Finite fields and functional reconstructions

Ray D. Sameshima

2016/09/23 $\sim$ 2016/11/21 23:48

Contents

Chapter 0

Preface

0.1 References

1. Scattering amplitudes over finite fields and multivariate functional reconstruction
   (Tiziano Peraro)
   `https://arxiv.org/pdf/1608.01902.pdf`

2. Haskell Language
   `www.haskell.org`

3. The Haskell Road to Logic, Maths and Programming
   (Kees Doets, Jan van Eijck)
   `http://homepages.cwi.nl/~jve/HR/`

4. Introduction to numerical analysis
   (Stoer Josef, Bulirsch Roland)

0.2 Set theoretical gadgets

0.2.1 Numbers

Here is a list of what we assumed that the readers are familiar with:

1. $\mathbb{N}$ (Peano axiom: $\emptyset, \mathrm{suc}$)

2. $\mathbb{Z}$

3. $\mathbb{Q}$

4. $\mathbb{R}$ (Dedekind cut)
5. $\mathbb{C}$

0.2.2 Algebraic structures

1. Monoid: $(\mathbb{N}, +), (\mathbb{N}, \times)$
2. Group: $(\mathbb{Z}, +), (\mathbb{Z}, \times)$
3. Ring: $\mathbb{Z}$
4. Field: $\mathbb{Q}$, $\mathbb{R}$ (continuous), $\mathbb{C}$ (algebraic closed)

0.3 Haskell language

From "A Brief, Incomplete and Mostly Wrong History of Programming Languages":[1]

> 1990 - A committee formed by Simon Peyton-Jones, Paul Hudak, Philip Wadler, Ashton Kutcher, and People for the Ethical Treatment of Animals creates Haskell, a pure, non-strict, functional language. Haskell gets some resistance due to the complexity of using monads to control side effects. Wadler tries to appease critics by explaining that "a monad is a monoid in the category of endofunctors, what's the problem?"

Figure 1: Haskell's logo, the combinations of λ and monad's bind `>>=`.

Haskell language is a standardized purely functional declarative statically typed programming language.

In declarative languages, we describe "what" or "definition" in its codes, however imperative languages, like C/C++, "how" or "procedure".

[1] `http://james-iry.blogspot.com/2009/05/brief-incomplete-and-mostly-wrong.html`

> Functional languages can be seen as 'executable mathematics'; the notation was designed to be as close as possible to the mathematical way of writing.[2]

Instead of loops, we use (implicit) recursions in functional language.[3]

```
> sum :: [Int] -> Int
> sum []     = 0
> sum (i:is) = i + sum is
```

[2] Algorithms: A Functional Programming Approach (Fethi A. Rabhi, Guy Lapalme)

[3] Of course, as a best practice, we should use higher order function (in this case `foldr` or `foldl`) rather than explicit recursions.

Chapter 1

Basics

We have assumed living knowledge on (axiomatic, i.e., ZFC) set theory, algebraic structures.

1.1 Finite fields

`Ffield.lhs`

`https://arxiv.org/pdf/1608.01902.pdf`

```
> module Ffield where
```

```
> import Data.Ratio
> import Data.Maybe
> import Data.Numbers.Primes
```

1.1.1 Rings

A ring $(R, +, *)$ is a structured set R with two binary operations

```
(+) :: R -> R -> R        (1.1)
(*) :: R -> R -> R        (1.2)
```

satisfying the following 3 (ring) axioms:

1. $(R, +)$ is an abelian, i.e., commutative group, i.e.,

$$\forall a, b, c \in R, (a + b) + c = a + (b + c) \quad \text{(associativity for +)} \quad (1.3)$$
$$\forall a, b, \in R, a + b = b + a \quad \text{(commutativity)} \quad (1.4)$$
$$\exists 0 \in R, \text{ s.t. } \forall a \in R, a + 0 = a \quad \text{(additive identity)} \quad (1.5)$$
$$\forall a \in R, \exists(-a) \in R \text{ s.t. } a + (-a) = 0 \quad \text{(additive inverse)} \quad (1.6)$$

2. $(R, *)$ is a monoid, i.e.,

$$\forall a, b, c \in R, (a * b) * c = a * (b * c) \quad \text{(associativity for } *) \quad (1.7)$$
$$\exists 1 \in R, \text{ s.t. } \forall a \in R, a * 1 = a = 1 * a \quad \text{(multiplicative identity)} \quad (1.8)$$

3. Multiplication is distributive w.r.t addition, i.e., $\forall a, b, c \in R$,

$$a * (b + c) = (a * b) + (a * c) \quad \text{(left distributivity)} \quad (1.9)$$
$$(a + b) * c = (a * c) + (b * c) \quad \text{(right distributivity)} \quad (1.10)$$

1.1.2 Fields

A field is a ring $(\mathbb{K}, +, *)$ whose non-zero elements form an abelian group under multiplication, i.e., $\forall r \in \mathbb{K}$,

$$r \neq 0 \Rightarrow \exists r^{-1} \in \mathbb{K} \text{ s.t. } r * r^{-1} = 1 = r^{-1} * r. \quad (1.11)$$

A field $\mathbb{K}$ is a finite field iff the underlying set $\mathbb{K}$ is finite. A field $\mathbb{K}$ is called infinite field iff the underlying set is infinite.

1.1.3 An example of finite rings $\mathbb{Z}_n$

Let $n(> 0) \in \mathbb{N}$ be a non-zero natural number. Then the quotient set

$$\mathbb{Z}_n := \mathbb{Z}/n\mathbb{Z} \quad (1.12)$$
$$\cong \{0, \cdots, (n-1)\} \quad (1.13)$$

with addition, subtraction and multiplication under modulo n is a ring.[1]

[1] Here we have taken an equivalence class,

$$0 \leq \forall k \leq (n-1), [k] := \{k + n * z | z \in \mathbb{Z}\} \quad (1.14)$$

1.1.4 Bézout's lemma

Consider $a, b \in \mathbb{Z}$ be nonzero integers. Then there exist $x, y \in \mathbb{Z}$ s.t.

$$a * x + b * y = \gcd(a, b), \tag{1.19}$$

where gcd is the greatest common divisor (function), see §1.1.5. We will prove this statement in §1.1.6.

1.1.5 Greatest common divisor

Before the proof, here is an implementation of gcd using Euclidean algorithm with Haskell language:

```
> -- Eucledian algorithm.
> myGCD :: Integral a => a -> a -> a
> myGCD a b
>   | b < 0 = myGCD a (-b)
> myGCD a b
>   | a == b = a
>   | b >  a = myGCD b a
>   | b <  a = myGCD (a-b) b
```

Example, by hands

Let us consider the gcd of 7 and 13. Since they are primes, the gcd should be 1. First it binds `a` with 7 and `b` with 13, and hit `b > a`.

$$\texttt{myGCD 7 13 == myGCD 13 7} \tag{1.20}$$

Then it hits main line:

$$\texttt{myGCD 13 7 == myGCD (13-7) 7} \tag{1.21}$$

---

with the following operations:

$$\begin{aligned} [k] + [l] &:= [k + l] && (1.15) \\ [k] * [l] &:= [k * l] && (1.16) \end{aligned}$$

This is equivalent to take modular n:

$$\begin{aligned} (k \mod n) + (l \mod n) &:= (k + l \mod n) && (1.17) \\ (k \mod n) * (l \mod n) &:= (k * l \mod n). && (1.18) \end{aligned}$$

In order to go to next step, Haskell evaluate $(13 - 7)$,[2] and

$$
\begin{aligned}
\texttt{myGCD (13-7) 7} &\ \texttt{==}\ \texttt{myGCD 6 7} && (1.22)\\
&\ \texttt{==}\ \texttt{myGCD 7 6} && (1.23)\\
&\ \texttt{==}\ \texttt{myGCD (7-6) 6} && (1.24)\\
&\ \texttt{==}\ \texttt{myGCD 1 6} && (1.25)\\
&\ \texttt{==}\ \texttt{myGCD 6 1} && (1.26)
\end{aligned}
$$

Finally it ends with 1:

$$\texttt{myGCD 1 1 == 1} \tag{1.27}$$

As another example, consider 15 and 25:

$$
\begin{aligned}
\texttt{myGCD 15 25} &\ \texttt{==}\ \texttt{myGCD 25 15} && (1.28)\\
&\ \texttt{==}\ \texttt{myGCD (25-15) 15} && (1.29)\\
&\ \texttt{==}\ \texttt{myGCD 10 15} && (1.30)\\
&\ \texttt{==}\ \texttt{myGCD 15 10} && (1.31)\\
&\ \texttt{==}\ \texttt{myGCD (15-10) 10} && (1.32)\\
&\ \texttt{==}\ \texttt{myGCD 5 10} && (1.33)\\
&\ \texttt{==}\ \texttt{myGCD 10 5} && (1.34)\\
&\ \texttt{==}\ \texttt{myGCD (10-5) 5} && (1.35)\\
&\ \texttt{==}\ \texttt{myGCD 5 5} && (1.36)\\
&\ \texttt{==}\ \texttt{5} && (1.37)
\end{aligned}
$$

Example, with Haskell

Let us check simple example using Haskell:

```
*Ffield> myGCD 7 13
1
*Ffield> myGCD 7 14
7
*Ffield> myGCD (-15) (20)
5
*Ffield> myGCD (-299) (-13)
13
```

[2] Since Haskell language adopts lazy evaluation, i.e., call by need, not call by name.

The final result is from

```
*Ffield> 13*23
299
```

1.1.6 Extended Euclidean algorithm

Here we treat the extended Euclidean algorithm, this is a constructive solution for Bézout's lemma.

As intermediate steps, this algorithm makes sequences of integers $\{r_i\}_i$, $\{s_i\}_i$, $\{t_i\}_i$ and quotients $\{q_i\}_i$ as follows. The base cases are

$$\begin{aligned}(r_0, s_0, t_0) &:= (a, 1, 0) && (1.38)\\ (r_1, s_1, t_1) &:= (b, 0, 1) && (1.39)\end{aligned}$$

and inductively, for $i \geq 2$,

$$\begin{aligned}q_i &:= \mathrm{quot}(r_{i-2}, r_{i-1}) && (1.40)\\ r_i &:= r_{i-2} - q_i * r_{i-1} && (1.41)\\ s_i &:= s_{i-2} - q_i * s_{i-1} && (1.42)\\ t_i &:= t_{i-2} - q_i * t_{i-1}. && (1.43)\end{aligned}$$

The termination condition[3] is

$$r_k = 0 \tag{1.44}$$

for some $k \in \mathbb{N}$ and

$$\begin{aligned}\gcd(a, b) &= r_{k-1} && (1.45)\\ x &= s_{k-1} && (1.46)\\ y &= t_{k-1}. && (1.47)\end{aligned}$$

Proof

By definition,

$$\begin{aligned}\gcd(r_{i-1}, r_i) &= \gcd(r_{i-1}, r_{i-2} - q_i * r_{i-1}) && (1.48)\\ &= \gcd(r_{i-1}, r_{i-2}) && (1.49)\end{aligned}$$

[3] This algorithm will terminate eventually, since the sequence $\{r_i\}_i$ is non-negative by definition of q_i, but strictly decreasing. Therefore, $\{r_i\}_i$ will meet 0 in finite step k.

and this implies

$$\gcd(a, b) =: \gcd(r_0, r_1) = \cdots = \gcd(r_{k-1}, 0), \tag{1.50}$$

i.e.,

$$r_{k-1} = \gcd(a, b). \tag{1.51}$$

Next, for $i = 0, 1$ observe

$$a * s_i + b * t_i = r_i. \tag{1.52}$$

Let $i \geq 2$, then

$$\begin{aligned} r_i &= r_{i-2} - q_i * r_{i-1} && (1.53)\\ &= a * s_{i-2} + b * t_{i-2} - q_i * (a * s_{i-1} + b * t_{i-1}) && (1.54)\\ &= a * (s_{i-2} - q_i * s_{i-1}) + b * (t_{i-2} - q_i * t_{i-1}) && (1.55)\\ &=: a * s_i + b * t_i. && (1.56) \end{aligned}$$

Therefore, inductively we get

$$\gcd(a, b) = r_{k-1} = a * s_{k-1} + b * t_{k-1}. =: a * x + b * y. \tag{1.57}$$

This prove Bézout's lemma.
■

Haskell implementation

Here I use lazy lists for intermediate lists of qs, rs, ss, ts, and pick up (second) last elements for the results.

```
Here we would like to implement the extended Euclidean algorithm.
See the algorithm, examples, and pseudo code at:

  https://en.wikipedia.org/wiki/Extended_Euclidean_algorithm

> exGCD' :: (Integral n) => n -> n -> ([n], [n], [n], [n])
> exGCD' a b = (qs, rs, ss, ts)
>   where
>     qs = zipWith quot rs (tail rs)
>     rs = takeUntil (==0) r'
>     r' = steps a b
```

```
>     ss = steps 1 0
>     ts = steps 0 1
>     steps a b = rr
>       where
>         rr@(_:rs) = a:b: zipWith (-) rr (zipWith (*) qs rs)
>
> takeUntil :: (a -> Bool) -> [a] -> [a]
> takeUntil p = foldr func []
>   where
>     func x xs
>       | p x = []
>       | otherwise = x : xs
```

Here we have used so called lazy lists, and higher order function[4]. The gcd of a and b should be the last element of second list `rs`, and our targets (s, t) are second last elements of last two lists `ss` and `ts`. The following example is from wikipedia:

```
*Ffield> exGCD' 240 46
([5,4,1,1,2],[240,46,10,6,4,2],[1,0,1,-4,5,-9,23],[0,1,-5,21,-26,47,-120])
```

Look at the second lasts of `[1,0,1,-4,5,-9,23]`, `[0,1,-5,21,-26,47,-120]`, i.e., -9 and 47:

```
*Ffield> gcd 240 46
2
*Ffield> 240*(-9) + 46*(47)
2
```

It works, and we have other simpler examples:

```
*Ffield> exGCD' 15 25
([0,1,1,2],[15,25,15,10,5],[1,0,1,-1,2,-5],[0,1,0,1,-1,3])
*Ffield> 15 * 2 + 25*(-1)
5
*Ffield> exGCD' 15 26
([0,1,1,2,1,3],[15,26,15,11,4,3,1],[1,0,1,-1,2,-5,7,-26],[0,1,0,1,-1,3,-4,15])
*Ffield> 15*7 + (-4)*26
1
```

[4] Naively speaking, the function whose inputs and/or outputs are functions is called a higher order function.

Now what we should do is extract gcd of a and b, and (x, y) from the tuple of lists:

```
> -- a*x + b*y = gcd a b
> exGCD :: Integral t => t -> t -> (t, t, t)
> exGCD a b = (g, x, y)
>   where
>     (_,r,s,t) = exGCD' a b
>     g = last r
>     x = last . init $ s
>     y = last . init $ t
```

where the underscore _ is a special symbol in Haskell that hits every pattern, since we do not need the quotient list. So, in order to get gcd and (x, y) we don't need quotients list.

```
*Ffield> exGCD 46 240
(2,47,-9)
*Ffield> 46*47 + 240*(-9)
2
*Ffield> gcd 46 240
2
```

1.1.7 Coprime as a binary relation

Let us define a binary relation as follows:

```
coprime :: Integral a => a -> a -> Bool
coprime a b = (gcd a b) == 1
```

1.1.8 Corollary (Inverses in $\mathbb{Z}_n$)

For a non-zero element

$$a \in \mathbb{Z}_n, \tag{1.58}$$

there is a unique number

$$b \in \mathbb{Z}_n \text{ s.t. } ((a * b) \mod n) = 1 \tag{1.59}$$

iff a and n are coprime.

Proof

From Bézout's lemma, a and n are coprime iff

$$\exists s, t \in \mathbb{Z}, a * s + n * t = 1. \tag{1.60}$$

Therefore

$$\begin{aligned} a \text{ and } n \text{ are coprime} &\Leftrightarrow \exists s, t \in \mathbb{Z}, a * s + n * t = 1 && (1.61) \\ &\Leftrightarrow \exists s, t' \in \mathbb{Z}, a * s = 1 + n * t'. && (1.62) \end{aligned}$$

This s, by taking its modulo n is our $b = a^{-1}$:

$$a * s = 1 \mod n. \tag{1.63}$$

We will make a Haskell implementation in §1.1.9.
■

1.1.9 Corollary (Finite field $\mathbb{Z}_p$)

If p is prime, then

$$\mathbb{Z}_p := \{0, \cdots, (p-1)\} \tag{1.64}$$

with addition, subtraction and multiplication under modulo n is a field.

Proof

It suffices to show that

$$\forall a \in \mathbb{Z}_p, a \neq 0 \Rightarrow \exists a^{-1} \in \mathbb{K} \text{ s.t. } a * a^{-1} = 1 = a^{-1} * a, \tag{1.65}$$

but since p is prime, and

$$\forall a \in \mathbb{Z}_p, a \neq 0 \Rightarrow \texttt{gcd a p == 1} \tag{1.66}$$

so all non-zero element has its inverse in $\mathbb{Z}_p$.
■

Example and implementation

Let us pick 11 as a prime and consider $\mathbb{Z}_{11}$:

```
Example Z_{11}

*Ffield> isField 11
True
*Ffield> map (exGCD 11) [0..10]
[(11,1,0),(1,0,1),(1,1,-5),(1,-1,4),(1,-1,3)
,(1,1,-2),(1,-1,2),(1,2,-3),(1,3,-4),(1,-4,5),(1,1,-1)
]
```

This list of three-tuple let us know the candidate of inverse. Take the last one, `(1,1,-1)`. This is the image of `exGcd 11 10`, and

$$1 = 10 * 1 + 11 * (-1) \tag{1.67}$$

holds. This suggests -1 is a candidate of the inverse of 10 in $\mathbb{Z}_{11}$:

$$\begin{aligned} 10^{-1} &= -1 \mod 11 && (1.68) \\ &= 10 \mod 11 && (1.69) \end{aligned}$$

In fact,

$$10 * 10 = 11 * 9 + 1. \tag{1.70}$$

So, picking up the third elements in tuple and zipping with nonzero elements, we have a list of inverses:

```
*Ffield> map ((`mod` 11) . (\(_,_,x)->x) . exGCD 11) [1..10]
[1,6,4,3,9,2,8,7,5,10]

We get non-zero elements with its inverse:

*Ffield> zip [1..10] it
[(1,1),(2,6),(3,4),(4,3),(5,9),(6,2),(7,8),(8,7),(9,5),(10,10)]
```

Let us generalize these flow into a function[5]:

[5] From `https://hackage.haskell.org/package/base-4.9.0.0/docs/Data-Maybe.html`:

> The Maybe type encapsulates an optional value. A value of type Maybe a either contains a value of type a (represented as Just a), or it is empty (represented as Nothing). Using Maybe is a good way to deal with errors or exceptional cases without resorting to drastic measures such as error.

```
> -- a^{-1} (in Z_p) == a `inversep` p
> inversep :: Integral a => a -> a -> Maybe a
> a `inversep` p = let (g,x,_) = exGCD a p in
>   if (g == 1) then Just (x `mod` p)
>               else Nothing
```

This `inversep` function returns the inverse with respect to second argument, if they are coprime, i.e. gcd is 1. So the second argument should not be prime.

```
> inversesp :: Integral a => a -> [Maybe a]
> inversesp p = map (`inversep` p) [1..(p-1)]

  *Ffield> inversesp 11
  [Just 1,Just 6,Just 4,Just 3,Just 9,Just 2,Just 8,Just 7,Just 5,Just 10]
  *Ffield> inversesp 9
  [Just 1,Just 5,Nothing,Just 7,Just 2,Nothing,Just 4,Just 8]
```

1.2 Rational number reconstruction

1.2.1 A map from $\mathbb{Q}$ to $\mathbb{Z}_p$

Let p be a prime. Now we have a map

$$- \mod p : \mathbb{Z} \to \mathbb{Z}_p; a \mapsto (a \mod p), \tag{1.71}$$

and a natural inclusion (or a forgetful map)[6]

$$¿ : \mathbb{Z}_p \hookrightarrow \mathbb{Z}. \tag{1.73}$$

Then we can define a map

$$- \mod p : \mathbb{Q} \to \mathbb{Z}_p \tag{1.74}$$

by[7]

$$q = \frac{a}{b} \mapsto (q \mod p) := \left(\left(a \times ¿ \left(b^{-1} \mod p \right) \right) \mod p \right). \tag{1.75}$$

[6] By introducing this forgetful map, we can use

$$\times : (\mathbb{Z}, \mathbb{Z}) \to \mathbb{Z} \tag{1.72}$$

of normal product on $\mathbb{Z}$.

[7] This is an example of operator overloadings.

Example and implementation

An easy implementation is the followings:[8]

```
A map from Q to Z_p.

> -- p should be prime.
> modp :: Integral a => Ratio a -> a -> a
> q `modp` p = (a * (bi `mod` p)) `mod` p
>   where
>     (a,b) = (numerator q, denominator q)
>     bi = fromJust (b `inversep` p)
```

Let us consider a rational number $\frac{3}{7}$ on a finite field $\mathbb{Z}_{11}$:

```
Example: on Z_{11}
Consider (3 % 7).

  *Ffield> let q = 3%7
  *Ffield> 3 `mod` 11
  3
  *Ffield> 7 `inversep` 11
  Just 8
  *Ffield> (3*8) `mod` 11
  2
```

For example, pick 7:

```
  *Ffield> 7*8 == 11*5+1
  True
```

Therefore, on $\mathbb{Z}_{11}$, $(7^{-1} \mod 11)$ is equal to $(8 \mod 11)$ and

$$\begin{aligned} \frac{3}{7} \in \mathbb{Q} \quad \mapsto \quad & (3 \times ¿(7^{-1} \mod 11) \mod 11) & (1.78) \\ & = (3 \times 8) \mod 11 & (1.79) \\ & = 24 \mod 11 & (1.80) \\ & = 2 \mod 11. & (1.81) \end{aligned}$$

[8] The backquotes makes any binary function infix operator. For example,

```
add 1 2 == 1 `add` 2        (1.76)
```

Similarly, use parenthesis we can use an infix binary operator to a function:

```
(+) 1 2 == 1 + 2        (1.77)
```

Haskell returns the same result

```
*Ffield> q `modp` 11
2
```

and consistent.

1.2.2 Reconstruction from $\mathbb{Z}_p$ to $\mathbb{Q}$

Consider a rational number q and its image $a \in \mathbb{Z}_p$.

$$a := q \mod p \tag{1.82}$$

The extended Euclidean algorithm can be used for guessing a rational number q from the images $a := q \mod p$ of several primes p's.

At each step, the extended Euclidean algorithm satisfies eq.(1.52).

$$a * s_i + p * t_i = r_i \tag{1.83}$$

Therefore

$$r_i = a * s_i \mod p. \tag{1.84}$$

Hence $\frac{r_i}{s_i}$ is a possible guess for q. We take

$$r_i^2, s_i^2 < p \tag{1.85}$$

as the termination condition for this reconstruction.

Haskell implementation

Let us first try to reconstruct from the image ($\frac{1}{3} \mod p$) of some prime p. Here we have chosen three primes

```
Reconstruction Z_p -> Q
  *Ffield> let q = (1%3)
  *Ffield> take 3 $ dropWhile (<100) primes
  [101,103,107]
```

The images are basically given by the first elements of second lists (s_0's):

```
*Ffield> q `modp` 101
34
*Ffield> let try x = exGCD' (q `modp` x) x
*Ffield> try 101
([0,2,1,33],[34,101,34,33,1],[1,0,1,-2,3,-101],[0,1,0,1,-1,34])
*Ffield> try 103
([0,1,2,34],[69,103,69,34,1],[1,0,1,-1,3,-103],[0,1,0,1,-2,69])
*Ffield> try 107
([0,2,1,35],[36,107,36,35,1],[1,0,1,-2,3,-107],[0,1,0,1,-1,36])
```

Look at the first hit of termination condition eq.(1.85), $r_4 = 1$ and $s_4 = 3$. They give us the same guess $\frac{1}{3}$, and that the reconstructed number.

From the above observations we can make a simple "guess" function:

```
> guess :: Integral t =>
>          (t, t)       -- (q `modp` p, p)
>       -> (Ratio t, t)
> guess (a, p) = let (_,rs,ss,_) = exGCD' a p in
>   (select rs ss p, p)
>     where
>       select :: Integral t => [t] -> [t] -> t -> Ratio t
>       select [] _ _ = 0%1
>       select (r:rs) (s:ss) p
>         | s /= 0 && r^2 <= p && s^2 <= p = r%s
>         | otherwise = select rs ss p
```

We have put a list of big primes as follows.

```
> -- Hard code of big primes
> -- For chinese reminder theorem we declare it as [Integer].
> bigPrimes :: [Integer]
> bigPrimes = dropWhile (< 897473) $ takeWhile (< 978948) primes
```

We choose 3 times match as the termination condition.

```
> matches3 :: Eq a => [a] -> a
> matches3 (a:bb@(b:c:cs))
>   | a == b && b == c = a
>   | otherwise        = matches3 bb
```

Finally, we can check our gadgets.

What we know is a list of (q `modp` p) and prime p for several (big) primes.

```
*Ffield> let q = 10%19
*Ffield> let knownData = zip (map (modp q) bigPrimes) bigPrimes
*Ffield> take 3 knownData
[(614061,897473),(377894,897497),(566842,897499)]
*Ffield> matches3 $  map (fst . guess) knownData
10 % 19
```

The following is the function we need, its input is the list of tuple which first element is the image in $\mathbb{Z}_p$ and second element is that prime p.

```
> reconstruct :: Integral a =>
>                [(a, a)]  -- :: [(Z_p, primes)]
>             -> Ratio a
> reconstruct aps = matches3 $ map (fst . guess) aps
```

```
Here is a naive test:
  > let qs = [1 % 3,10 % 19,41 % 17,30 % 311,311 % 32
             ,869 % 232,778 % 123,331 % 739]
  > let modmap q = zip (map (modp q) bigPrimes) bigPrimes
  > let longList = map modmap qs
  > map reconstruct longList
  [1 % 3,10 % 19,41 % 17,30 % 311,311 % 32
  ,869 % 232,778 % 123,331 % 739]
  > it == qs
  True
```

For later use, let us define

```
> imagesAndPrimes ::  Rational-> [(Integer, Integer)]
> imagesAndPrimes q = zip (map (modp q) bigPrimes) bigPrimes
```

to generate a list of images (of our target rational number) in Z_p and the base primes.

As another example, we have slightly involved function:

```
> matches3' :: Eq a => [(a, t)] -> (a, t)
> matches3' (a0@(a,_):bb@((b,_):(c,_):cs))
>   | a == b && b == c = a0
>   | otherwise        = matches3' bb
```

Let us see the first good guess, Haskell tells us that in order to reconstruct, say $\frac{331}{739}$, we should take three primes start from 614693:

```
*Ffield> let knowData q = zip (map (modp q) primes) primes
*Ffield> matches3' $ map guess $ knowData (331%739)
(331 % 739,614693)
(18.31 secs, 12,393,394,032 bytes)

*Ffield> matches3' $ map guess $ knowData (11%13)
(11 % 13,311)
(0.02 secs, 2,319,136 bytes)
*Ffield> matches3' $ map guess $ knowData (1%13)
(1 % 13,191)
(0.01 secs, 1,443,704 bytes)
*Ffield> matches3' $ map guess $ knowData (1%3)
(1 % 3,13)
(0.01 secs, 268,592 bytes)
*Ffield> matches3' $ map guess $ knowData (11%31)
(11 % 31,1129)
(0.03 secs, 8,516,568 bytes)
*Ffield> matches3' $ map guess $ knowData (12%312)
(1 % 26,709)
```

A problem

Since our choice of `bigPrimes` are order 10^6, our reconstruction can fail for rational numbers of

$$\frac{O(10^3)}{O(10^3)}, \tag{1.86}$$

say

```
*Ffield> let q = 895%922
*Ffield> let knownData = imagesAndPrimes q
*Ffield> take 4 knownData
[(882873,897473)
,(365035,897497)
,(705735,897499)
,(511060,897517)
]
*Ffield> map guess it
[((-854) % 123,897473)
,((-656) % 327,897497)
,((-192) % 805,897499)
```

```
  ,((-491) % 497,897517)
  ]
```

We can solve this by introducing the following theorem.

1.2.3 Chinese remainder theorem

From wikipedia[9]

> There are certain things whose number is unknown. If we count them by threes, we have two left over; by fives, we have three left over; and by sevens, two are left over. How many things are there?

Here is a solution with Haskell:

```
*Ffield> let lst = [n|n<-[0..], mod n 3==2, mod n 5==3, mod n 7==2]
*Ffield> head lst
23
```

We define an infinite list of natural numbers that satisfy

$$n \mod 3 = 2, n \mod 5 = 3, n \mod 7 = 2. \tag{1.87}$$

Then take the first element, and this is the answer.

Claim

The statement for binary case is the following. Let $n_1, n_2 \in \mathbb{Z}$ be coprime, then for arbitrary $a_1, a_2 \in \mathbb{Z}$, the following a system of equations

$$x = a_1 \mod n_1 \tag{1.88}$$
$$x = a_2 \mod n_2 \tag{1.89}$$

have a unique solution modular $n_1 * n_2$[10].

[9] https://en.wikipedia.org/wiki/Chinese_remainder_theorem

[10] Note that, this is equivalent that there is a unique solution a in

$$0 \leq a < n_1 \times n_2. \tag{1.90}$$

Proof

(existence) With §1.1.6, there are $m_1, m_2 \in \mathbb{Z}$ s.t.

$$n_1 * m_1 + n_2 * m_2 = 1. \tag{1.91}$$

Now we have

$$\begin{aligned} n_1 * m_1 &= 1 \mod n_2 & (1.92)\\ n_2 * m_2 &= 1 \mod n_1 & (1.93) \end{aligned}$$

that is[11]

$$\begin{aligned} m_1 &= n_1^{-1} \mod n_2 & (1.94)\\ m_2 &= n_2^{-1} \mod n_1. & (1.95) \end{aligned}$$

Then

$$a := a_1 * n_2 * m_2 + a_2 * n_1 * m_1 \mod (n_1 * n_2) \tag{1.96}$$

is a solution.

(uniqueness) If a' is also a solution, then

$$\begin{aligned} a - a' &= 0 \mod n_1 & (1.97)\\ a - a' &= 0 \mod n_2. & (1.98) \end{aligned}$$

Since n_1 and n_2 are coprime, i.e., no common divisors, this difference is divisible by $n_1 * n_2$, and

$$a - a' = 0 \mod (n_1 * n_2). \tag{1.99}$$

Therefore, the solution is unique modular $n_1 * n_2$.
■

Haskell implementation

Let us see how our naive `guess` function fail one more time:

`Chinese Remainder Theorem, and its usage`

```
*Ffield> let q = 895%922
*Ffield> let knownData = imagesAndPrimes q
```

[11] Here we have used slightly different notions from 1. m_1 in 1 is our m_2 times our n_2.

```
*Ffield> let [(a1,p1),(a2,p2)] = take 2 knownData
*Ffield> take 2 knownData
[(882873,897473),(365035,897497)]
*Ffield> map guess it
[((-854) % 123,897473),((-656) % 327,897497)]
```

It suffices to make a binary version of Chinese Remainder theorem in Haskell:

```
> crtRec' :: Integral t => (t, t) -> (t, t) -> (t, t)
> crtRec' (a1,p1) (a2,p2) = (a,p)
>   where
>     a = (a1*p2*m2 + a2*p1*m1) `mod` p
>     m1 = fromJust (p1 `inversep` p2)
>     m2 = fromJust (p2 `inversep` p1)
>     p = p1*p2
```

`crtRec'` function takes two tuples of image in $\mathbb{Z}_p$ and primes, and returns these combination. Now let us fold.

```
> pile :: (a -> a -> a) -> [a] -> [a]
> pile f [] = []
> pile f dd@(d:ds) = d : zipWith' f (pile f dd) ds
```

Schematically, this `pile f` function takes

$$[d_0, d_1, d_2, d_3, \cdots] \tag{1.100}$$

and returns

$$[d_0, f(d_0, d_1), f(f(d_0, d_1), d_2), f(f(f(d_0, d_1), d_2), d_3), \cdots] \tag{1.101}$$

We have used another higher order function which is slightly modified from standard definition:

```
> -- Strict zipWith, from:
> --   http://d.hatena.ne.jp/kazu-yamamoto/touch/20100624/1277348961
> zipWith' :: (a -> b -> c) -> [a] -> [b] -> [c]
> zipWith' f (a:as) (b:bs) = (x `seq` x) : zipWith' f as bs
>   where x = f a b
> zipWith' _ _      _      = []
```

Let us check our implementation.

```
*Ffield> let q = 895%922
*Ffield> let knownData = imagesAndPrimes q
*Ffield> take 4 knownData
[(882873,897473)
,(365035,897497)
,(705735,897499)
,(511060,897517)
]
*Ffield> pile crtRec' it
[(882873,897473)
,(86488560937,805479325081)
,(397525881357811624,722916888780872419)
,(2329314482599662599376l4,648830197267942270883623)
]
*Ffield> map guess it
[((-854) % 123,897473)
,(895 % 922,805479325081)
,(895 % 922,722916888780872419)
,(895 % 922,648830197267942270883623)
```

So on a product ring $\mathbb{Z}_{805479325081}$, we get the right answer.

1.2.4 `recCRT`: from image in $\mathbb{Z}_p$ to rational number

From above discussion, here we can define a function which takes a list of images in $\mathbb{Z}_p$ and returns the rational number. What we do is, basically, to take a list of image (of our target rational number) and primes, then applying Chinese Remainder theorem recursively, return several guess of rational number.

```
> recCRT :: Integral a => [(a,a)] -> Ratio a
> recCRT = reconstruct . pile crtRec'

> recCRT' = matches3' . map guess . pile crtRec'
```

```
*Ffield> let q = 895%922
*Ffield> let knownData = imagesAndPrimes q
*Ffield> recCRT knownData
895 % 922
*Ffield> recCRT' knownData
(895 % 922,805479325081)
```

Here is some random checks and results.

```
todo: use QuickCheck

> trial = do
>   n <- randomRIO (0,10000) :: IO Integer
>   d <- randomRIO (1,10000) :: IO Integer
>   let q = (n%d)
>   putStrLn $ "input: " ++ show q
>   return $ recCRT' . imagesAndPrimes $ q

  *Ffield> trial
  input: 1080 % 6931
  (1080 % 6931,805479325081)
  *Ffield> trial
  input: 2323 % 1248
  (2323 % 1248,805479325081)
  *Ffield> trial
  input: 6583 % 1528
  (6583 % 1528,805479325081)
  *Ffield> trial
  input: 721 % 423
  (721 % 423,897473)
  *Ffield> trial
  input: 9967 % 7410
  (9967 % 7410,805479325081)
```

1.3 Polynomials and rational functions

The following discussion on an arbitrary field $\mathbb{K}$.

1.3.1 Notations

Let $n \in \mathbb{N}$ be positive. We use multi-index notation:

$$\alpha = (\alpha_1, \cdots, \alpha_n) \in \mathbb{N}^n. \tag{1.102}$$

A monomial is defined as

$$z^\alpha := \prod_i z_i^{\alpha_i}. \tag{1.103}$$

The total degree of this monomial is given by

$$|\alpha| := \sum_i \alpha_i. \tag{1.104}$$

1.3.2 Polynomials and rational functions

Let $\mathbb{K}$ be a field. Consider a map

$$f : \mathbb{K}^n \to \mathbb{K}; z \mapsto f(z) := \sum_\alpha c_\alpha z^\alpha, \tag{1.105}$$

where

$$c_\alpha \in \mathbb{K}. \tag{1.106}$$

We call the value $f(z)$ at the dummy $z \in \mathbb{K}^n$ a polynomial:

$$f(z) := \sum_\alpha c_\alpha z^\alpha. \tag{1.107}$$

We denote

$$\mathbb{K}[z] := \left\{ \sum_\alpha c_\alpha z^\alpha \right\} \tag{1.108}$$

as the ring of all polynomial functions in the variable z with $\mathbb{K}$-coefficients.

Similarly, a rational function can be expressed as a ratio of two polynomials $p(z), q(z) \in \mathbb{K}[z]$:

$$\frac{p(z)}{q(z)} = \frac{\sum_\alpha n_\alpha z^\alpha}{\sum_\beta d_\beta z^\beta}. \tag{1.109}$$

We denote

$$\mathbb{K}(z) := \left\{ \frac{\sum_\alpha n_\alpha z^\alpha}{\sum_\beta d_\beta z^\beta} \right\} \tag{1.110}$$

as the field of rational functions in the variable z with $\mathbb{F}$-coefficients. Similar to fractional numbers, there are several equivalent representation of a rational function, even if we simplify with gcd. However there still is an overall constant ambiguity. To have a unique representation, usually we put the lowest degree of term of the denominator to be 1.

1.3.3 As data, coefficients list

We can identify a polynomial

$$\sum_{\alpha} c_\alpha z^\alpha \tag{1.111}$$

as a set of coefficients

$$\{c_\alpha\}_\alpha. \tag{1.112}$$

Similarly, for a rational function, we can identify

$$\frac{\sum_\alpha n_\alpha z^\alpha}{\sum_\beta d_\beta z^\beta} \tag{1.113}$$

as an ordered pair of coefficients

$$(\{n_\alpha\}_\alpha, \{d_\beta\}_\beta). \tag{1.114}$$

However, there still is an overall factor ambiguity even after gcd simplifications.

1.4 Haskell implementation of univariate polynomials

Here we basically follows some part of §9 of ref.3, and its addendum[12].

```
Univariate.lhs

> module Univariate where
> import Data.Ratio
> import Polynomials
```

1.4.1 A polynomial as a list of coefficients

Let us start `instance` declaration, which enable us to use basic arithmetics, e.g., addition and multiplication.

[12] See `http://homepages.cwi.nl/~jve/HR/PolAddendum.pdf`

```
-- Polynomials.hs
-- http://homepages.cwi.nl/~jve/rcrh/Polynomials.hs

module Polynomials where

default (Integer, Rational, Double)

-- polynomials, as coefficients lists
instance (Num a, Ord a) => Num [a] where
  fromInteger c = [fromInteger c]
  -- operator overloading
  negate []     = []
  negate (f:fs) = (negate f) : (negate fs)

  signum [] = []
  signum gs
    | signum (last gs) < (fromInteger 0) = negate z
    | otherwise = z

  abs [] = []
  abs gs
    | signum gs == z = gs
    | otherwise      = negate gs

  fs     + []     = fs
  []     + gs     = gs
  (f:fs) + (g:gs) = f+g : fs+gs

  fs     * []     = []
  []     * gs     = []
  (f:fs) * gg@(g:gs) = f*g : (f .* gs + fs * gg)

delta :: (Num a, Ord a) => [a] -> [a]
delta = ([1,-1] *)

shift :: [a] -> [a]
shift = tail

p2fct :: Num a => [a] -> a -> a
p2fct [] x = 0
```

```
p2fct (a:as) x = a + (x * p2fct as x)

comp :: (Eq a, Num a, Ord a) => [a] -> [a] -> [a]
comp _      []      = error ".."
comp []     _       = []
comp (f:fs) g0@(0:gs) = f : gs * (comp fs g0)
comp (f:fs) gg@(g:gs) = ([f] + [g] * (comp fs gg))
                      + (0 : gs * (comp fs gg))

deriv :: Num a => [a] -> [a]
deriv []     = []
deriv (f:fs) = deriv1 fs 1
  where
    deriv1 []     _ = []
    deriv1 (g:gs) n = n*g : deriv1 gs (n+1)
```

Note that the above operators are overloaded, say `(*)`, `f*g` is a multiplication of two numbers but `fs*gg` is a multiplication of two list of coefficients. We can not extend this overloading to scalar multiplication, since Haskell type system takes the operands of `(*)` the same type

$$\texttt{(*) :: Num a => a -> a -> a} \tag{1.115}$$

```
> -- scalar multiplication
> infixl 7 .*
> (.*) :: Num a => a -> [a] -> [a]
> c .* []     = []
> c .* (f:fs) = c*f : c .* fs
```

Let us see few examples. If we take a scalar multiplication, say

$$3 * \left(1 + 2z + 3z^2 + 4z^3\right) \tag{1.116}$$

the result should be

$$3 * \left(1 + 2z + 3z^2 + 4z^3\right) = 3 + 6z + 9z^2 + 12z^3 \tag{1.117}$$

In Haskell

```
*Univariate> 3 .* [1,2,3,4]
[3,6,9,12]
```

and this is exactly same as map with section:

```
*Univariate> map (3*) [1,2,3,4]
[3,6,9,12]
```

When we multiply two polynomials, say

$$(1 + 2z) * (3 + 4z + 5z^2 + 6z^3) \tag{1.118}$$

the result should be

$$\begin{aligned} (1 + 2z) * (3 + 4z + 5z^2 + 6z^3) &= 1 * (3 + 4z + 5z^2 + 6z^3) + 2z * (3 + 4z + 5z^2 + 6z^3) \\ &= 3 + (4 + 2 * 3)z + (5 + 2 * 4)z^2 + (6 + 2 * 5)z^3 + 2 * 6z^4 \\ &= 3 + 10z + 13z^2 + 16z^3 + 12z^4 \end{aligned} \tag{1.119}$$

In Haskell,

```
*Univariate> [1,2] * [3,4,5,6]
[3,10,13,16,12]
```

Now the (dummy) variable is given as

```
> -- z of f(z), variable
> z :: Num a => [a]
> z = [0,1]
```

A polynomial of degree R is given by a finite sum of the following form:

$$f(z) := \sum_{i=0}^{R} c_i z^i. \tag{1.120}$$

Therefore, it is natural to represent $f(z)$ by a list of coefficient $\{c_i\}_i$. Here is the translator from the coefficient list to a polynomial function:

```
> p2fct :: Num a => [a] -> a -> a
> p2fct [] x = 0
> p2fct (a:as) x = a + (x * p2fct as x)
```

This gives us[13]

```
*Univariate> take 10 $ map (p2fct [1,2,3]) [0..]
[1,6,17,34,57,86,121,162,209,262]
*Univariate> take 10 $ map (\n -> 1+2*n+3*n^2) [0..]
[1,6,17,34,57,86,121,162,209,262]
```

[13] Here we have used lambda, or so called anonymous function. From `http://learnyouahaskell.com/higher-order-functions`

To make a lambda, we write a \(because it kind of looks like the greek

1.4.2 Difference analysis

We do not know in general this canonical form of the polynomial, nor the degree. That means, what we can access is the graph of f, i.e., the list of inputs and outputs. Without loss of generality, we can take

$$\texttt{[0..]} \tag{1.123}$$

as the input data. Usually we take a finite sublist of this, but we assume it is sufficiently long. The outputs should be

$$\texttt{map f [0..] = [f 0, f 1 ..]} \tag{1.124}$$

For example

```
*Univariate> take 10 $ map (\n -> n^2+2*n+1) [0..]
[1,4,9,16,25,36,49,64,81,100]
```

Let us consider the difference sequence

$$\Delta(f)(n) := f(n+1) - f(n). \tag{1.125}$$

Its Haskell version is

```
> -- difference analysis
> difs :: (Num a) => [a] -> [a]
> difs [] = []
> difs [_] = []
> difs (i:jj@(j:js)) = j-i : difs jj
```

This gives

```
*Univariate> difs [1,4,9,16,25,36,49,64,81,100]
[3,5,7,9,11,13,15,17,19]
*Univariate> difs [3,5,7,9,11,13,15,17,19]
[2,2,2,2,2,2,2,2]
```

letter lambda if you squint hard enough) and then we write the parameters, separated by spaces.

For example,

$$f(x) := x^2 + 1 \tag{1.121}$$
$$f := \lambda x.x^2 + 1 \tag{1.122}$$

are the same definition.

We claim that if $f(z)$ is a polynomial of degree R, then $\Delta(f)(z)$ is a polynomial of degree $R-1$. Since the degree is given, we can write $f(z)$ in canonical form

$$f(n) = \sum_{i=0}^{R} c_i n^i \tag{1.126}$$

and

$$\begin{aligned}
\Delta(f)(n) &:= f(n+1) - f(n) && (1.127)\\
&= \sum_{i=0}^{R} c_i \left\{(n+1)^i - n^i\right\} && (1.128)\\
&= \sum_{i=1}^{R} c_i \left\{(n+1)^i - n^i\right\} && (1.129)\\
&= \sum_{i=1}^{R} c_i \left\{i * n^{i-1} + O(n^{i-2})\right\} && (1.130)\\
&= c_R * R * n^{R-1} + O(n^{R-2}) && (1.131)
\end{aligned}$$

where $O(n^{i-2})$ is some polynomial(s) of degree $i-2$.

This guarantees the following function will terminate in finite steps[14]; `difLists` keeps generating difference lists until the difference get constant.

```
> difLists :: (Eq a, Num a) => [[a]] -> [[a]]
> difLists [] = []
> difLists xx@(xs:xss) =
>   if isConst xs then xx
>                 else difLists $ difs xs : xx
>   where
>     isConst (i:jj@(j:js)) = all (==i) jj
>     isConst _ = error "difLists: lack of data, or not a polynomial"
```

Let us try:

```
*Univariate> difLists [[-12,-11,6,45,112,213,354,541,780,1077]]
[[6,6,6,6,6,6,6]
,[16,22,28,34,40,46,52,58]
,[1,17,39,67,101,141,187,239,297]
,[-12,-11,6,45,112,213,354,541,780,1077]
]
```

[14] If a given lists is generated by a polynomial.

The degree of the polynomial can be computed by difference analysis:

```
> degree' :: (Eq a, Num a) => [a] -> Int
> degree' xs = length (difLists [xs]) -1
```

For example,

```
*Univariate> degree [1,4,9,16,25,36,49,64,81,100]
2
*Univariate> take 10 $ map (\n -> n^2+2*n+1) [0..]
[1,4,9,16,25,36,49,64,81,100]
*Univariate> degree $ take 10 $ map (\n -> n^5+4*n^3+1) [0..]
5
```

Above `degree'` function can only treat finite list, however, the following function can compute the degree of infinite list.

```
> degreeLazy :: (Eq a, Num a) => [a] -> Int
> degreeLazy xs = helper xs 0
>   where
>     helper as@(a:b:c:_) n
>       | a==b && b==c = n
>       | otherwise    = helper (difs as) (n+1)
```

Note that this lazy function only sees the first two elements of the list (of difference). So first take the lazy `degreeLazy` and guess the degree, take sufficient finite sublist of output and apply `degree'`. Here is the hybrid version:

```
> degree :: (Num a, Eq a) => [a] -> Int
> degree xs = let l = degreeLazy xs in
>   degree' $ take (l+2) xs
```

Chapter 2

Functional reconstruction over $\mathbb{Q}$

The goal of a functional reconstruction algorithm is to identify the monomials appearing in their definition and the corresponding coefficients.

From here, we use $\mathbb{Q}$ as our base field, but every algorithm can be computed on any field, e.g., finite field $\mathbb{Z}_p$.

2.1 Univariate polynomials

2.1.1 Newtons' polynomial representation

Consider a univariate polynomial $f(z)$. Given a sequence of distinct values $y_n|_{n\in\mathbb{N}}$, we evaluate the polynomial form $f(z)$ sequentially:

$$
\begin{aligned}
f_0(z) &= a_0 && (2.1)\\
f_1(z) &= a_0 + (z-y_0)a_1 && (2.2)\\
&\vdots\\
f_r(z) &= a_0 + (z-y_0)\left(a_1 + (z-y_1)(\cdots + (z-y_{r-1})a_r\right) && (2.3)\\
&= f_{r-1}(z) + (z-y_0)(z-y_1)\cdots(z-y_{r-1})a_r, && (2.4)
\end{aligned}
$$

where

$$
\begin{aligned}
a_0 &= f(y_0) && (2.5)\\
a_1 &= \frac{f(y_1)-a_0}{y_1-y_0} && (2.6)\\
&\vdots\\
a_r &= \left(\left((f(y_r)-a_0)\frac{1}{y_r-y_0}-a_1\right)\frac{1}{y_r-y_1}-\cdots-a_{r-1}\right)\frac{1}{y_r-y_{r-1}} && (2.7)
\end{aligned}
$$

It is easy to see that, $f_r(z)$ and the original $f(z)$ match on the given data points, i.e.,

$$f_r(n) = f(n), 0 \le n \le r. \tag{2.8}$$

When we have already known the total degree of $f(z)$, say R, then we can terminate this sequential trial:

$$
\begin{aligned}
f(z) &= f_R(z) && (2.9)\\
&= \sum_{r=0}^{R} a_r \prod_{i=0}^{r-1}(z-y_i). && (2.10)
\end{aligned}
$$

In practice, a consecutive zero on the sequence a_r can be taken as the termination condition for this algorithm.[1]

2.1.2 Towards canonical representations

Once we get the Newton's representation

$$\sum_{r=0}^{R} a_r \prod_{i=0}^{r-1}(z-y_i) = a_0 + (z-y_0)\left(a_1 + (z-y_1)(\cdots + (z-y_{R-1})a_R\right) \tag{2.11}$$

as the reconstructed polynomial, it is convenient to convert it into the canonical form:

$$\sum_{r=0}^{R} c_r z^r. \tag{2.12}$$

This conversion only requires addition and multiplication of univariate polynomials. These operations are reasonably cheap, especially on $\mathbb{Z}_p$.

[1] We have not proved, but higher power will be dominant when we take sufficiently big input, so we terminate this sequence when we get a consecutive zero in a_r.

2.1.3 Simplification of our problem

Without loss of generality, we can put

$$\texttt{[0..]} \tag{2.13}$$

as our input list. We usually take its finite part but we assume it has enough length. Corresponding to above input,

$$\texttt{map f [0..] = [f 0, f 1, ..]} \tag{2.14}$$

of `f :: Ratio Int -> Ratio Int` is our output list.

Then we have slightly simpler forms of coefficients:

$$
\begin{aligned}
f_r(z) &:= a_0 + z * (a_1 + (z-1)\,(a_2 + (z-2)\,(a_3 + \cdots + (z-r+1)a_r))) && (2.15)\\
a_0 &= f(0) && (2.16)\\
a_1 &= f(y_1) - a_0 && (2.17)\\
&= f(1) - f(0) =: \Delta(f)(0) && (2.18)\\
a_2 &= \frac{f(2)-a_0}{2} - a_1 && (2.19)\\
&= \frac{f(2)-f(0)}{2} - (f(1)-f(0)) && (2.20)\\
&= \frac{f(2)-2f(1)-f(0)}{2} && (2.21)\\
&= \frac{(f(2)-f(1))-(f(1)-f(0))}{2} =: \frac{\Delta^2(f)(0)}{2} && (2.22)\\
&\vdots\\
a_r &= \frac{\Delta^r(f)(0)}{r!}, && (2.23)
\end{aligned}
$$

where Δ is the difference operator in eq.(1.125):

$$\Delta(f)(n) := f(n+1) - f(n). \tag{2.24}$$

In order to simplify our expression, we introduce a falling power:

$$
\begin{aligned}
(x)_0 &:= 1 && (2.25)\\
(x)_n &:= x(x-1)\cdots(x-n+1) && (2.26)\\
&= \prod_{i=0}^{n-1}(x-i). && (2.27)
\end{aligned}
$$

Under these settings, we have

$$\begin{aligned} f(z) &= f_R(z) && (2.28)\\ &= \sum_{r=0}^{R} \frac{\Delta^r(f)(0)}{r!}(x)_r, && (2.29)\end{aligned}$$

where we have assume

$$\Delta^{R+1}(f) = [0, 0, \cdots]. \tag{2.30}$$

Example

Consider a polynomial

$$f(z) := 2 * z^3 + 3 * z, \tag{2.31}$$

and its out put list

$$[f(0), f(1), f(3), \cdots] = [0, 5, 22, 63, 140, 265, \cdots] \tag{2.32}$$

This polynomial is 3rd degree, so we compute up to $\Delta^3(f)(0)$:

$$\begin{aligned} f(0) &= 0 && (2.33)\\ \Delta(f)(0) &= f(1) - f(0) = 5 && (2.34)\\ \Delta^2(f)(0) &= \Delta(f)(1) - \Delta(f)(0) \\ &= f(2) - f(1) - 5 = 22 - 5 - 5 = 12 && (2.35)\\ \Delta^3(f)(0) &= \Delta^2(f)(1) - \Delta^2(f)(0) \\ &= f(3) - f(2) - \{f(2) - f(1)\} - 12 = 12 && (2.36)\end{aligned}$$

so we get

$$[0, 5, 12, 12] \tag{2.37}$$

as the first difference list. Therefore, we get the falling power representation of f:

$$\begin{aligned} f(z) &= 5(x)_1 + \frac{12}{2}(x)_2 + \frac{12}{3!}(x)_3 && (2.38)\\ &= 5(x)_1 + 6(x)_2 + 2(x)_3. && (2.39)\end{aligned}$$

2.2 Univariate polynomial reconstruction with Haskell

2.2.1 Newton interpolation formula with Haskell

First, the falling power is naturally given by recursively:

```
> infixr 8 ^- -- falling power
> (^-) :: (Integral a) => a -> a -> a
> x ^- 0 = 1
> x ^- n = (x ^- (n-1)) * (x - n + 1)
```

Assume the differences are given in a list

$$\texttt{xs} = \left[f(0), \Delta(f)(0), \Delta^2(f)(0), \cdots\right]. \tag{2.40}$$

Then the implementation of the Newton interpolation formula is as follows:

```
> newtonC :: (Fractional t, Enum t) => [t] -> [t]
> newtonC xs = [x / factorial k | (x,k) <- zip xs [0..]]
>   where
>     factorial k = product [1..fromInteger k]
```

Consider a polynomial

$$\texttt{f x = 2*x^3+3*x} \tag{2.41}$$

Let us try to reconstruct this polynomial from output list. In order to get the list `[x_0, x_1 ..]`, take `difLists` and pick the first elements:

```
> let f x = 2*x^3+3*x
> take 10 $ map f [0..]
[0,5,22,63,140,265,450,707,1048,1485]
> difLists [it]
[[12,12,12,12,12,12,12]
,[12,24,36,48,60,72,84,96]
,[5,17,41,77,125,185,257,341,437]
,[0,5,22,63,140,265,450,707,1048,1485]
]
> reverse $ map head it
[0,5,12,12]
```

This list is the same as eq.(2.37) and we get the same expression as eq.(2.39) $5(x)_1 + 6(x)_2 + 2(x)_3$:

```
> newtonC it
[0 % 1,5 % 1,6 % 1,2 % 1]
```

The list of first differences, i.e.,

$$\left[f(0), \Delta(f)(0), \Delta^2(f)(0), \cdots\right] \tag{2.42}$$

can be computed as follows:

```
> firstDifs :: (Eq a, Num a) => [a] -> [a]
> firstDifs xs = reverse $ map head $ difLists [xs]
```

Mapping a list of integers to a Newton representation:

```
> list2npol :: (Integral a) => [Ratio a] -> [Ratio a]
> list2npol = newtonC . firstDifs

  *NewtonInterpolation> take 10 $ map f [0..]
  [0,5,22,63,140,265,450,707,1048,1485]
  *NewtonInterpolation> list2npol it
  [0 % 1,5 % 1,6 % 1,2 % 1]
```

Therefore, we get the Newton coefficients from the output list.

2.2.2 Stirling numbers of the first kind

We need to map Newton falling powers to standard powers to get the canonical representation. This is a matter of applying combinatorics, by means of a convention formula that uses the so-called Stirling cyclic numbers

$$\begin{bmatrix} n \\ k \end{bmatrix} \tag{2.43}$$

Its defining relation is, $\forall n > 0$,

$$(x)_n = \sum_{k=1}^{n} (-)^{n-k} \begin{bmatrix} n \\ k \end{bmatrix} x^k, \tag{2.44}$$

and

$$\begin{bmatrix} 0 \\ 0 \end{bmatrix} := 1. \tag{2.45}$$

From the highest order, x^n, we get

$$\begin{bmatrix} n \\ n \end{bmatrix} = 1, \forall n > 0. \tag{2.46}$$

We also put

$$\begin{bmatrix} 0 \\ 1 \end{bmatrix} = \begin{bmatrix} 0 \\ 2 \end{bmatrix} = \begin{bmatrix} 0 \\ 3 \end{bmatrix} = \cdots = 0, \tag{2.47}$$

and

$$\begin{bmatrix} 1 \\ 0 \end{bmatrix} = \begin{bmatrix} 2 \\ 0 \end{bmatrix} = \begin{bmatrix} 3 \\ 0 \end{bmatrix} = \cdots = 0. \tag{2.48}$$

The key equation is

$$(x)_n = (x)_{n-1} * (x - n + 1) \tag{2.49}$$

and we get

$$\begin{aligned}
(x)_n &= \sum_{k=1}^{n} (-)^{n-k} \begin{bmatrix} n \\ k \end{bmatrix} x^k && (2.50)\\
&= x^n + \sum_{k=1}^{n-1} (-)^{n-k} \begin{bmatrix} n \\ k \end{bmatrix} x^k && (2.51)\\
(x)_{n-1} * (x - n + 1) &= \sum_{k=1}^{n-1} (-)^{n-1-k} \left\{ \begin{bmatrix} n-1 \\ k \end{bmatrix} x^{k+1} - (n-1) \begin{bmatrix} n-1 \\ k \end{bmatrix} x^k \right\} && (2.52)\\
&= \sum_{l=2}^{n} (-)^{n-l} \begin{bmatrix} n-1 \\ l-1 \end{bmatrix} x^l + (n-1) \sum_{k=1}^{n-1} (-)^{n-k} \begin{bmatrix} n \\ k \end{bmatrix} x^k && (2.53)\\
&= x^n + (n-1)(-)^{n-1} x \\
&\quad + \sum_{k=2}^{n-1} (-)^{n-k} \left\{ \begin{bmatrix} n-1 \\ k-1 \end{bmatrix} + (n-1) \begin{bmatrix} n-1 \\ k \end{bmatrix} \right\} x^k && (2.54)\\
&= x^n + \sum_{k=1}^{n-1} (-)^{n-k} \left\{ \begin{bmatrix} n-1 \\ k-1 \end{bmatrix} + (n-1) \begin{bmatrix} n-1 \\ k \end{bmatrix} \right\} x^k && (2.55)
\end{aligned}$$

Therefore, $\forall n, k > 0$,

$$\begin{bmatrix} n \\ k \end{bmatrix} = \begin{bmatrix} n-1 \\ k-1 \end{bmatrix} + (n-1) \begin{bmatrix} n-1 \\ k \end{bmatrix} \tag{2.56}$$

Now we have the following canonical, power representation of reconstructed polynomial

$$f(z) = f_R(z) \tag{2.57}$$

$$= \sum_{r=0}^{R} \frac{\Delta^r(f)(0)}{r!}(x)_r \tag{2.58}$$

$$= \sum_{r=0}^{R} \frac{\Delta^r(f)(0)}{r!} \sum_{k=1}^{r} (-)^{r-k} \left[\begin{array}{c} r \\ k \end{array} \right] x^k, \tag{2.59}$$

So, what shall we do is to sum up order by order.

Here is an implementation, first the Stirling numbers:

```
> stirlingC :: Integer -> Integer -> Integer
> stirlingC 0 0 = 1
> stirlingC 0 _ = 0
> stirlingC n k = (n-1)*(stirlingC (n-1) k) + stirlingC (n-1) (k-1)
```

This definition can be used to convert from falling powers to standard powers.

```
> fall2pol :: (Integral a) => a -> [a]
> fall2pol 0 = [1]
> fall2pol n = 0   -- No constant term.
>            : [(-1)^(n-k) * stirlingC n k| k<-[1..n]]
```

We use `fall2pol` to convert Newton representations to standard polynomials in coefficients list representation. Here we have uses `sum` to collect same order terms in list representation.

```
> npol2pol :: (Integral a) => [Ratio a] -> [Ratio a]
> npol2pol xs = sum [ [x] * map fromInteger (fall2pol k)
>                   | (x,k) <- zip xs [0..]
>                   ]
```

2.2.3 `list2pol`: from output list to canonical coefficients

Finally, here is the function for computing a polynomial from an output sequence:

```
> list2pol :: (Integral a) => [Ratio a] -> [Ratio a]
> list2pol = npol2pol . list2npol
```

Here are some checks on these functions:

```
Reconstruction as curve fitting
  *NewtonInterpolation> list2pol $ map (\n -> 7*n^2+3*n-4) [0..100]
  [(-4) % 1,3 % 1,7 % 1]

  *NewtonInterpolation> list2pol [0,1,5,14,30]
  [0 % 1,1 % 6,1 % 2,1 % 3]
  *NewtonInterpolation> map (\n -> n%6 + n^2%2 + n^3%3) [0..4]
  [0 % 1,1 % 1,5 % 1,14 % 1,30 % 1]

  *NewtonInterpolation> map (p2fct $ list2pol [0,1,5,14,30]) [0..8]
  [0 % 1,1 % 1,5 % 1,14 % 1,30 % 1,55 % 1,91 % 1,140 % 1,204 % 1]
```

First example shows that from the sufficiently long output list, we can reconstruct the list of coefficients. Second example shows that from a given outputs, we have a list coefficients. Then use these coefficients, we define the output list of the function, and they match. The last example shows that from a limited (but sufficient) output information, we reconstruct a function and get extra outputs outside from the given data.

2.3 Univariate rational functions

We use the same notion, i.e., what we can know is the output-list of a univariate rational function, say `f::Int -> Ratio Int`:

```
map f [0..]  == [f 0, f 1 ..]                (2.60)
```

2.3.1 Thiele's interpolation formula

We evaluate the polynomial form $f(z)$ as a continued fraction:

$$
\begin{aligned}
f_0(z) &= a_0 && (2.61)\\
f_1(z) &= a_0 + \frac{z}{a_1} && (2.62)\\
&\vdots\\
f_r(z) &= a_0 + \cfrac{z}{a_1 + \cfrac{z-1}{a_2 + \cfrac{z-2}{\ddots\; a_{r-2} + \cfrac{\vdots}{a_{r-1} + \cfrac{z-r+1}{a_r}}}}}, && (2.63)
\end{aligned}
$$

where

$$
\begin{aligned}
a_0 &= f(0) && (2.64)\\
a_1 &= \frac{1}{f(1) - a_0} && (2.65)\\
a_2 &= \cfrac{1}{\cfrac{2}{f(2) - a_0} - a_1} && (2.66)\\
&\vdots\\
a_r &= \cfrac{1}{\cfrac{2}{\cfrac{3}{\cfrac{\vdots}{\cfrac{r}{f(r) - a_0} - a_1} - a_2} - a_{r-2}} - a_{r-1}} && (2.67)\\
&= \left(\left((f(r) - a_0)^{-1} r - a_1\right)^{-1} (r-1) - \cdots - a_{r-1}\right)^{-1} 1 && (2.68)
\end{aligned}
$$

2.3.2 Towards canonical representations

In order to get a unique representation of canonical form

$$\frac{\sum_\alpha n_\alpha z^\alpha}{\sum_\beta d_\beta z^\beta} \tag{2.69}$$

we put

$$d_{\min r'} = 1 \tag{2.70}$$

as a normalization, instead of d_0. However, if we meet 0 as a singular value, then we can shift s.t. the new $d_0 \neq 0$. So without loss of generality, we can assume $f(0)$ is not singular, i.e., the denominator of f has a nonzero constant term:

$$d_0 = 1 \tag{2.71}$$

$$f(z) = \frac{\sum_i n_i z^i}{1 + \sum_{j>0} d_z^j}. \tag{2.72}$$

2.4 Univariate rational function reconstruction with Haskell

Here we the same notion of

`https://rosettacode.org/wiki/Thiele%27s_interpolation_formula`

and especially

`https://rosettacode.org/wiki/Thiele%27s_interpolation_formula#C`

2.4.1 Reciprocal difference

We claim, without proof[2], that the Thiele coefficients are given by

$$a_0 := f(0) \tag{2.73}$$

$$a_n := \rho_{n,0} - \rho_{n-2,0}, \tag{2.74}$$

[2] See the ref.4, Theorem (2.2.2.5) in 2nd edition.

where ρ is so called the reciprocal difference:

$$\rho_{n,i} := 0, n < 0 \tag{2.75}$$
$$\rho_{0,i} := f(i), i = 0, 1, 2, \cdots \tag{2.76}$$
$$\rho_{n,i} := \frac{n}{\rho_{n-1,i+1} - \rho_{n-1,i}} + \rho_{n-2,i+1} \tag{2.77}$$

These preparation helps us to write the following codes:

```
Thiele's interpolation formula

Reciprocal difference rho, using the same notation of
https://rosettacode.org/wiki/Thiele%27s_interpolation_formula#C

> rho :: [Ratio Int] -- A list of output of f :: Int -> Ratio Int
>     -> Int -> Int -> Ratio Int
> rho fs 0 i = fs !! i
> rho fs n _
>   | n < 0 = 0
> rho fs n i = (n*den)%num + rho fs (n-2) (i+1)
>   where
>     num  = numerator next
>     den  = denominator next
>     next = (rho fs (n-1) (i+1)) - (rho fs (n-1) i)

Note that (%) has the following type,
  (%) :: Integral a => a -> a -> Ratio a

> a fs 0 = fs !! 0
> a fs n = rho fs n 0 - rho fs (n-2) 0
```

2.4.2 `tDegree` for termination

Now let us consider a simple example which is given by the following Thiele coefficients

$$a_0 = 1, a_1 = 2, a_2 = 3, a_3 = 4. \tag{2.78}$$

The function is now

$$
\begin{aligned}
f(x) &:= 1 + \cfrac{x}{2 + \cfrac{x-1}{3 + \cfrac{x-2}{4}}} && (2.79)\\
&= \frac{x^2 + 16x + 16}{16 + 6x} && (2.80)
\end{aligned}
$$

Using Maxima[3], we can verify this:

```
(%i25) f(x) := 1+(x/(2+(x-1)/(3+(x-2)/4)));
(%o25) f(x):=x/(2+(x-1)/(3+(x-2)/4))+1
(%i26) ratsimp(f(x));
(%o26) (x^2+16*x+16)/(16+6*x)
```

Let us come back Haskell, and try to get the Thiele coefficients of

```
*Univariate> let func x = (x^2 + 16*x + 16)%(6*x + 16)
*Univariate> let fs = map func [0..]
*Univariate> map (a fs) [0..]
[1 % 1,2 % 1,3 % 1,4 % 1,*** Exception: Ratio has zero denominator
```

This is clearly unsafe, so let us think more carefully. Observe the reciprocal differences

```
*Univariate> let fs = map func [0..]
*Univariate> take 5 $ map (rho fs 0) [0..]
[1 % 1,3 % 2,13 % 7,73 % 34,12 % 5]
*Univariate> take 5 $ map (rho fs 1) [0..]
[2 % 1,14 % 5,238 % 69,170 % 43,230 % 53]
*Univariate> take 5 $ map (rho fs 2) [0..]
[4 % 1,79 % 16,269 % 44,667 % 88,413 % 44]
*Univariate> take 5 $ map (rho fs 3) [0..]
[6 % 1,6 % 1,6 % 1,6 % 1,6 % 1]
```

So, the constancy of the reciprocal differences can be used to get the depth of Thiele series:

```
> tDegree :: [Ratio Int] -> Int
> tDegree fs = helper fs 0
```

[3] http://maxima.sourceforge.net

```
>   where
>     helper fs n
>       | isConstants fs' = n
>       | otherwise       = helper fs (n+1)
>       where
>         fs' = map (rho fs n) [0..]
>     isConstants (i:j:_) = i==j -- 2 times match
> --  isConstants (i:j:k:_) = i==j && j==k
```

Using this `tDegree` function, we can safely take the (finite) Thiele sequence.

2.4.3 `thieleC`: from output list to Thiele coefficients

From the equation (3.26) of ref.1,

```
*Univariate> let h t = (3+6*t+18*t^2)%(1+2*t+20*t^2)
*Univariate> let hs = map h [0..]
*Univariate> tDegree hs
4
```

So we get the Thiele coefficients

```
*Univariate> map (a hs) [0..(tDegree hs)]
[3 % 1,(-23) % 42,(-28) % 13,767 % 14,7 % 130]
```

Plug these in the continued fraction, and simplify with Maxima

```
(%i35) h(t):=3+t/((-23/42)+(t-1)/((-28/13)+(t-2)/((767/14)+(t-3)/(7/130))));
(%o35) h(t):=t/((-23)/42+(t-1)/((-28)/13+(t-2)/(767/14+(t-3)/(7/130))))+3
(%i36) ratsimp(h(t));
(%o36) (18*t^2+6*t+3)/(1+2*t+20*t^2)
```

Finally we make a function `thieleC` that returns the Thiele coefficients:

```
> thieleC :: [Ratio Int] -> [Ratio Int]
> thieleC lst = map (a lst) [0..(tDegree lst)]
```

```
*Univariate> thieleC hs
[3 % 1,(-23) % 42,(-28) % 13,767 % 14,7 % 130]
```

We need a convertor from this Thiele sequence to continuous form of rational function.

```
> nextStep [a0,a1] (v:_)  = a0 + v/a1
> nextStep (a:as)  (v:vs) = a + (v / nextStep as vs)
>
> -- From thiele sequence to (rational) function.
> thiele2ratf :: Integral a => [Ratio a] -> (Ratio a -> Ratio a)
> thiele2ratf as x
>   | x == 0 = head as
>   | otherwise = nextStep as [x,x-1 ..]
```

The following example shows that, the given output lists `hs`, we can interpolate the value between our discrete data.

```
*Univariate> let h t = (3+6*t+18*t^2)%(1+2*t+20*t^2)
*Univariate> let hs = map h [0..]
*Univariate> take 5 hs
[3 % 1,27 % 23,87 % 85,183 % 187,45 % 47]
*Univariate> let as = thieleC hs
*Univariate> as
[3 % 1,(-23) % 42,(-28) % 13,767 % 14,7 % 130]
*Univariate> let th x = thiele2ratf as x
*Univariate> map th [0..5]
[3 % 1,27 % 23,87 % 85,183 % 187,45 % 47,69 % 73]
*Univariate> th 0.5
3 % 2
```

2.4.4 Haskell representation for rational functions

We represent a rational function by a tuple of coefficient lists, like,

$$\texttt{(ns,ds) :: ([Ratio Int],[Ratio Int])} \tag{2.81}$$

Here is a translator from coefficients lists to rational function.

```
> lists2ratf :: (Integral a) =>
>   ([Ratio a],[Ratio a]) -> (Ratio a -> Ratio a)
> lists2ratf (ns,ds) x = (p2fct ns x)/(p2fct ds x)
```

```
*Univariate> let frac x = lists2ratf ([1,1%2,1%3],[2,2%3]) x
*Univariate> take 10 $ map frac [0..]
[1 % 2,11 % 16,1 % 1,11 % 8,25 % 14,71 % 32,8 % 3,25 % 8,79 % 22,65 % 16]
*Univariate> let ffrac x = (1+(1%2)*x+(1%3)*x^2)/(2+(2%3)*x)
*Univariate> take 10 $ map ffrac [0..]
[1 % 2,11 % 16,1 % 1,11 % 8,25 % 14,71 % 32,8 % 3,25 % 8,79 % 22,65 % 16]
```

Simply taking numerator and denominator polynomials.

The following `canonicalizer` reduces the tuple-rep of rational function in canonical form, i.e., the coefficient of the lowest degree term of the denominator to be 1[4].

```
> canonicalize :: (Integral a) =>
>   ([Ratio a],[Ratio a]) -> ([Ratio a],[Ratio a])
> canonicalize rat@(ns,ds)
>   | dMin == 1 = rat
>   | otherwise = (map (/dMin) ns, map (/dMin) ds)
>   where
>     dMin = firstNonzero ds
>     firstNonzero [a] = a -- head
>     firstNonzero (a:as)
>       | a /= 0 = a
>       | otherwise = firstNonzero as
```

```
*Univariate> canonicalize ([1,1%2,1%3],[2,2%3])
([1 % 2,1 % 4,1 % 6],[1 % 1,1 % 3])
*Univariate> canonicalize ([1,1%2,1%3],[0,0,2,2%3])
([1 % 2,1 % 4,1 % 6],[0 % 1,0 % 1,1 % 1,1 % 3])
*Univariate> canonicalize ([1,1%2,1%3],[0,0,0,2%3])
([3 % 2,3 % 4,1 % 2],[0 % 1,0 % 1,0 % 1,1 % 1])
```

What we need is a translator from Thiele coefficients to this tuple-rep. Since the list of Thiele coefficients is finite, we can naturally think recursively.

Before we go to a general case, consider

$$f(x) := 1 + \cfrac{x}{2 + \cfrac{x-1}{3 + \cfrac{x-2}{4}}} \tag{2.82}$$

[4] Here our data point start from 0, i.e., the output data is given by `map f [0..]`, 0 is not singular, i.e., the denominator should have constant term and that means non empty. Therefore, the function firstNonzero is actually `head`.

When we simplify this expression, we should start from the bottom:

$$
\begin{aligned}
f(x) &= 1+\cfrac{x}{2+\cfrac{x-1}{\cfrac{4*3+x-2}{4}}} && (2.83)\\
&= 1+\cfrac{x}{2+\cfrac{x-1}{\cfrac{x+10}{4}}} && (2.84)\\
&= 1+\cfrac{x}{\cfrac{2*(x+10)+4*(x-1)}{x+10}} && (2.85)\\
&= 1+\cfrac{x}{\cfrac{6x+16}{x+10}} && (2.86)\\
&= \frac{1*(6x+16)+x*(x+10)}{6x+16} && (2.87)\\
&= \frac{x^2+16x+16}{6x+16} && (2.88)
\end{aligned}
$$

Finally, if we need, we take its canonical form:

$$
f(x) = \frac{1+x+\frac{1}{16}x^2}{1+\frac{3}{8}x} \tag{2.89}
$$

In general, we have the following Thiele representation:

$$
a_0+\cfrac{z}{a_1+\cfrac{z-1}{a_2+\cfrac{z-2}{\cfrac{\vdots}{a_n+\cfrac{z-n}{a_{n+1}}}}}} \tag{2.90}
$$

The base case should be

$$
a_n+\frac{z-n}{a_{n+1}} = \frac{a_{n+1}*a_n-n+z}{a_{n+1}} \tag{2.91}
$$

and induction step $0 \leq r \leq n$ should be

$$\begin{aligned} a_r(z) &= a_r + \frac{z-r}{a_{r+1}(z)} && (2.92) \\ &= \frac{a_r a_{r+1}(z) + z - r}{a_{r+1}(z)} && (2.93) \\ &= \frac{a_r * \mathtt{num}\,(a_{r+1}(z)) + \mathtt{den}\,(a_{r+1}(z)) * (z-r)}{\mathtt{num}\,(a_{r+1}(z))} && (2.94) \end{aligned}$$

where

$$a_{r+1}(z) = \frac{\mathtt{num}\,(a_{r+1}(z))}{\mathtt{den}\,(a_{r+1}(z))} \tag{2.95}$$

is a canonical representation of $a_{n+1}(z)$[5].

Thus, the implementation is the followings.

```
> thiele2coef :: (Integral a) =>
>   [Ratio a] -> ([Ratio a],[Ratio a])
> thiele2coef as = canonicalize $ t2r as 0
>   where
>     t2r [an,an'] n = ([an*an'-n,1],[an'])
>     t2r (a:as)   n = ((a .* num) + ([-n,1] * den), num)
>       where
>         (num, den) = t2r as (n+1)
```

From the first example,

```
*Univariate> let func x = (x^2+16*x+16)%(6*x+16)
*Univariate> let funcList = map func [0..]
*Univariate> tDegree funcList
3
*Univariate> take 5 funcList
[1 % 1,3 % 2,13 % 7,73 % 34,12 % 5]
*Univariate> let aFunc = thieleC funcList
*Univariate> aFunc
[1 % 1,2 % 1,3 % 1,4 % 1]
*Univariate> thiele2coef aFunc
([1 % 1,1 % 1,1 % 16],[1 % 1,3 % 8])
```

From the other example, equation (3.26) of ref.1,

[5] Not necessary being a canonical representation, it suffices to express $a_{n+1}(z)$ in a polynomial over polynomial form, that is, two lists in Haskell.

```
*Univariate> let h t = (3+6*t+18*t^2)%(1+2*t+20*t^2)
*Univariate> let hs = map h [0..]
*Univariate> take 5 hs
[3 % 1,27 % 23,87 % 85,183 % 187,45 % 47]
*Univariate> let th x = thiele2ratf as x
*Univariate> map th [0..5]
[3 % 1,27 % 23,87 % 85,183 % 187,45 % 47,69 % 73]
*Univariate> as
[3 % 1,(-23) % 42,(-28) % 13,767 % 14,7 % 130]
*Univariate> thiele2coef as
([3 % 1,6 % 1,18 % 1],[1 % 1,2 % 1,20 % 1])
```

2.4.5 `list2rat`: from output list to canonical coefficients

Finally, we get

```
> list2rat :: (Integral a) => [Ratio a] -> ([Ratio a], [Ratio a])
> list2rat = thiele2Coef . thieleC
```

as the reconstruction function from the output sequence.

```
*Univariate> let h t = (3+6*t+18*t^2)%(1+2*t+20*t^2)
*Univariate> list2rat $ map h [0..]
([3 % 1,6 % 1,18 % 1],[1 % 1,2 % 1,20 % 1])
```

2.5 Multivariate polynomials

From now on, we will use only the following functions from univariate cases.

Multivariate.lhs

```
> module Multivariate
>   where

> import Data.Ratio
> import Univariate
>   ( degree, list2pol
>   , thiele2ratf, lists2ratf, thiele2coef, lists2rat
>   )
```

2.5.1 Foldings as recursive applications

Consider an arbitrary multivariate polynomial

$$f(z_1, \cdots, z_n) \in \mathbb{K}[z_1, \cdots, z_n]. \tag{2.96}$$

First, fix all the variable but 1st and apply the univariate Newton's reconstruction:

$$f(z_1, z_2, \cdots, z_n) = \sum_{r=0}^{R} a_r(z_2, \cdots, z_n) \prod_{i=0}^{r-1}(z_1 - y_i) \tag{2.97}$$

Recursively, pick up one "coefficient" and apply the univariate Newton's reconstruction on z_2:

$$a_r(z_2, \cdots, z_n) = \sum_{s=0}^{S} b_s(z_3, \cdots, z_n) \prod_{j=0}^{s-1}(z_2 - x_j) \tag{2.98}$$

The terminate cotndition should be the univariate case.

2.5.2 Experiments, 2 variables case

Let us take a polynomial from the denominator in eq.(3.23) of ref.1.

$$f(z_1, z_2) = 3 + 2z_1 + 4z_2 + 7z_1^2 + 5z_1z_2 + 6z_2^2 \tag{2.99}$$

In Haskell, first, fix $z_2 = 0, 1, 2$ and identify $f(z_1, 0), f(z_1, 1), f(z_1, 2)$ as our univariate polynomials.

```
*Multivariate> let f z1 z2 = 3+2*z1+4*z2+7*z1^2+5*z1*z2+6*z2^2
*Multivariate> let fs z = map (‘f‘ z) [0..]
*Multivariate> let llst = map fs [0,1,2]
*Multivariate> map degree llst
[2,2,2]
```

Fine, so the canonical form can be

$$f(z_1, z) = c_0(z) + c_1(z)z_1 + c_2(z)z_1^2. \tag{2.100}$$

Now our new target is three univariate polynomials $c_0(z), c_1(z), c_2(z)$.

```
*Multivariate> list2pol $ take 10 $ fs 0
[3 % 1,2 % 1,7 % 1]
*Multivariate> list2pol $ take 10 $ fs 1
[13 % 1,7 % 1,7 % 1]
*Multivariate> list2pol $ take 10 $ fs 2
[35 % 1,12 % 1,7 % 1]
```

That is

$$\begin{aligned} f(z,0) &= 3+2z+7z^2 && (2.101)\\ f(z,1) &= 13+7z+7z^2 && (2.102)\\ f(z,2) &= 35+12z+7z^2. && (2.103)\end{aligned}$$

From these observation, we can determine $c_2(z)$, since it already a constant sequence.

$$c_2(z) = 7 \qquad (2.104)$$

Consider $c_1(z)$, the sequence is now enough to determine $c_1(z)$:

```
*Multivariate> degree [2,7,12]
1
*Multivariate> list2pol [2,7,12]
[2 % 1,5 % 1]
```

i.e.,

$$c_1(z) = 2+5z. \qquad (2.105)$$

However, for $c_1(z)$

```
*Multivariate> degree [3, 13, 35]
*** Exception: difLists: lack of data, or not a polynomial
CallStack (from HasCallStack):
  error, called at ./Univariate.lhs:61:19 in main:Univariate
```

so we need more numbers. Let us try one more:

```
*Multivariate> list2pol $ take 10 $ map (‘f‘ 3) [0..]
[69 % 1,17 % 1,7 % 1]
*Multivariate> degree [3, 13, 35, 69]
2
*Multivariate> list2pol [3,13,35,69]
[3 % 1,4 % 1,6 % 1]
```

Thus we have

$$c_0(z) = 3+4z+6z^2 \qquad (2.106)$$

and these fully determine our polynomial:

$$f(z_1,z_2) = (3+4z_2+6z_2^2)+(2+5z_2)z_1+7z_1^2. \qquad (2.107)$$

As another experiment, take the denominator.

```
*Multivariate> let g x y = 1+7*x + 8*y + 10*x^2 + x*y+9*y^2
*Multivariate> let gs x = map (g x) [0..]
*Multivariate> map degree $ map gs [0..3]
[2,2,2,2]
```

So the canonical form should be

$$g(x, y) = c_0(x) + c_1(x)y + c_2(x)y^2 \tag{2.108}$$

Let us look at these coefficient polynomial:

```
*Multivariate> list2pol $ take 10 $ gs 0
[1 % 1,8 % 1,9 % 1]
*Multivariate> list2pol $ take 10 $ gs 1
[18 % 1,9 % 1,9 % 1]
*Multivariate> list2pol $ take 10 $ gs 2
[55 % 1,10 % 1,9 % 1]
*Multivariate> list2pol $ take 10 $ gs 3
[112 % 1,11 % 1,9 % 1]
```

So we get

$$c_2(x) = 9 \tag{2.109}$$

and

```
*Multivariate> map (list2pol . (take 10) . gs) [0..4]
[[1 % 1,8 % 1,9 % 1]
,[18 % 1,9 % 1,9 % 1]
,[55 % 1,10 % 1,9 % 1]
,[112 % 1,11 % 1,9 % 1]
,[189 % 1,12 % 1,9 % 1]
]
*Multivariate> map head it
[1 % 1,18 % 1,55 % 1,112 % 1,189 % 1]
*Multivariate> list2pol it
[1 % 1,7 % 1,10 % 1]
*Multivariate> list2pol $ map (head . list2pol . (take 10) . gs) [0..4]
[1 % 1,7 % 1,10 % 1]
```

Using index operator `(!!)`,

```
*Multivariate> list2pol $ map ((!! 0) . list2pol . (take 10) . gs) [0..4]
[1 % 1,7 % 1,10 % 1]
*Multivariate> list2pol $ map ((!! 1) . list2pol . (take 10) . gs) [0..4]
[8 % 1,1 % 1]
*Multivariate> list2pol $ map ((!! 2) . list2pol . (take 10) . gs) [0..4]
[9 % 1]
```

Finally we get

$$c_0(x) = 1 + 7x + 10x^2, c_1(x) = 8 + x, (c_2(x) = 9,) \tag{2.110}$$

and

$$g(x, y) = (1 + 7x + 10x^2) + (8 + x)y + 9y^2 \tag{2.111}$$

2.5.3 Haskell implementation, 2 variables case

Let us assume that we are given a "table" of the values of a 2-variate function. We represent this table as a list of lists.

```
*Multivariate> let f z1 z2 = 3+2*z1+4*z2+7*z1^2+5*z1*z2+6*z2^2
*Multivariate> [[f x y | y <- [0..9]] | x <- [0..9]]
[[3,13,35,69,115,173,243,325,419,525]
,[12,27,54,93,144,207,282,369,468,579]
,[35,55,87,131,187,255,335,427,531,647]
,[72,97,134,183,244,317,402,499,608,729]
,[123,153,195,249,315,393,483,585,699,825]
,[188,223,270,329,400,483,578,685,804,935]
,[267,307,359,423,499,587,687,799,923,1059]
,[360,405,462,531,612,705,810,927,1056,1197]
,[467,517,579,653,739,837,947,1069,1203,1349]
,[588,643,710,789,880,983,1098,1225,1364,1515]
]
```

```
> tablize :: (Enum t1, Num t1) => (t1 -> t1 -> t) -> Int -> [[t]]
> tablize f n = [[f x y | y <- range] | x <- range]
>   where
>     range = take n [0..]
```

So, this "table" is like

$$\begin{pmatrix} f_{0,0} & f_{0,1} & \cdots \\ f_{1,0} & f_{1,1} & \cdots \\ f_{2,0} & f_{2,1} & \cdots \\ \vdots & & \ddots \end{pmatrix} \tag{2.112}$$

Then we can apply the univariate technique.

```
*Multivariate> let fTable = tablize f 10
*Multivariate> map list2pol fTable
[[3 % 1,4 % 1,6 % 1]
,[12 % 1,9 % 1,6 % 1]
,[35 % 1,14 % 1,6 % 1]
,[72 % 1,19 % 1,6 % 1]
,[123 % 1,24 % 1,6 % 1]
,[188 % 1,29 % 1,6 % 1]
,[267 % 1,34 % 1,6 % 1]
,[360 % 1,39 % 1,6 % 1]
,[467 % 1,44 % 1,6 % 1]
,[588 % 1,49 % 1,6 % 1]
]
```

Now we need to see the behavior of each coefficient, so take the "transpose" of it:

```
> wellOrd :: [[a]] -> [[a]]
> wellOrd xss
>   | null (head xss) = []
>   | otherwise       = map head xss : wellOrd (map tail xss)
```

```
*Multivariate> let f z1 z2 = 3+2*z1+4*z2+7*z1^2+5*z1*z2+6*z2^2
*Multivariate> let fTable = tablize f 10
*Multivariate> map list2pol fTable
[[3 % 1,4 % 1,6 % 1]
,[12 % 1,9 % 1,6 % 1]
,[35 % 1,14 % 1,6 % 1]
,[72 % 1,19 % 1,6 % 1]
,[123 % 1,24 % 1,6 % 1]
,[188 % 1,29 % 1,6 % 1]
,[267 % 1,34 % 1,6 % 1]
```

```
,[360 % 1,39 % 1,6 % 1]
,[467 % 1,44 % 1,6 % 1]
,[588 % 1,49 % 1,6 % 1]
]
*Multivariate> wellOrd it
[[3 % 1,12 % 1,35 % 1,72 % 1,123 % 1,188 % 1,267 % 1,360 % 1,467 % 1,588 % 1]
,[4 % 1,9 % 1,14 % 1,19 % 1,24 % 1,29 % 1,34 % 1,39 % 1,44 % 1,49 % 1]
,[6 % 1,6 % 1,6 % 1,6 % 1,6 % 1,6 % 1,6 % 1,6 % 1,6 % 1,6 % 1]
]
*Multivariate> map list2pol it
[[3 % 1,2 % 1,7 % 1]
,[4 % 1,5 % 1]
,[6 % 1]]
```

Therefore, the whole procedure becomes

```
> table2pol :: [[Ratio Integer]] -> [[Ratio Integer]]
> table2pol = map list2pol . wellOrd . map list2pol

  *Multivariate> let g x y = 1+7*x + 8*y + 10*x^2 + x*y+9*y^2
  *Multivariate> table2pol $ tablize g 5
  [[1 % 1,7 % 1,10 % 1],[8 % 1,1 % 1],[9 % 1]]
```

2.6 Multivariate rational functions

2.6.1 The canonical normalization

Our target is a pair of coefficients $(\{n_\alpha\}_\alpha, \{d_\beta\}_\beta)$ in

$$\frac{\sum_\alpha n_\alpha z^\alpha}{\sum_\beta d_\beta z^\beta} \tag{2.113}$$

A canonical choice is

$$d_0 = d_{(0,\cdots,0)} = 1. \tag{2.114}$$

Accidentally we might face $d_0 = 0$, but we can shift our function and make

$$d'_0 = d_s \neq 0. \tag{2.115}$$

2.6.2 An auxiliary t

Introducing an auxiliary variable t, let us define

$$h(z,t) := f(tz_1, \cdots, tz_n), \tag{2.116}$$

and reconstruct $h(t,z)$ as a univariate rational function of t:

$$h(z,t) = \frac{\sum_{r=0}^{R} p_r(z) t^r}{1 + \sum_{r'=1}^{R'} q_{r'}(z) t^{r'}} \tag{2.117}$$

where

$$p_r(z) = \sum_{|\alpha|=r} n_\alpha z^\alpha \tag{2.118}$$

$$q_{r'}(z) = \sum_{|\beta|=r'} n_\beta z^\beta \tag{2.119}$$

are homogeneous polynomials.

Thus, what we shall do is the (homogeneous) polynomial reconstructions of $p_r(z)|_{0\le r\le R}$, $q_{r'}|_{1\le r'\le R'}$.

A simplification

Since our new targets are homogeneous polynomials, we can consider, say,

$$p_r(1, z_2, \cdots, z_n) \tag{2.120}$$

instead of $p_r(z_1, z_2, \cdots, z_n)$, reconstruct it using multivariate Newton's method, and homogenize with z_1.

2.6.3 Experiments, 2 variables case

Consider the equation (3.23) in ref.1.

```
*Multivariate> let f x y = (3+2*x+4*y+7*x^2+5*x*y+6*y^2)
                             % (1+7*x+8*y+10*x^2+x*y+9*y^2)
*Multivariate> :t f
f :: Integral a => a -> a -> Ratio a
*Multivariate> let h x y t = f (t*x) (t*y)
*Multivariate> let hs x y = map (h x y) [0..]
*Multivariate> take 5 $ hs 0 0
[3 % 1,3 % 1,3 % 1,3 % 1,3 % 1]
```

```
*Multivariate> take 5 $ hs 0 1
[3 % 1,13 % 18,35 % 53,69 % 106,115 % 177]
*Multivariate> take 5 $ hs 1 0
[3 % 1,2 % 3,7 % 11,9 % 14,41 % 63]
*Multivariate> take 5 $ hs 1 1
[3 % 1,3 % 4,29 % 37,183 % 226,105 % 127]
```

Here we have introduced the auxiliary t as third argument.

We take $(x, y) = (1, 0), (1, 1), (1, 2), (1, 3)$ and reconstruct them[6].

```
*Multivariate> lists2rat $ hs 1 0
([3 % 1,2 % 1,7 % 1],[1 % 1,7 % 1,10 % 1])
*Multivariate> lists2rat $ hs 1 1
([3 % 1,6 % 1,18 % 1],[1 % 1,15 % 1,20 % 1])
*Multivariate> lists2rat $ hs 1 2
([3 % 1,10 % 1,41 % 1],[1 % 1,23 % 1,48 % 1])
*Multivariate> lists2rat $ hs 1 3
([3 % 1,14 % 1,76 % 1],[1 % 1,31 % 1,94 % 1])
```

So we have

$$h(1,0,t) = \frac{3+2t+7t^2}{1+7t+10t^2} \quad (2.121)$$

$$h(1,1,t) = \frac{3+6t+18t^2}{1+15t+20t^2} \quad (2.122)$$

$$h(1,2,t) = \frac{3+10t+41t^2}{1+23t+48t^2} \quad (2.123)$$

$$h(1,3,t) = \frac{3+14t+76t^2}{1+31t+94t^2} \quad (2.124)$$

Our next targets are the coefficients as polynomials in y [7].

Let us consider numerator first. This `list` is Haskell representation for eq.(2.121), eq.(2.122), eq.(2.123) and eq.(2.124).

```
*Multivariate> let list = map (lists2rat . (hs 1)) [0..4]
*Multivariate> let numf = map fst list
*Multivariate> list
[([3 % 1,2 % 1,7 % 1],[1 % 1,7 % 1,10 % 1])
,([3 % 1,6 % 1,18 % 1],[1 % 1,15 % 1,20 % 1])
```

[6] Eq.(3.26) in ref.1 is different from our reconstruction.

[7] In our example, we take $x = 1$ fixed and reproduce x-dependence using homogenization

```
,([3 % 1,10 % 1,41 % 1],[1 % 1,23 % 1,48 % 1])
,([3 % 1,14 % 1,76 % 1],[1 % 1,31 % 1,94 % 1])
,([3 % 1,18 % 1,123 % 1],[1 % 1,39 % 1,158 % 1])
]
*Multivariate> numf
[[3 % 1,2 % 1,7 % 1]
,[3 % 1,6 % 1,18 % 1]
,[3 % 1,10 % 1,41 % 1]
,[3 % 1,14 % 1,76 % 1]
,[3 % 1,18 % 1,123 % 1]
]
```

From this information, we reconstruct each polynomials

```
*Multivariate> list2pol $ map head numf
[3 % 1]
*Multivariate> list2pol $ map (head . tail) numf
[2 % 1,4 % 1]
*Multivariate> list2pol $ map last numf
[7 % 1,5 % 1,6 % 1]
```

that is we have $3, 2 + 4y, 7 + 5y + 6y^2$ as results. Similarly,

```
*Multivariate> let denf = map snd list
*Multivariate> denf
[[1 % 1,7 % 1,10 % 1]
,[1 % 1,15 % 1,20 % 1]
,[1 % 1,23 % 1,48 % 1]
,[1 % 1,31 % 1,94 % 1]
,[1 % 1,39 % 1,158 % 1]
]
*Multivariate> list2pol $ map head denf
[1 % 1]
*Multivariate> list2pol $ map (head . tail) denf
[7 % 1,8 % 1]
*Multivariate> list2pol $ map last denf
[10 % 1,1 % 1,9 % 1]
```

So we get

$$h(1, y, t) = \frac{3 + (2 + 4y)t + (7 + 5y + 6y^2)t^2}{1 + (7 + 8y)t + (10 + y + 9y^2)t^2} \tag{2.125}$$

Finally, we use the homogeneous property for each powers:

$$h(x,y,t) = \frac{3+(2x+4y)t+(7x^2+5xy+6y^2)t^2}{1+(7x+8y)t+(10x^2+xy+9y^2)t^2} \tag{2.126}$$

Putting $t=1$, we get

$$\begin{aligned} f(x,y) &= h(x,y,1) && (2.127)\\ &= \frac{3+(2x+4y)+(7x^2+5xy+6y^2)}{1+(7x+8y)+(10x^2+xy+9y^2)} && (2.128)\end{aligned}$$

2.6.4 Haskell implementation, 2 variables case

Assume we have a "table" of data:

```
*Multivariate> let h x y = (3+2*x+4*y+7*x^2+5*x*y+6*y^2) % (1+7*x+8*y+10*x^2+x*y+9*y^2)
*Multivariate> let auxh x y t = h (t*x) (t*y)
*Multivariate> let h x y = (3+2*x+4*y+7*x^2+5*x*y+6*y^2)% (1+7*x+8*y+10*x^2+x*y+9*y^2)
*Multivariate> let auxh x y t = h (t*x) (t*y)
```

```
Using the homogenious property, we just take x=1:
```

```
*Multivariate> let auxhs = [map (auxh 1 y) [0..5] | y <- [0..5]]
*Multivariate> auxhs
[[3 % 1,2 % 3,7 % 11,9 % 14,41 % 63,94 % 143]
,[3 % 1,3 % 4,29 % 37,183 % 226,105 % 127,161 % 192]
,[3 % 1,3 % 4,187 % 239,201 % 251,233 % 287,77 % 94]
,[3 % 1,31 % 42,335 % 439,729 % 940,425 % 543,1973 % 2506]
,[3 % 1,8 % 11,59 % 79,291 % 385,681 % 895,528 % 691]
,[3 % 1,23 % 32,155 % 211,1707 % 2302,1001 % 1343,4663 % 6236]
]
```

Now, each list can be seen as a univariate rational function:

```
*Multivariate> map list2rat auxhs
[([3 % 1,2 % 1,7 % 1],[1 % 1,7 % 1,10 % 1])
,([3 % 1,6 % 1,18 % 1],[1 % 1,15 % 1,20 % 1])
,([3 % 1,10 % 1,41 % 1],[1 % 1,23 % 1,48 % 1])
,([3 % 1,14 % 1,76 % 1],[1 % 1,31 % 1,94 % 1])
,([3 % 1,18 % 1,123 % 1],[1 % 1,39 % 1,158 % 1])
,([3 % 1,22 % 1,182 % 1],[1 % 1,47 % 1,240 % 1])
]
```

```
*Multivariate> map fst it
[[3 % 1,2 % 1,7 % 1]
,[3 % 1,6 % 1,18 % 1]
,[3 % 1,10 % 1,41 % 1]
,[3 % 1,14 % 1,76 % 1]
,[3 % 1,18 % 1,123 % 1]
,[3 % 1,22 % 1,182 % 1]
]
```

We need to see the behavior of each coefficients:

```
*Multivariate> wellOrd it
[[3 % 1,3 % 1,3 % 1,3 % 1,3 % 1,3 % 1]
,[2 % 1,6 % 1,10 % 1,14 % 1,18 % 1,22 % 1]
,[7 % 1,18 % 1,41 % 1,76 % 1,123 % 1,182 % 1]
]
*Multivariate> map list2pol it
[[3 % 1],[2 % 1,4 % 1],[7 % 1,5 % 1,6 % 1]]
```

So, the numerator is given by

```
*Multivariate> map list2pol . wellOrd . map (fst . list2rat) $ auxhs
[[3 % 1],[2 % 1,4 % 1],[7 % 1,5 % 1,6 % 1]]
```

and the denominator is

```
*Multivariate> map list2pol . wellOrd . map (snd . list2rat) $ auxhs
[[1 % 1],[7 % 1,8 % 1],[10 % 1,1 % 1,9 % 1]]
```

Thus, we finally get the following function

```
> table2ratf table = (t2r fst table, t2r snd table)
>   where
>     t2r third = map list2pol . wellOrd . map (third . list2rat)
```

```
*Multivariate> table2ratf auxhs
([[3 % 1],[2 % 1,4 % 1],[7 % 1,5 % 1,6 % 1]],[[1 % 1],[7 % 1,8 % 1],[10 % 1,1 %
```

Chapter 3

Functional reconstruction over finite fields

3.1 Univariate polynomials

We choose our new target the first differences, since once we get it, to reconstruct polynomial is an easy task. Once we get the first differences of a polynomial, we get the coefficient list by applying `npol2pol . newtonC` on it.

3.1.1 Pre-cook

We need a convertor, or function-modular which takes function and prime, and returns a function on $\mathbb{Z}_p$.

```
> -- Function-modular.
> fmodp :: Integral c => (a -> Ratio c) -> c -> a -> c
> f `fmodp` p = (`modp` p) . f
```

```
  *FROverZp> let f x = (1%3) + (3%5)*x + (7%6)*x^2
  *FROverZp> let fs = map f [0..]
  *FROverZp> take 5 $ map (f `fmodp` 101) [0..]
  [34,93,87,16,82]
  *FROverZp> take 5 $ map (`modp` 101) fs
  [34,93,87,16,82]
```

What we can access is the output list of our target polynomial. On $\mathbb{Z}_p$, our input is a finite list

$$\texttt{[0,1,2 .. (p-1)]} \tag{3.1}$$

so as the output list.

```
> accessibleData :: (Ratio Int -> Ratio Int) -> Int -> [Int]
> accessibleData f p = take p $ map (f `fmodp` p) [0..]
>
> accessibleData' :: [Ratio Int] -> Int -> [Int]
> accessibleData' fs p = take p $ map (`modp` p) fs
```

3.1.2 Difference analysis on $\mathbb{Z}_p$

Play the same game over prime field $\mathbb{Z}_p$, i.e., every arithmetic in under mod p.

```
Difference analysis over Z_p
Every arithmetic should be on Z_p, i.e., (`mod` p).

> difsp :: Integral b => b -> [b] -> [b]
> difsp p xs = map (`mod` p) (zipWith (-) (tail xs) xs)

  *FROverZp> let f x = (1%3) + (3%5)*x + (7%6)*x^2
  *FROverZp> take 5 $ accessibleData f 101
  [34,93,87,16,82]
  *FROverZp> difsp 101 it
  [59,95,30,66]
  *FROverZp> difsp 101 it
  [36,36,36]
  *FROverZp> difsp 101 it
  [0,0]
```

Here what we do is, first to take the differences (`zipWith (-) (tail xs) xs`) and take modular p for all element (`map (`mod` p)`). Now we can recursively apply this `difsp` over our data.

```
> difListsp :: Integral b => b -> [[b]] -> [[b]]
> difListsp _ [] = []
> difListsp p xx@(xs:xxs) =
>   if isConst xs then xx
>                 else difListsp p $ difsp p xs : xx
```

```
>   where
>     isConst (i:jj@(j:js)) = all (==i) jj
>     isConst _ = error "difListsp: "
```

```
  *FROverZp> let f x = (1%3) + (3%5)*x + (7%6)*x^2
  *FROverZp> map head $ difListsp 101 [accessibleData f 101]
  [36,59,34]
```

3.1.3 Eager and lazy degree

From the above difference analysis on $\mathbb{Z}_p$, we get degree of the polynomial. Here we have a combination of two degree functions, one is eager and the other lazy:

```
Degree, eager and lazy versions

> degreep' p xs = length (difListsp p [xs]) -1
> degreep'Lazy p xs = helper xs 0
>   where
>     helper as@(a:b:c:_) n
>       | a==b && b==c = n -- two times matching
>       | otherwise    = helper (difsp p as) (n+1)
>
> degreep :: Integral b => b -> [b] -> Int
> degreep p xs = let l = degreep'Lazy p xs in
>   degreep' p $ take (l+2) xs
```

```
  *FROverZp> let f x = (1%3) + (3%5)*x + (7%6)*x^2
  *FROverZp> let myDeg p = degreep p $ accessibleData f p
  *FROverZp> myDeg 101
  2
  *FROverZp> myDeg 103
  2
  *FROverZp> myDeg 107
  2
  *FROverZp> degreep 101 $ accessibleData
    (\n -> (1%2)+(2%3)*n+(3%4)*n^2+(6%7)*n^7) 101
  7
```

Now we can take first differences.

```
> firstDifsp :: Integral a => a -> [a] -> [a]
> firstDifsp p xs = reverse $ map head $ difListsp p [xs']
>   where
>     xs' = take n xs
>     n   = 2+ degreep p xs
```

```
*FROverZp> let f x = (1%3) + (3%5)*x + (7%6)*x^2
*FROverZp> firstDifsp 101 $ accessibleData f 101
[34,59,36]
*FROverZp> firstDifsp 101 $ accessibleData
  (\n -> (1%2)+(2%3)*n+(3%4)*n^2+(6%7)*n^7) 101
[51,66,59,33,29,58,32,78]
```

3.1.4 Term by term reconstruction

The output list of `firstDifsp` are basically the coefficients of Newton representation on $\mathbb{Z}_p$. So we zip it with our base prime p and map these pair over several primes.

```
> wellOrd :: [[a]] -> [[a]]
> wellOrd xss
>   | null (head xss) = []
>   | otherwise       = map head xss : wellOrd (map tail xss)
```

```
*FROverZp> let f x = (1%3) + (3%5)*x + (7%6)*x^2
*FROverZp> let fps p = accessibleData f p
*FROverZp> let ourData p = firstDifsp p (fps p)
*FROverZp> let fivePrimes = take 5 bigPrimes
*FROverZp> map (\p -> zip (ourData p) (repeat p)) fivePrimes
[[(299158,897473),(867559,897473),(299160,897473)]
,[(299166,897497),(329084,897497),(299168,897497)]
,[(598333,897499),(388918,897499),(598335,897499)]
,[(598345,897517),(29919,897517),(598347,897517)]
,[(299176,897527),(329095,897527),(299178,897527)]
]
*FROverZp> wellOrd it
[[(299158,897473),(299166,897497),(598333,897499)
 ,(598345,897517),(299176,897527)]
,[(867559,897473),(329084,897497),(388918,897499)
 ,(29919,897517),(329095,897527)]
```

```
,[(299160,897473),(299168,897497),(598335,897499)
 ,(598347,897517),(299178,897527)]
]
*FROverZp> :t it
it :: [[(Int, Int)]]
```

Finally we get the images of first differences over prime fields.

One minor issue is to change the data type, since our tools (say the functions of Chinese Remainder Theorem) use limit-less integer `Integer`.

```
> toInteger2 :: (Integral a1, Integral a) => (a, a1) -> (Integer, Integer)
> toInteger2 (a,b) = (toInteger a, toInteger b)
```

Let us take an example:

```
*FROverZp> let f x = (895 % 922) + (1080 % 6931)*x + (2323 % 1248)*x^2
*FROverZp> let fps p = accessibleData f p
*FROverZp> let longList = map (map toInteger2) $ wellOrd $
  map (\p -> zip (firstDifsp p (fps p)) (repeat p)) bigPrimes
*FROverZp> map recCRT' longList
[(895 % 922,805479325081)
,(17448553 % 8649888,722916888780872419)
,(2323 % 624,805479325081)
]
```

This result is consistent to that of on Q:

```
*FROverZp> :l Univariate
[1 of 2] Compiling Polynomials      ( Polynomials.hs, interpreted )
[2 of 2] Compiling Univariate       ( Univariate.lhs, interpreted )
Ok, modules loaded: Univariate, Polynomials.
*Univariate> let f x = (895 % 922) + (1080 % 6931)*x + (2323 % 1248)*x^2
*Univariate> firstDifs (map f [0..20])
[895 % 922,17448553 % 8649888,2323 % 624]
```

3.1.5 `list2polZp`: from the output list to coefficient lists

Finally we get the function which takes an output list of our unknown univariate polynomial and returns the coefficient.

```
> list2firstDifZp' fs =
>   map recCRT' $ map (map toInteger2) $ wellOrd $ map helper bigPrimes
```

```
>   where helper p = zip (firstDifsp p (accessibleData' fs p)) (repeat p)
```

```
*FROverZp> let f x = (895 % 922) + (1080 % 6931)*x + (2323 % 1248)*x^2
*FROverZp> let fs = map f [0..]
*FROverZp> list2firstDifZp' fs
[(895 % 922,805479325081)
,(17448553 % 8649888,722916888780872419)
,(2323 % 624,805479325081)
]
*FROverZp> map fst it
[895 % 922,17448553 % 8649888,2323 % 624]
*FROverZp> newtonC it
[895 % 922,17448553 % 8649888,2323 % 1248]
*FROverZp> npol2pol it
[895 % 922,1080 % 6931,2323 % 1248]
```

```
> list2polZp :: [Ratio Int] -> [Ratio Integer]
> list2polZp = npol2pol . newtonC . (map fst) . list2firstDifZp'
```

3.2 TBA Univariate rational functions

Chapter 4

Codes

4.1 Ffield.lhs

Listing 4.1: Ffield.lhs

```
Ffield.lhs

https://arxiv.org/pdf/1608.01902.pdf

> module Ffield where

> import Data.Ratio
> import Data.Maybe
> import Data.Numbers.Primes
> import Test.QuickCheck

> coprime :: Integral a => a -> a -> Bool
> coprime a b = gcd a b == 1

Consider a finite ring
  Z_n := [0..(n-1)]
of some Int number.
If any non-zero element has its multiplication inverse,
    then the ring is a field:

> -- Our target should be in Int.
> isField :: Int -> Bool
> isField = isPrime

Here we would like to implement the extended Euclidean
    algorithm.
```

```
See the algorithm, examples, and pseudo code at:

  https://en.wikipedia.org/wiki/Extended_Euclidean_algorithm
  http://qiita.com/bra_cat_ket/items/205c19611e21f3d422b7

> exGCD' :: (Integral n) => n -> n -> ([n], [n], [n], [n])
> exGCD' a b = (qs, rs, ss, ts)
>   where
>     qs = zipWith quot rs (tail rs)
>     rs = takeUntil (==0) r'
>     r' = steps a b
>     ss = steps 1 0
>     ts = steps 0 1
>     steps a b = rr
>       where
>         rr@(_:rs) = a:b: zipWith (-) rr (zipWith (*) qs rs)
>
> takeUntil :: (a -> Bool) -> [a] -> [a]
> takeUntil p = foldr func []
>   where
>     func x xs
>       | p x       = []
>       | otherwise = x : xs
>
> -- a*x + b*y = gcd a b
> exGCD :: Integral t => t -> t -> (t, t, t)
> exGCD a b = (g, x, y)
>   where
>     (_,r,s,t) = exGCD' a b
>     g = last r
>     x = last . init $ s
>     y = last . init $ t
>
> -- a^{-1} (in Z_p) == a `inversep` p
> inversep :: Integral a => a -> a -> Maybe a -- We also use in CRT.
> a `inversep` p = let (g,x,_) = exGCD a p in
>   if (g == 1)
>     then Just (x `mod` p) -- g==1 <=> coprime a p
>     else Nothing
>
> inversesp :: Int -> [Maybe Int]
```

```
> inversesp p = map (`inversep` p) [1..(p-1)]
>
> -- A map from Q to Z_p.
> -- p should be prime.
> modp :: Ratio Int -> Int -> Maybe Int
> q `modp` p
>   | coprime b p = Just $ (a * (bi `mod` p)) `mod` p
>   | otherwise   = Nothing
>   where
>     (a,b)   = (numerator q, denominator q)
>     Just bi = b `inversep` p
>
> -- This is guess function without Chinese Reminder Theorem.
> guess :: Integral t =>
>          (Maybe t, t)        -- (q `modp` p, p)
>       -> Maybe (Ratio t, t)
> guess (Nothing, _) = Nothing
> guess (Just a, p) = let (_,rs,ss,_) = exGCD' a p in
>   Just (select rs ss p, p)
>     where
>       select :: Integral t => [t] -> [t] -> t -> Ratio t
>       select [] _ _ = 0%1
>       select (r:rs) (s:ss) p
>         | s /= 0 && r*r <= p && s*s <= p = r%s
>         | otherwise = select rs ss p
>
> -- Hard code of big primes
> bigPrimes :: [Int]
> bigPrimes = dropWhile (<10^4) primes
> -- bigPrimes = take 10 $ dropWhile (<10^4) primes
> -- 10 is just for "practical" reason.

  *Ffield> let knownData q = zip (map (modp q) bigPrimes) bigPrimes
  *Ffield> let ds = knownData (12%13)
  *Ffield> map guess ds
  [Just (12 % 13,10007)
  ,Just (12 % 13,10009)
  ,Just (12 % 13,10037)
  ,Just (12 % 13,10039) ..

  *Ffield> let ds = knownData (112%113)
  *Ffield> map guess ds
```

```
  [Just ((-39) % 50,10007)
  ,Just ((-41) % 48,10009)
  ,Just ((-69) % 20,10037)
  ,Just ((-71) % 18,10039) ..

--
Chinese Remainder Theorem, and its usage

> imagesAndPrimes :: Ratio Int -> [(Maybe Int, Int)]
> imagesAndPrimes q = zip (map (modp q) bigPrimes) bigPrimes

Our data is a list of the type
  [(Maybe Int, Int)]
In order to use CRT, we should cast its type.

> toInteger2 :: [(Maybe Int, Int)] -> [(Maybe Integer, Integer)]
> toInteger2 = map helper
>   where
>     helper (x,y) = (fmap toInteger x, toInteger y)
>
> crtRec':: Integral a => (Maybe a, a) -> (Maybe a, a) -> (Maybe a, a)
> crtRec' (Nothing,p) (_,q)       = (Nothing, p*q)
> crtRec' (_,p)       (Nothing,q) = (Nothing, p*q)
> crtRec' (Just a1,p1) (Just a2,p2) = (Just a,p)
>   where
>     a = (a1*p2*m2 + a2*p1*m1) `mod` p
>     Just m1 = p1 `inversep` p2
>     Just m2 = p2 `inversep` p1
>     p = p1*p2
>
> matches3 :: Eq a => [Maybe (a,b)] -> Maybe (a,b)
> matches3  (b1@(Just (q1,p1)):bb@((Just (q2,_)):(Just (q3,_)):_))
>   | q1==q2 && q2==q3 = b1
>   | otherwise        = matches3 bb
> matches3 _ = Nothing

  *Ffield> let ds = imagesAndPrimes (1123%1135)
  *Ffield> map guess ds
  [Just (25 % 52,10007)
  ,Just ((-81) % 34,10009)
  ,Just ((-88) % 63,10037) ..
```

```
  *Ffield> matches3 it
  Nothing

  *Ffield> scanl1 crtRec' ds

  *Ffield> scanl1 crtRec' . toInteger2 $ ds
  [(Just 3272,10007)
  ,(Just 14913702,100160063)
  ,(Just 298491901442,1005306552331) ..

  *Ffield> map guess it
  [Just (25 % 52,10007)
  ,Just (1123 % 1135,100160063)
  ,Just (1123 % 1135,1005306552331)
  ,Just (1123 % 1135,10092272478850909) ..

  *Ffield> matches3 it
  Just (1123 % 1135,100160063)
```

The final reconstruction function takes a list of Z_p values and returns the three times matched guess.

```
> reconstruct :: [(Maybe Int, Int)] -> Maybe (Ratio Integer, Integer)
> reconstruct = matches3 . map guess . scanl1 crtRec' . toInteger2

--

> aux :: [(Maybe Int, Int)] -> Ratio Int
> aux = fromRational . fst . fromJust . reconstruct
>
> prop_rec :: Ratio Int -> Bool
> prop_rec q = q == aux ds
>   where
>     ds = imagesAndPrimes q

  *Ffield> quickCheck prop_rec
  +++ OK, passed 100 tests.

  *Ffield> verboseCheck prop_rec
  Passed:
  0 % 1
  Passed:
```

```
1329564279259 % 689966835691
Passed:
(-4749106441063) % 5354272769957
Passed:
24393617614032 % 3429811214201
Passed:
11317350866338 % 2376141260619
Passed:
(-39223552414434) % 7602900239041
Passed:
11646784360896 % 1038773197715
Passed:
(-49349171845055) % 3752783115927
Passed:
(-2655767154053) % 2023542479048
Passed:
46861895449961 % 9804154238025
Passed:
1432195821037 % 1223607419737
Passed:
36703237585644 % 1206154505383
Passed:
85403617062851 % 2493712836857
Passed:
(-932690014756) % 2008903765881
Passed:
(-1361504657430) % 5532549345619
Passed:
(-29205487754653) % 2287912091229
Passed:
(-41868516551014) % 4135899903201
Passed:
(-165759270559576) % 2472941838409
Passed:
133806538589611 % 4499617533687
Passed:
(-17620738360931) % 6240948007885
Passed:
39459205935 % 497205472096
Passed:
117939084165907 % 6428931743209
Passed:
(-39205881063583) % 914594476839
Passed:
85731534523169 % 177493065396
```

```
Passed:
22844351052383 % 625751286364
Passed:
(-41929177046869) % 7717513998384
Passed:
(-239104997136076) % 6797515055957
Passed:
5217952768959 % 58314961636
Passed:
(-16757243373579) % 51622205471
Passed:
(-20666882197275) % 1468620478768
Passed:
(-136479419289527) % 6830615084400
Passed:
240244445293065 % 3788154937217
Passed:
(-33372393134923) % 371591225618
Passed:
(-101391102970683) % 1939379047834
Passed:
(-309270804437319) % 8714166158161
Passed:
51669639177678 % 6275195507707
Passed:
(-55827295694032) % 179632819809
Passed:
(-114812798039573) % 700296073152
Passed:
(-178448558145370) % 7607296029601
Passed:
23572878370179 % 1057593174922
Passed:
20398146164919 % 2939076586582
Passed:
(-291447760874087) % 185244184723
Passed:
329433704076007 % 1794485745740
Passed:
(-336625542209231) % 1599122245520
Passed:
(-19873460891647) % 1422075885668
Passed:
(-152680242051531) % 4981173771512
Passed:
```

```
(-424983220468388) % 7464862254053
Passed:
(-7574293347638) % 1657127826349
Passed:
(-203671927794075) % 6166500691591
Passed:
(-152003488487855) % 4130192768847
Passed:
(-63707410228315) % 4680142309544
Passed:
(-10545640427299) % 2918419691242
Passed:
(-53037754892432) % 2711758675081
Passed:
213538441367933 % 1513965455185
Passed:
(-431088172705814) % 1197611173437
Passed:
55694987217273 % 1329086881162
Passed:
(-39053206137251) % 861738484033
Passed:
135704343771202 % 568815043263
Passed:
250530684965279 % 1811055503527
Passed:
43497985509891 % 2566986060653
Passed:
141711472568173 % 98312031422
Passed:
121278120735174 % 7291914184429
Passed:
(-147562462382707) % 1771917424401
Passed:
(-324880460707855) % 3174757048141
Passed:
(-524141830329171) % 4257616728179
Passed:
402302293029904 % 6006135033215
Passed:
109760026030327 % 9235795361984
Passed:
299680927494929 % 2702925540266
Passed:
(-543368241802641) % 2107283788940
```

```
Passed:
247271165585597 % 5434442463210
Passed:
(-139088214814555) % 1552244371702
Passed:
599098456669089 % 385667799901
Passed:
(-203684593105028) % 2274842279547
Passed:
42525242468213 % 48119845245
Passed:
(-105766208155669) % 1558441423963
Passed:
(-412407833111131) % 4188726274137
Passed:
225552331901157 % 8818333576793
Passed:
(-129129125253969) % 1828301954564
Passed:
281086187565442 % 6778036064417
Passed:
(-566341489448993) % 7383824017887
Passed:
10836500052331 % 198735103052
Passed:
807851942917132 % 4690669292593
Passed:
59853908586384 % 436894123951
Passed:
(-373353472029451) % 2562548788743
Passed:
28725866844478 % 3261694948765
Passed:
(-277342703389826) % 9273579280753
Passed:
(-538083608317924) % 6023228287859
Passed:
(-705036515362431) % 9703276577398
Passed:
(-95322573966737) % 2904547141235
Passed:
77781855022925 % 2086621157657
Passed:
(-199176972632101) % 474722076748
Passed:
```

```
  827853802907850 % 4443227503433
  Passed:
  286584402406559 % 1003462078535
  Passed:
  (-195595743485494) % 4643981758567
  Passed:
  244786496629207 % 2885416954375
  Passed:
  (-830786791620) % 23171578669
  Passed:
  (-886324448647) % 5672970176
  Passed:
  1249713170911 % 159111156329
  Passed:
  (-99490220475093) % 1255980833900
  Passed:
  (-445170783092013) % 2379282629701
  +++ OK, passed 100 tests.
```

4.2 Polynomials.hs

Listing 4.2: Polynomials.hs

```
-- Polynomials.hs
-- http://homepages.cwi.nl/~jve/rcrh/Polynomials.hs

module Polynomials where

default (Integer, Rational, Double)

-- scalar multiplication
infixl 7 .*
(.*) :: Num a => a -> [a] -> [a]
c .* []     = []
c .* (f:fs) = c*f : c .* fs

z :: Num a => [a]
z = [0,1]

-- polynomials, as coefficients lists
instance (Num a, Ord a) => Num [a] where
  fromInteger c = [fromInteger c]
  -- operator overloading
  negate []     = []
  negate (f:fs) = (negate f) : (negate fs)
```

```

  signum [] = []
  signum gs
    | signum (last gs) < (fromInteger 0) = negate z
    | otherwise = z

  abs [] = []
  abs gs
    | signum gs == z = gs
    | otherwise      = negate gs

  fs     + []     = fs
  []     + gs     = gs
  (f:fs) + (g:gs) = f+g : fs+gs

  fs     * []     = []
  []     * gs     = []
  (f:fs) * gg@(g:gs) = f*g : (f .* gs + fs * gg)

delta :: (Num a, Ord a) => [a] -> [a]
delta = ([1,-1] *)

shift :: [a] -> [a]
shift = tail

p2fct :: Num a => [a] -> a -> a
p2fct [] x = 0
p2fct (a:as) x = a + (x * p2fct as x)

comp :: (Eq a, Num a, Ord a) => [a] -> [a] -> [a]
comp _      []      = error ".."
comp []     _       = []
comp (f:fs) g0@(0:gs) = f : gs * (comp fs g0)
comp (f:fs) gg@(g:gs) = ([f] + [g] * (comp fs gg))
                      + (0 : gs * (comp fs gg))

deriv :: Num a => [a] -> [a]
deriv []     = []
deriv (f:fs) = deriv1 fs 1
  where
    deriv1 []     _ = []
    deriv1 (g:gs) n = n*g : deriv1 gs (n+1)
```

4.3 Univariate.lhs

Listing 4.3: Univariate.lhs

```
Univariate.lhs

> module Univariate where
> import Data.Ratio
> import Polynomials

From the output list
  map f [0..]
of a polynomial
  f :: Int -> Ratio Int
we reconstrunct the canonical form of f.

> -- difference analysis
> difs :: (Num a) => [a] -> [a]
> difs [] = []
> difs [_] = []
> difs (i:jj@(j:js)) = j-i : difs jj
>
> difLists :: (Eq a, Num a) => [[a]] -> [[a]]
> difLists [] = []
> difLists xx@(xs:xss) =
>   if isConst xs then xx
>                 else difLists $ difs xs : xx
>   where
>     isConst (i:jj@(j:js)) = all (==i) jj
>     isConst _ = error "difLists: lack of data, or not a polynomial"
>
> -- This degree function is "strict", so only take finite list.
> degree' :: (Eq a, Num a) => [a] -> Int
> degree' xs = length (difLists [xs]) -1
>
> -- This degree function can compute the degree of infinite list.
> degreeLazy :: (Eq a, Num a) => [a] -> Int
> degreeLazy xs = helper xs 0
>   where
>     helper as@(a:b:c:_) n
>       | a==b && b==c = n
>       | otherwise    = helper (difs as) (n+1)
>
> -- This is a hyblid version, safe and lazy.
> degree :: (Num a, Eq a) => [a] -> Int
```

```
> degree xs = let l = degreeLazy xs in
>   degree' $ take (l+2) xs

Newton interpolation formula
First we introduce a new infix symbol for the operation of taking a falling power.

> infixr 8 ^- -- falling power
> (^-) :: (Eq a, Num a) => a -> a -> a
> x ^- 0 = 1
> x ^- n = (x ^- (n-1)) * (x - n + 1)

Claim (Newton interpolation formula)
A polynomial f of degree n is expressed as
  f(z) = \sum_{k=0}^n  (diff^n(f)(0)/k!) * (x ^- n)
where diff^n(f) is the n-th difference of f.

Example
Consider a polynomial f = 2*x^3+3*x.

In general, we have no prior knowledge of this form, but we know the sequences as a list of outputs:

  Univariate> let f x = 2*x^3+3*x
  Univariate> take 10 $ map f [0..]
  [0,5,22,63,140,265,450,707,1048,1485]
  Univariate> degree $ take 10 $ map f [0..]
  3

Let us try to get differences:

  Univariate> difs $ take 10 $ map f [0..]
  [5,17,41,77,125,185,257,341,437]
  Univariate> difs it
  [12,24,36,48,60,72,84,96]
  Univariate> difs it
  [12,12,12,12,12,12,12]

Or more simply take difLists:

  Univariate> difLists [take 10 $ map f [0..]]
  [[12,12,12,12,12,12,12]
  ,[12,24,36,48,60,72,84,96]
  ,[5,17,41,77,125,185,257,341,437]
  ,[0,5,22,63,140,265,450,707,1048,1485]
```

```
  ]

What we need is the heads of above lists.

  Univariate> map head it
  [12,12,5,0]

Newton interpolation formula gives
  f' x = 0*(x ^- 0) `div` (0!) + 5*(x ^- 1) `div` (1!) + 12*(x ^- 2) `div` (2!) + 12*(x ^- 3) `div` (3!)
       = 5*(x ^- 1) + 6*(x ^- 2) + 2*(x ^- 3)
So

  Univariate> let f x = 2*x^3+3*x
  Univariate> let f' x = 5*(x ^- 1) + 6*(x ^- 2) + 2*(x ^- 3)
  Univariate> take 10 $ map f [0..]
  [0,5,22,63,140,265,450,707,1048,1485]
  Univariate> take 10 $ map f' [0..]
  [0,5,22,63,140,265,450,707,1048,1485]

Assume the differences are given in a list
  [x_0, x_1 ..]
where x_i = diff^k(f)(0).
Then the implementation of the Newton interpolation formula is as follows:

> newtonC :: (Fractional t, Enum t) => [t] -> [t]
> newtonC xs = [x / factorial k | (x,k) <- zip xs [0..]]
>   where
>     factorial k = product [1..fromInteger k]

  Univariate> let f x = 2*x^3+3*x
  Univariate> take 10 $ map f [0..]
  [0,5,22,63,140,265,450,707,1048,1485]
  Univariate> difLists [it]
  [[12,12,12,12,12,12,12]
  ,[12,24,36,48,60,72,84,96]
  ,[5,17,41,77,125,185,257,341,437]
  ,[0,5,22,63,140,265,450,707,1048,1485]
  ]
  Univariate> reverse $ map head it
  [0,5,12,12]
  Univariate> newtonC it
  [0 % 1,5 % 1,6 % 1,2 % 1]
```

```

The list of first differences can be computed as follows:

> firstDifs :: (Eq a, Num a) => [a] -> [a]
> firstDifs xs = reverse $ map head $ difLists [xs]

Mapping a list of integers to a Newton representation:

> -- This implementation can take infinite list.
> list2npol :: (Integral a) => [Ratio a] -> [Ratio a]
> list2npol xs = newtonC . firstDifs $ take n xs
>   where n = (degree xs) + 2

  *Univariate> let f x = 2*x^3 + 3*x + 1%5
  *Univariate> take 10 $ map f [0..]
  [1 % 5,26 % 5,111 % 5,316 % 5,701 % 5,1326 % 5,2251 % 5,3536 % 5,5241 % 5,7426 % 5]
  *Univariate> list2npol it
  [1 % 5,5 % 1,6 % 1,2 % 1]
  *Univariate> list2npol $ map f [0..]
  [1 % 5,5 % 1,6 % 1,2 % 1]

We need to map Newton falling powers to standard powers.
This is a matter of applying combinatorics, by means of a convention formula that uses the so-called Stirling cyclic numbers (of the first kind.)
Its defining relation is
  (x ^- n) = \sum_{k=1}^n (stirlingC n k) * (-1)^(n-k) * x^k.
The key equation is
  (x ^- n) = (x ^- (n-1)) * (x-n+1)
           = x*(x ^- (n-1)) - (n-1)*(x ^- (n-1))

Therefore, an implementation is as follows:

> stirlingC :: (Integral a) => a -> a -> a
> stirlingC 0 0 = 1
> stirlingC 0 _ = 0
> stirlingC n k = stirlingC (n-1) (k-1) + (n-1)*stirlingC (n-1) k

This definition can be used to convert from falling powers to standard powers.

> fall2pol :: (Integral a) => a -> [a]
```

```
> fall2pol 0 = [1]
> fall2pol n = 0   -- No constant term.
>             : [(-1)^(n-k) * stirlingC n k| k<-[1..n]]

We use this to convert Newton representations to standard polynomials in coefficients list representation.
Here we have uses sum to collect same order terms in list representation.

> -- For later convenience, we relax the type annotation.
> -- npol2pol :: (Integral a) => [Ratio a] -> [Ratio a]
> npol2pol :: (Ord t, Num t) => [t] -> [t]
> npol2pol xs = sum [ [x] * map fromInteger (fall2pol k)
>                   | (x,k) <- zip xs [0..]
>                   ]

Finally, here is the function for computing a polynomial from an output sequence:

> list2pol :: (Integral a) => [Ratio a] -> [Ratio a]
> list2pol = npol2pol . list2npol

Reconstruction as curve fitting
  *Univariate> let f x = 2*x^3 + 3*x + 1%5
  *Univariate> take 10 $ map f [0..]
  [1 % 5,26 % 5,111 % 5,316 % 5,701 % 5,1326 % 5,2251 % 5,3536 % 5,5241 % 5,7426 % 5]
  *Univariate> list2npol it
  [1 % 5,5 % 1,6 % 1,2 % 1]
  *Univariate> list2npol $ map f [0..]
  [1 % 5,5 % 1,6 % 1,2 % 1]
  *Univariate> list2pol $ map (\n -> 1%3 + (3%5)*n + (5%7)*n^2) [0..]
  [1 % 3,3 % 5,5 % 7]
  *Univariate>  list2pol [0,1,5,14,30,55]
  [0 % 1,1 % 6,1 % 2,1 % 3]
  *Univariate> map (p2fct $ list2pol [0,1,5,14,30,55]) [0..6]
  [0 % 1,1 % 1,5 % 1,14 % 1,30 % 1,55 % 1,91 % 1]

--

Thiele's interpolation formula
https://rosettacode.org/wiki/Thiele%27s_interpolation_formula#Haskell
```

```
http://mathworld.wolfram.com/ThielesInterpolationFormula.html

reciprocal difference
Using the same notation of
https://rosettacode.org/wiki/Thiele%27s_interpolation_formula#C

> rho :: (Integral a) =>
>        [Ratio a] -- A list of output of f :: a -> Ratio a
>     -> a -> Int -> Ratio a
> rho fs 0 i = fs !! i
> rho fs n i
>   | n < 0     = 0
>   | otherwise = (n*den)%num + rho fs (n-2) (i+1)
>   where
>     num  = numerator next
>     den  = denominator next
>     next = rho fs (n-1) (i+1) - rho fs (n-1) i

Note that (%) has the following type,
  (%) :: Integral a => a -> a -> Ratio a

> a :: (Integral a) => [Ratio a] -> a -> Ratio a
> a fs 0 = head fs
> a fs n = rho fs n 0 - rho fs (n-2) 0

Consider the following continuous fraction form.
  (%i25) f(x) := 1+(x/(2+(x-1)/(3+(x-2)/4)));
  (%o25) f(x):=x/(2+(x-1)/(3+(x-2)/4))+1
  (%i26) ratsimp(f(x));
  (%o26) (x^2+16*x+16)/(16+6*x)

  *Univariate> map (a fs) [0..]
  [1 % 1,2 % 1,3 % 1,4 % 1,*** Exception: Ratio has zero denominator

  *Univariate> let func x = (x^2 + 16*x + 16)%(6*x + 16)
  *Univariate> let fs = map func [0..]
  *Univariate> take 5 $ map (rho fs 0) [0..]
  [1 % 1,3 % 2,13 % 7,73 % 34,12 % 5]
  *Univariate> take 5 $ map (rho fs 1) [0..]
  [2 % 1,14 % 5,238 % 69,170 % 43,230 % 53]
  *Univariate> take 5 $ map (rho fs 2) [0..]
```

```
  [4 % 1,79 % 16,269 % 44,667 % 88,413 % 44]
  *Univariate> take 5 $ map (rho fs 3) [0..]
  [6 % 1,6 % 1,6 % 1,6 % 1,6 % 1]

> tDegree :: Integral a => [Ratio a] -> a
> tDegree fs = helper fs 0
>   where
>     helper fs n
>       | isConstants fs' = n
>       | otherwise       = helper fs (n+1)
>       where
>         fs' = map (rho fs n) [0..]
>     isConstants (i:j:_) = i==j -- 2 times match
> --  isConstants (i:j:k_) = i==j && j==k -- 3 times match

  *Univariate> let h t = (3+6*t+18*t^2)%(1+2*t+20*t^2)
  *Univariate> let hs = map h [0..]
  *Univariate> tDegree hs
  4
  *Univariate> map (a hs) [0..(tDegree hs)]
  [3 % 1,(-23) % 42,(-28) % 13,767 % 14,7 % 130]

With Maxima,
  (%i35) h(t) := 3+t/((-23/42)+(t-1)/((-28/13)+(t-2)/((767/14)+(t-3)/(7/130))));

  (%o35) h(t):=t/((-23)/42+(t-1)/((-28)/13+(t-2)/(767/14+(t-3)/(7/130))))+3
  (%i36) ratsimp(h(t));

  (%o36) (18*t^2+6*t+3)/(1+2*t+20*t^2)

> thieleC :: (Integral a) => [Ratio a] -> [Ratio a]
> thieleC lst = map (a lst) [0..(tDegree lst)]

  *Univariate> thieleC hs
  [3 % 1,(-23) % 42,(-28) % 13,767 % 14,7 % 130]

We need a convertor from this thiele sequence to continuous form of rational function.

> nextStep [a0,a1] (v:_)  = a0 + v/a1
> nextStep (a:as)  (v:vs) = a + (v / nextStep as vs)
>
```

```
> -- From thiele sequence to (rational) function.
> thiele2ratf :: Integral a => [Ratio a] -> (Ratio a -> Ratio a)
> thiele2ratf as x
>   | x == 0    = head as
>   | otherwise = nextStep as [x,x-1 ..]

  *Univariate> let h t = (3+6*t+18*t^2)%(1+2*t+20*t^2)
  *Univariate> let hs = map h [0..]
  *Univariate> let as = thieleC hs
  *Univariate> as
  [3 % 1,(-23) % 42,(-28) % 13,767 % 14,7 % 130]
  *Univariate> let th x = thiele2ratf as x
  *Univariate> take 5 hs
  [3 % 1,27 % 23,87 % 85,183 % 187,45 % 47]
  *Univariate> map th [0..5]
  [3 % 1,27 % 23,87 % 85,183 % 187,45 % 47,69 % 73]

We represent a rational function by a tuple of coefficient lists:
  (ns,ds) :: ([Ratio Int],[Ratio Int])
where ns and ds are coef-list-rep of numerator polynomial and denominator polynomial.
Here is a translator from coefficients lists to rational function.

> -- similar to p2fct
> lists2ratf :: (Integral a) =>
>               ([Ratio a],[Ratio a]) -> (Ratio a -> Ratio a)
> lists2ratf (ns,ds) x = p2fct ns x / p2fct ds x

  *Univariate> let frac x = lists2ratf ([1,1%2,1%3],[2,2%3]) x
  *Univariate> take 10 $ map frac [0..]
  [1 % 2,11 % 16,1 % 1,11 % 8,25 % 14,71 % 32,8 % 3,25 % 8,79 % 22,65 % 16]
  *Univariate> let ffrac x = (1+(1%2)*x+(1%3)*x^2)/(2+(2%3)*x)
  *Univariate> take 10 $ map ffrac [0..]
  [1 % 2,11 % 16,1 % 1,11 % 8,25 % 14,71 % 32,8 % 3,25 % 8,79 % 22,65 % 16]

The following canonicalizer reduces the tuple-rep of rational function in canonical form
```

```
That is, the coefficien of the lowest degree term of the denominator to be 1.
However, since our input starts from 0 and this means firstNonzero is the same as head.

> canonicalize :: (Integral a) => ([Ratio a],[Ratio a]) -> ([Ratio a],[Ratio a])
> canonicalize rat@(ns,ds)
>   | dMin == 1 = rat
>   | otherwise = (map (/dMin) ns, map (/dMin) ds)
>   where
>     dMin = firstNonzero ds
>     firstNonzero [a] = a -- head
>     firstNonzero (a:as)
>       | a /= 0    = a
>       | otherwise = firstNonzero as

What we need is a translator from Thiele coefficients to this tuple-rep.

> thiele2coef :: (Integral a) => [Ratio a] -> ([Ratio a],[Ratio a])
> thiele2coef as = canonicalize $ t2r as 0
>   where
>     t2r [an,an'] n = ([an*an'-n,1],[an'])
>     t2r (a:as)   n = ((a .* num) + ([-n,1] * den), num)
>       where
>         (num, den) = t2r as (n+1)
>
> list2rat :: (Integral a) => [Ratio a] -> ([Ratio a], [Ratio a])
> list2rat = thiele2coef . thieleC

  *Univariate> let h t = (3+6*t+18*t^2)%(1+2*t+20*t^2)
  *Univariate> let hs = map h [0..]
  *Univariate> take 5 hs
  [3 % 1,27 % 23,87 % 85,183 % 187,45 % 47]
  *Univariate> let th x = thiele2ratf as x
  *Univariate> map th [0..5]
  [3 % 1,27 % 23,87 % 85,183 % 187,45 % 47,69 % 73]
  *Univariate> as
  [3 % 1,(-23) % 42,(-28) % 13,767 % 14,7 % 130]
  *Univariate> thiele2coef as
  ([3 % 1,6 % 1,18 % 1],[1 % 1,2 % 1,20 % 1])
```

4.4 Multivariate.lhs

Listing 4.4: Multivariate.lhs

```
Multivariate.lhs

> module Multivariate
>   where

> import Data.Ratio
> import Univariate
>   ( degree, list2pol
>   , thiele2ratf, lists2ratf, thiele2coef, list2rat
>   )

Let us start 2-variate polynomials.

  *Multivariate> let f z1 z2 = 3+2*z1+4*z2+7*z1^2+5*z1*z2+6*z2^2
  *Multivariate> [[f x y | y <- [0..9]] | x <- [0..9]]
  [[3,13,35,69,115,173,243,325,419,525]
  ,[12,27,54,93,144,207,282,369,468,579]
  ,[35,55,87,131,187,255,335,427,531,647]
  ,[72,97,134,183,244,317,402,499,608,729]
  ,[123,153,195,249,315,393,483,585,699,825]
  ,[188,223,270,329,400,483,578,685,804,935]
  ,[267,307,359,423,499,587,687,799,923,1059]
  ,[360,405,462,531,612,705,810,927,1056,1197]
  ,[467,517,579,653,739,837,947,1069,1203,1349]
  ,[588,643,710,789,880,983,1098,1225,1364,1515]
  ]

Assuming the list of lists is a matrix of 2-variate function's values, (f i j).

> tablize :: (Enum t1, Num t1) => (t1 -> t1 -> t) -> Int -> [[t]]
> tablize f n = [[f x y | y <- range] | x <- range]
>   where
>     range = take n [0..]

  *Multivariate> let fTable = tablize f 10
  *Multivariate> map list2pol fTable
  [[3 % 1,4 % 1,6 % 1]
  ,[12 % 1,9 % 1,6 % 1]
```

```
 ,[35 % 1,14 % 1,6 % 1]
 ,[72 % 1,19 % 1,6 % 1]
 ,[123 % 1,24 % 1,6 % 1]
 ,[188 % 1,29 % 1,6 % 1]
 ,[267 % 1,34 % 1,6 % 1]
 ,[360 % 1,39 % 1,6 % 1]
 ,[467 % 1,44 % 1,6 % 1]
 ,[588 % 1,49 % 1,6 % 1]
 ]

> wellOrd :: [[a]] -> [[a]]
> wellOrd xss
>   | null (head xss) = []
>   | otherwise       = map head xss : wellOrd (map tail xss)

 *Multivariate> let f z1 z2 = 3+2*z1+4*z2+7*z1^2+5*z1*z2+6*z2^2
 *Multivariate> let fTable = tablize f 10
 *Multivariate> map list2pol fTable
 [[3 % 1,4 % 1,6 % 1]
 ,[12 % 1,9 % 1,6 % 1]
 ,[35 % 1,14 % 1,6 % 1]
 ,[72 % 1,19 % 1,6 % 1]
 ,[123 % 1,24 % 1,6 % 1]
 ,[188 % 1,29 % 1,6 % 1]
 ,[267 % 1,34 % 1,6 % 1]
 ,[360 % 1,39 % 1,6 % 1]
 ,[467 % 1,44 % 1,6 % 1]
 ,[588 % 1,49 % 1,6 % 1]
 ]
 *Multivariate> wellOrd it
 [[3 % 1,12 % 1,35 % 1,72 % 1,123 % 1,188 % 1,267 % 1,360 % 1,467 % 1,588 % 1]
 ,[4 % 1,9 % 1,14 % 1,19 % 1,24 % 1,29 % 1,34 % 1,39 % 1,44 % 1,49 % 1]
 ,[6 % 1,6 % 1,6 % 1,6 % 1,6 % 1,6 % 1,6 % 1,6 % 1,6 % 1,6 % 1]
 ]
 *Multivariate> map list2pol it
 [[3 % 1,2 % 1,7 % 1]
 ,[4 % 1,5 % 1]
 ,[6 % 1]]

> table2pol :: [[Ratio Integer]] -> [[Ratio Integer]]
```

```
> table2pol = map list2pol . wellOrd . map list2pol

  *Multivariate> let g x y = 1+7*x + 8*y + 10*x^2 + x*y
      +9*y^2
  *Multivariate> table2pol $ tablize g 5
  [[1 % 1,7 % 1,10 % 1],[8 % 1,1 % 1],[9 % 1]]

--

Next, 2-variate rational functions.

  *Multivariate> let h x y = (3+2*x+4*y+7*x^2+5*x*y+6*y
      ^2) % (1+7*x+8*y+10*x^2+x*y+9*y^2)
  *Multivariate> let auxh x y t = h (t*x) (t*y)
  *Multivariate> let h x y = (3+2*x+4*y+7*x^2+5*x*y+6*y
      ^2)% (1+7*x+8*y+10*x^2+x*y+9*y^2)
  *Multivariate> let auxh x y t = h (t*x) (t*y)

Using the homogenious property, we just take x=1:

  *Multivariate> let auxhs = [map (auxh 1 y) [0..5] | y
      <- [0..5]]
  *Multivariate> auxhs
  [[3 % 1,2 % 3,7 % 11,9 % 14,41 % 63,94 % 143]
  ,[3 % 1,3 % 4,29 % 37,183 % 226,105 % 127,161 % 192]
  ,[3 % 1,3 % 4,187 % 239,201 % 251,233 % 287,77 % 94]
  ,[3 % 1,31 % 42,335 % 439,729 % 940,425 % 543,1973 %
      2506]
  ,[3 % 1,8 % 11,59 % 79,291 % 385,681 % 895,528 % 691]
  ,[3 % 1,23 % 32,155 % 211,1707 % 2302,1001 % 1343,4663
      % 6236]
  ]

  *Multivariate> map list2rat auxhs
  [([3 % 1,2 % 1,7 % 1],[1 % 1,7 % 1,10 % 1])
  ,([3 % 1,6 % 1,18 % 1],[1 % 1,15 % 1,20 % 1])
  ,([3 % 1,10 % 1,41 % 1],[1 % 1,23 % 1,48 % 1])
  ,([3 % 1,14 % 1,76 % 1],[1 % 1,31 % 1,94 % 1])
  ,([3 % 1,18 % 1,123 % 1],[1 % 1,39 % 1,158 % 1])
  ,([3 % 1,22 % 1,182 % 1],[1 % 1,47 % 1,240 % 1])
  ]
  *Multivariate> map fst it
  [[3 % 1,2 % 1,7 % 1]
  ,[3 % 1,6 % 1,18 % 1]
  ,[3 % 1,10 % 1,41 % 1]
```

```
  ,[3 % 1,14 % 1,76 % 1]
  ,[3 % 1,18 % 1,123 % 1]
  ,[3 % 1,22 % 1,182 % 1]
  ]
  *Multivariate> wellOrd it
  [[3 % 1,3 % 1,3 % 1,3 % 1,3 % 1,3 % 1]
  ,[2 % 1,6 % 1,10 % 1,14 % 1,18 % 1,22 % 1]
  ,[7 % 1,18 % 1,41 % 1,76 % 1,123 % 1,182 % 1]
  ]
  *Multivariate> map list2pol it
  [[3 % 1],[2 % 1,4 % 1],[7 % 1,5 % 1,6 % 1]]

So, the numerator is given by

  *Multivariate> map list2pol . wellOrd . map (fst . list2rat) $ auxhs
  [[3 % 1],[2 % 1,4 % 1],[7 % 1,5 % 1,6 % 1]]

and the denominator is

  *Multivariate> map list2pol . wellOrd . map (snd . list2rat) $ auxhs
  [[1 % 1],[7 % 1,8 % 1],[10 % 1,1 % 1,9 % 1]]

> table2ratf table = (t2r fst table, t2r snd table)
>   where
>     t2r third = map list2pol . wellOrd . map (third . list2rat)

  *Multivariate> table2ratf auxhs
  ([[3 % 1],[2 % 1,4 % 1],[7 % 1,5 % 1,6 % 1]],[[1 % 1],[7 % 1,8 % 1],[10 % 1,1 % 1,9 % 1]])
```

4.5 FROverZp.lhs

Listing 4.5: FROverZp.lhs

```
FROverZp.lhs

> module FROverZp where

Functional Reconstruction over finite field Z_p

> import Data.Ratio
> import Data.Maybe
```

```
> import Data.Numbers.Primes
> import Data.List (null)
> import Control.Monad (sequence)
>
> import Ffield (modp)
> -- , inversep, bigPrimes, recCRT, recCRT')
> -- import Univariate (npol2pol, newtonC)

Univariate Polynomial case
Our target is a univariate polynomial
  f :: (Integral a) =>
       Ratio a -> Ratio a -- Real?

> -- Function-modular, now our modp function is wrapped by Maybe.
> fmodp :: (a -> Ratio Int) -> Int -> a -> Maybe Int
> f `fmodp` p = (`modp` p) . f

  *FROverZp> let f x = (1%3) + (3%5)*x + (7%13)*x^2
  *FROverZp> take 10 $ map (f `fmodp` 13)  [0..]
  [Just 9,Nothing,Nothing,Nothing,Nothing,Nothing,Nothing,Nothing,Nothing,Nothing]
  *FROverZp> take 10 $ map (f `fmodp` 19)  [0..]
  [Just 13,Just 8,Just 7,Just 10,Just 17,Just 9,Just 5,Just 5,Just 9,Just 17]

Difference analysis over Z_p
Every arithmetic should be on Z_p, i.e., (`mod` p).

> accessibleData :: (Num a, Enum a) => (a -> Ratio Int) -> Int -> [Maybe Int]
> accessibleData f p = take p $ map (f `fmodp` p) [0..]
>
> accessibleData' :: [Ratio Int] -> Int -> [Maybe Int]
> accessibleData' fs p = take p $ map (`modp` p) fs

  *FROverZp> let helper x y = (-) <$> x <*> y
  *FROverZp> :type helper
  helper :: (Applicative f, Num b) => f b -> f b -> f b
  *FROverZp> :t map helper
  map helper :: (Applicative f, Num b) => [f b] -> [f b -> f b]
  *FROverZp> let myDif xs = zipWith helper (tail xs) xs
  *FROverZp> myDif [Just 5,Just 0,Just 2,Just 4,Just 6,Just 1,Just 3]
```

```
[Just (-5),Just 2,Just 2,Just 2,Just (-5),Just 2]
*FROverZp> map (fmap (`mod` 13)) it
[Just 8,Just 2,Just 2,Just 2,Just 8,Just 2]
*FROverZp> map (fmap (`mod` 13)) . myDif $ [Just 5,Just 0,Just 2,Just 4,Just 6,Just 1,Just 3, Nothing, Just 1, Just 2]
[Just 8,Just 2,Just 2,Just 2,Just 8,Just 2,Nothing, Nothing,Just 1]

> -- difsp :: (Applicative f, Integral b) => b -> [f b] -> [f b]
> difsp :: Applicative f => Int -> [f Int] -> [f Int]
> difsp p = map (fmap (`mod` p)) . difsp'
>   where
>     difsp' :: (Applicative f, Num b) => [f b] -> [f b]
>     difsp' xs = zipWith helper (tail xs) xs
>     helper :: (Applicative f, Num b) => f b -> f b -> f b
>     helper x y = (-) <$> x <*> y

*FROverZp> difsp 13 [Just 5,Just 0,Just 2,Just 4,Just 6,Just 1,Just 3, Nothing, Just 1, Just 2]
[Just 8,Just 2,Just 2,Just 2,Just 8,Just 2,Nothing, Nothing,Just 1]

> difListsp :: (Applicative f, Eq (f Int)) => Int -> [[f Int]] -> [[f Int]]
> difListsp _ [] = []
> difListsp p xx@(xs:xss) =
>   if isConst xs
>     then xx
>     else difListsp p $ difsp p xs : xx
>   where
>     isConst (i:jj@(j:js)) = all (==i) jj
>     isConst _ = error "difListsp:␣"

*FROverZp> let f x = (1%3) + (3%5)*x + (7%13)*x^2
*FROverZp> let ds = accessibleData f 101
*FROverZp> map head $ difListsp 101 [ds]
[Just 71,Just 26,Just 34]

Degree, eager and lazy versions

> degreep' :: (Applicative f, Eq (f Int)) => Int -> [f Int] -> Int
```

```
> degreep' p xs = length (difListsp p [xs]) -1
>
> degreepLazy :: (Applicative f, Num t, Eq (f Int)) => Int -> [f Int] -> t
> degreepLazy p xs = helper xs 0
>   where
>     helper as@(a:b:c:_) n
>       | a==b && b==c = n -- two times matching
>       | otherwise    = helper (difsp p as) (n+1)
>
> degreep :: (Applicative f, Eq (f Int)) => Int -> [f Int] -> Int
> degreep p xs = let l = degreepLazy p xs in
>   degreep' p $ take (l+2) xs

  *FROverZp> let f x = (1%3) + (3%5)*x + (7%13)*x^2
  *FROverZp> degreep 101 $ accessibleData f 101
  2
  *FROverZp> degreep 103 $ accessibleData f 103
  2
  *FROverZp> degreep 107 $ accessibleData f 107
  2
  *FROverZp> degreep 11 $ accessibleData f 11
  2
  *FROverZp> degreep 13 $ accessibleData f 13
  1
  *FROverZp> degreep 17 $ accessibleData f 17
  2

> -- firstDifsp :: Integral a => a -> [a] -> [a]
> firstDifsp :: (Applicative f, Eq (f Int)) => Int -> [f Int] -> [f Int]
> firstDifsp p xs = reverse $ map head $ difListsp p [xs']
>   where
>     xs' = take n xs
>     n   = 2 + degreep p xs

  *FROverZp> let f x = (1%3) + (3%5)*x + (7%13)*x^2
  *FROverZp> let fs p = accessibleData f p
  *FROverZp> firstDifsp 101 (fs 101)
  [Just 34,Just 26,Just 71]
  *FROverZp> firstDifsp 103 (fs 103)
  [Just 69,Just 36,Just 9]
  *FROverZp> firstDifsp 107 (fs 107)
```

```
  [Just 36,Just 39,Just 34]

  *FROverZp> map ourData [11,13,17,19,101,103,107]
  [[Just 4,Just 3,Just 7]
  ,[Just 9,Nothing]
  ,[Just 6,Just 15,Just 5]
  ,[Just 13,Just 14,Just 4]
  ,[Just 34,Just 26,Just 71]
  ,[Just 69,Just 36,Just 9]
  ,[Just 36,Just 39,Just 34]
  ]

  *FROverZp> let ourData' = sequence . ourData
  *FROverZp> map ourData' ourPrimes
  [Just [4,3,7]
  ,Nothing
  ,Just [6,15,5]
  ,Just [13,14,4]
  ,Just [34,26,71]
  ,Just [69,36,9]
  ,Just [36,39,34]
  ]

  *FROverZp> zip (map (sequence . ourData) smallPrimes)
     smallPrimes
  [(Just [4,3,7],11)
  ,(Nothing,13)
  ,(Just [6,15,5],17)
  ,(Just [13,14,4],19)
  ,(Just [34,26,71],101)
  ,(Just [69,36,9],103)
  ,(Just [36,39,34],107)
  ]







  Our target is this diff-list, since once we reconstruct
      the diflists from several prime fields to rational
      field, we can fully convert it to canonical form
     in Q, by applying Univariate.npol2pol.

> wellOrd :: [[a]] -> [[a]]
```

```
> wellOrd xss
>   | null (head xss) = []
>   | otherwise       = map head xss : wellOrd (map tail xss)

  *FROverZp> let f x = (1%3) + (3%5)*x + (7%6)*x^2
  *FROverZp> let fps p = accessibleData f p
  *FROverZp> let ourData p = firstDifsp p (fps p)
  *FROverZp> let fivePrimes = take 5 bigPrimes
  *FROverZp> map (\p -> zip (ourData p) (repeat p)) fivePrimes
  [[(299158,897473),(867559,897473),(299160,897473)]
  ,[(299166,897497),(329084,897497),(299168,897497)]
  ,[(598333,897499),(388918,897499),(598335,897499)]
  ,[(598345,897517),(29919,897517),(598347,897517)]
  ,[(299176,897527),(329095,897527),(299178,897527)]
  ]
  *FROverZp> wellOrd it
  [[(299158,897473),(299166,897497),(598333,897499)
   ,(598345,897517),(299176,897527)]
  ,[(867559,897473),(329084,897497),(388918,897499)
   ,(29919,897517),(329095,897527)]
  ,[(299160,897473),(299168,897497),(598335,897499)
   ,(598347,897517),(299178,897527)]
  ]
  *FROverZp> :t it
  it :: [[(Int, Int)]]

We need to transform
  Int -> Integer
to use recCRT :: Integral a => [(a, a)] -> Ratio a

  *FROverZp> let impCasted =
  [[(299158,897473),(299166,897497),(598333,897499)
   ,(598345,897517),(299176,897527)]
  ,[(867559,897473),(329084,897497),(388918,897499)
   ,(29919,897517),(329095,897527)]
  ,[(299160,897473),(299168,897497),(598335,897499)
   ,(598347,897517),(299178,897527)]
  ]
  *FROverZp> :t impCasted
  impCasted :: (Num t1, Num t) => [[(t, t1)]]
  *FROverZp> map recCRT impCasted
  [1 % 3,53 % 30,7 % 3]
  *FROverZp> map recCRT' impCastet
```

```
  [(1 % 3,897473),(53 % 30,897473),(7 % 3,897473)]

This result is consistent:
  *Univariate> let f x = (1%3) + (3%5)*x + (7%6)*x^2
  *Univariate> firstDifs (map f [0..10])
  [1 % 3,53 % 30,7 % 3]

Let us define above casting function

> toInteger2 :: (Integral a1, Integral a) => (a, a1) -> (Integer, Integer)
> toInteger2 (a,b) = (toInteger a, toInteger b)

  *FROverZp> let f x = (1%3) + (3%5)*x + (7%6)*x^2
  *FROverZp> let fps p = accessibleData f p
  *FROverZp> let ourData p = firstDifsp p (fps p)
  *FROverZp> let longList' = map (\p -> zip (ourData p) (repeat p)) bigPrimes
  *FROverZp> let longList = wellOrd longList'
  *FROverZp> :t longList
  longList :: [[(Int, Int)]]
  *FROverZp> let longList'' = map (map toInteger2) longList
  *FROverZp> :t longList''
  longList'' :: [[(Integer, Integer)]]
  *FROverZp> map recCRT longList''
  [1 % 3,53 % 30,7 % 3]
  *FROverZp> map recCRT' longList''
  [(1 % 3,897473),(53 % 30,897473),(7 % 3,897473)]
  *FROverZp> let f x = (1%3) + (3%5)*x + (7%6)*x^2
  *FROverZp> let fps p = accessibleData f p
  *FROverZp> let ourData p = firstDifsp p (fps p)
  *FROverZp> let longList' = map (\p -> zip (ourData p) (repeat p)) bigPrimes
  *FROverZp> let longList = wellOrd longList'
  *FROverZp> :t longList
  longList :: [[(Int, Int)]]
  *FROverZp> let longList'' = map (map toInteger2) longList
  *FROverZp> :t longList''
  longList'' :: [[(Integer, Integer)]]
  *FROverZp> map recCRT longList''
  [1 % 3,53 % 30,7 % 3]
  *FROverZp> map recCRT' longList''
  [(1 % 3,897473),(53 % 30,897473),(7 % 3,897473)]
```

```

Let us try another example:

  *FROverZp> let f x = (895 % 922) + (1080 % 6931)*x + (2323 % 1248)*x^2
  *FROverZp> let fps p = accessibleData f p
  *FROverZp> let longList = map (map toInteger2) $ wellOrd $ map (\p -> zip (firstDifsp p (fps p)) (repeat p)) bigPrimes
  *FROverZp> map recCRT' longList
  [(895 % 922,805479325081)
  ,(17448553 % 8649888,722916888780872419)
  ,(2323 % 624,805479325081)
  ]

This result is consistent to that of on Q:

  *FROverZp> :l Univariate
  [1 of 2] Compiling Polynomials      ( Polynomials.hs, interpreted )
  [2 of 2] Compiling Univariate       ( Univariate.lhs, interpreted )
  Ok, modules loaded: Univariate, Polynomials.
  *Univariate> let f x = (895 % 922) + (1080 % 6931)*x + (2323 % 1248)*x^2
  *Univariate> firstDifs (map f [0..20])
  [895 % 922,17448553 % 8649888,2323 % 624]

> {-
> list2firstDifZp' fs = map (recCRT' . map toInteger2) $ wellOrd $ map helper bigPrimes
>   where
>     helper p = zip (firstDifsp p (accessibleData' fs p)) (repeat p)

  *FROverZp> let f x = (895 % 922) + (1080 % 6931)*x + (2323 % 1248)*x^2
  *FROverZp> let fs = map f [0..]
  *FROverZp> list2firstDifZp' fs
  [(895 % 922,805479325081)
  ,(17448553 % 8649888,722916888780872419)
  ,(2323 % 624,805479325081)
  ]
  *FROverZp> map fst it
  [895 % 922,17448553 % 8649888,2323 % 624]
```

```
  *FROverZp> newtonC it
  [895 % 922,17448553 % 8649888,2323 % 1248]
  *FROverZp> npol2pol it
  [895 % 922,1080 % 6931,2323 % 1248]

> list2polZp :: [Ratio Int] -> [Ratio Integer]
> list2polZp = npol2pol . newtonC . (map fst) . list2firstDifZp'

--
Univariate Rational function case
Since thiele2coef uses only (*), (+) and (-) operations, we don't have to do these calculation over prime fields.
So, our target should be rho function (matrix?) calculation.

Reciprocal difference

> rhoZp :: Integral a => [Ratio a] -> a -> Int -> a -> a
> rhoZp fs 0 i p = (fs !! i) `modp` p
> rhoZp fs n i p
>   | n <= 0      = 0
>   | otherwise = (n*inv + rhoZp fs (n-2) (i+1) p) `mod` p
>   where
>     inv = fromJust inv'
>     inv'= (rhoZp fs (n-1) (i+1) p - rhoZp fs (n-1) i p) `inversep` p
>
> aZp :: Integral a => [Ratio a] -> a -> a -> a
> aZp fs 0 p = head fs `modp` p
> aZp fs n p = (rhoZp fs n 0 p - rhoZp fs (n-2) 0 p) `mod` p
>
> tDegreeZp fs p = helper fs 0 p
>   where
>     helper fs n p
>       | isConst fs' = n
>       | otherwise   = helper fs (n+1) p
>       where
>         fs' = map (rhoZp fs n p) [0..]
>     isConst (i:j:_) = i==j

  *FROverZp> let h t = (3+6*t+18*t^2)%(1+2*t+20*t^2)
```

```
  *FROverZp> let hs = map h [0..]
  *FROverZp> take 5 $ map (\n -> rhoZp hs 0 n 101) [0..]
  [3,89,64,8,16]
  *FROverZp> take 5 $ map (\n -> rhoZp hs 1 n 101) [0..]
  [74,4,9,38,65]
  *FROverZp> take 5 $ map (\n -> rhoZp hs 2 n 101) [0..]
  [*** Exception: Maybe.fromJust: Nothing

> -}
```