Finite fields and functional reconstructions

Ray D. Sameshima

2016/09/23 $\sim$ 2017/03/30 13:11

Contents

0 Preface **7**
- 0.1 References . . . 7
- 0.2 Set theoretical gadgets . . . 7
  - 0.2.1 Numbers . . . 7
  - 0.2.2 Algebraic structures . . . 8
- 0.3 Haskell language . . . 8

1 Basics **11**
- 1.1 Finite fields . . . 11
  - 1.1.1 Rings . . . 11
  - 1.1.2 Fields . . . 12
  - 1.1.3 An example of finite rings $\mathbb{Z}_n$ . . . 12
  - 1.1.4 Bézout's lemma . . . 13
  - 1.1.5 Greatest common divisor . . . 13
  - 1.1.6 Extended Euclidean algorithm . . . 15
  - 1.1.7 Coprime as a binary relation . . . 18
  - 1.1.8 Corollary (Inverses in $\mathbb{Z}_n$) . . . 18
  - 1.1.9 Corollary (Finite field $\mathbb{Z}_p$) . . . 19
- 1.2 Rational number reconstruction . . . 21
  - 1.2.1 A map from $\mathbb{Q}$ to $\mathbb{Z}_p$ . . . 21
  - 1.2.2 Reconstruction from $\mathbb{Z}_p$ to $\mathbb{Q}$ . . . 23
  - 1.2.3 Chinese remainder theorem . . . 27
  - 1.2.4 `recCRT`: from image in $\mathbb{Z}_p$ to rational number . . . 30
- 1.3 Polynomials and rational functions . . . 31
  - 1.3.1 Notations . . . 31
  - 1.3.2 Polynomials and rational functions . . . 32
  - 1.3.3 As data, coefficients list . . . 33
- 1.4 Haskell implementation of univariate polynomials . . . 33
  - 1.4.1 A polynomial as a list of coefficients . . . 33

1.4.2 Difference analysis 37

2 Functional reconstruction over $\mathbb{Q}$ 41
2.1 Univariate polynomials 41
2.1.1 Newtons' polynomial representation 41
2.1.2 Towards canonical representations 42
2.1.3 Simplification of our problem 43
2.2 Univariate polynomial reconstruction with Haskell 45
2.2.1 Newton interpolation formula with Haskell 45
2.2.2 Stirling numbers of the first kind 46
2.2.3 `list2pol`: from output list to canonical coefficients . 48
2.3 Univariate rational functions 49
2.3.1 Thiele's interpolation formula 50
2.3.2 Towards canonical representations 51
2.4 Univariate rational function reconstruction with Haskell . . . 51
2.4.1 Reciprocal difference 51
2.4.2 `tDegree` for termination 52
2.4.3 `thieleC`: from output list to Thiele coefficients 54
2.4.4 Haskell representation for rational functions 55
2.4.5 `list2rat`: from output list to canonical coefficients . 59
2.5 Multivariate polynomials 59
2.5.1 Foldings as recursive applications 60
2.5.2 Experiments, 2 variables case 60
2.5.3 Haskell implementation, 2 variables case 63
2.6 Multivariate rational functions 65
2.6.1 The canonical normalization 65
2.6.2 An auxiliary t 66
2.6.3 Experiments, 2 variables case 66
2.6.4 Haskell implementation, 2 variables case 69

3 Functional reconstruction over finite fields 71
3.1 Univariate polynomials 71
3.1.1 Pre-cook 71
3.1.2 Difference analysis on $\mathbb{Z}_p$ 72
3.1.3 Eager and lazy degree 73
3.1.4 Term by term reconstruction 74
3.1.5 `list2polZp`: from the output list to coefficient lists . 75
3.2 TBA Univariate rational functions 76

4 Codes **77**
4.1 `Ffield.lhs` 77
4.2 `Polynomials.hs` 82
4.3 `Univariate.lhs` 84
4.4 `Multivariate.lhs` 97
4.5 `FROverZp.lhs` 107

Chapter 0

Preface

0.1 References

1. Scattering amplitudes over finite fields and multivariate functional reconstruction
(Tiziano Peraro)
`https://arxiv.org/pdf/1608.01902.pdf`

2. Haskell Language
`www.haskell.org`

3. The Haskell Road to Logic, Maths and Programming
(Kees Doets, Jan van Eijck)
`http://homepages.cwi.nl/~jve/HR/`

4. Introduction to numerical analysis
(Stoer Josef, Bulirsch Roland)

0.2 Set theoretical gadgets

0.2.1 Numbers

Here is a list of what we assumed that the readers are familiar with:

1. $\mathbb{N}$ (Peano axiom: $\emptyset, \mathrm{suc}$)

2. $\mathbb{Z}$

3. $\mathbb{Q}$

4. $\mathbb{R}$ (Dedekind cut)
5. $\mathbb{C}$

0.2.2 Algebraic structures

1. Monoid: $(\mathbb{N}, +), (\mathbb{N}, \times)$
2. Group: $(\mathbb{Z}, +), (\mathbb{Z}, \times)$
3. Ring: $\mathbb{Z}$
4. Field: $\mathbb{Q}$, $\mathbb{R}$ (continuous), $\mathbb{C}$ (algebraic closed)

0.3 Haskell language

From "A Brief, Incomplete and Mostly Wrong History of Programming Languages":[1]

> 1990 - A committee formed by Simon Peyton-Jones, Paul Hudak, Philip Wadler, Ashton Kutcher, and People for the Ethical Treatment of Animals creates Haskell, a pure, non-strict, functional language. Haskell gets some resistance due to the complexity of using monads to control side effects. Wadler tries to appease critics by explaining that "a monad is a monoid in the category of endofunc[illegible] w[illegible]s the problem?"

Figure 1: Haskell's logo, the combinations of λ and monad's bind `>>=`.

Haskell language is a standardized purely functional declarative statically typed programming language.

In declarative languages, we describe "what" or "definition" in its codes, however imperative languages, like C/C++, "how" or "procedure".

[1] `http://james-iry.blogspot.com/2009/05/brief-incomplete-and-mostly-wrong.html`

> Functional languages can be seen as 'executable mathematics'; the notation was designed to be as close as possible to the mathematical way of writing.[2]

Instead of loops, we use (implicit) recursions in functional language.[3]

```
> sum :: [Int] -> Int
> sum []     = 0
> sum (i:is) = i + sum is
```

[2] Algorithms: A Functional Programming Approach (Fethi A. Rabhi, Guy Lapalme)

[3] Of course, as a best practice, we should use higher order function (in this case `foldr` or `foldl`) rather than explicit recursions.

Chapter 1

Basics

We have assumed living knowledge on (axiomatic, i.e., ZFC) set theory, algebraic structures.

1.1 Finite fields

```
Ffield.lhs

https://arxiv.org/pdf/1608.01902.pdf

> module Ffield where

> import Data.Ratio
> import Data.Maybe
> import Data.Numbers.Primes
```

1.1.1 Rings

A ring $(R, +, *)$ is a structured set R with two binary operations

```
(+) :: R -> R -> R                    (1.1)
(*) :: R -> R -> R                    (1.2)
```

satisfying the following 3 (ring) axioms:

1. $(R, +)$ is an abelian, i.e., commutative group, i.e.,

$$\forall a, b, c \in R, (a + b) + c = a + (b + c) \quad \text{(associativity for } +) \tag{1.3}$$
$$\forall a, b, \in R, a + b = b + a \quad \text{(commutativity)} \tag{1.4}$$
$$\exists 0 \in R, \text{ s.t. } \forall a \in R, a + 0 = a \quad \text{(additive identity)} \tag{1.5}$$
$$\forall a \in R, \exists (-a) \in R \text{ s.t. } a + (-a) = 0 \quad \text{(additive inverse)} \tag{1.6}$$

2. $(R, *)$ is a monoid, i.e.,

$$\forall a, b, c \in R, (a * b) * c = a * (b * c) \quad \text{(associativity for } *) \tag{1.7}$$
$$\exists 1 \in R, \text{ s.t. } \forall a \in R, a * 1 = a = 1 * a \quad \text{(multiplicative identity)} \tag{1.8}$$

3. Multiplication is distributive w.r.t addition, i.e., $\forall a, b, c \in R$,

$$a * (b + c) = (a * b) + (a * c) \quad \text{(left distributivity)} \tag{1.9}$$
$$(a + b) * c = (a * c) + (b * c) \quad \text{(right distributivity)} \tag{1.10}$$

1.1.2 Fields

A field is a ring $(\mathbb{K}, +, *)$ whose non-zero elements form an abelian group under multiplication, i.e., $\forall r \in \mathbb{K}$,

$$r \neq 0 \Rightarrow \exists r^{-1} \in \mathbb{K} \text{ s.t. } r * r^{-1} = 1 = r^{-1} * r. \tag{1.11}$$

A field $\mathbb{K}$ is a finite field iff the underlying set $\mathbb{K}$ is finite. A field $\mathbb{K}$ is called infinite field iff the underlying set is infinite.

1.1.3 An example of finite rings $\mathbb{Z}_n$

Let $n(> 0) \in \mathbb{N}$ be a non-zero natural number. Then the quotient set

$$\mathbb{Z}_n := \mathbb{Z}/n\mathbb{Z} \tag{1.12}$$
$$\cong \{0, \cdots, (n-1)\} \tag{1.13}$$

with addition, subtraction and multiplication under modulo n is a ring.[1]

[1] Here we have taken an equivalence class,

$$0 \leq \forall k \leq (n-1), [k] := \{k + n * z | z \in \mathbb{Z}\} \tag{1.14}$$

1.1.4 Bézout's lemma

Consider $a, b \in \mathbb{Z}$ be nonzero integers. Then there exist $x, y \in \mathbb{Z}$ s.t.

$$a * x + b * y = \gcd(a, b), \tag{1.19}$$

where gcd is the greatest common divisor (function), see §1.1.5. We will prove this statement in §1.1.6.

1.1.5 Greatest common divisor

Before the proof, here is an implementation of gcd using Euclidean algorithm with Haskell language:

```
> -- Eucledian algorithm.
> myGCD :: Integral a => a -> a -> a
> myGCD a b
>   | b < 0 = myGCD a (-b)
> myGCD a b
>   | a == b = a
>   | b >  a = myGCD b a
>   | b <  a = myGCD (a-b) b
```

Example, by hands

Let us consider the gcd of 7 and 13. Since they are primes, the gcd should be 1. First it binds `a` with 7 and `b` with 13, and hit `b > a`.

```
myGCD 7 13 == myGCD 13 7
```
(1.20)

Then it hits main line:

```
myGCD 13 7 == myGCD (13-7) 7
```
(1.21)

---

with the following operations:

$$[k] + [l] := [k + l] \tag{1.15}$$
$$[k] * [l] := [k * l] \tag{1.16}$$

This is equivalent to take modular n:

$$(k \mod n) + (l \mod n) := (k + l \mod n) \tag{1.17}$$
$$(k \mod n) * (l \mod n) := (k * l \mod n). \tag{1.18}$$

In order to go to next step, Haskell evaluate $(13-7)$,[2] and

```
myGCD (13-7) 7  ==  myGCD 6 7          (1.22)
                ==  myGCD 7 6          (1.23)
                ==  myGCD (7-6) 6      (1.24)
                ==  myGCD 1 6          (1.25)
                ==  myGCD 6 1          (1.26)
```

Finally it ends with 1:

```
myGCD 1 1 == 1                         (1.27)
```

As another example, consider 15 and 25:

```
myGCD 15 25  ==  myGCD 25 15           (1.28)
             ==  myGCD (25-15) 15      (1.29)
             ==  myGCD 10 15           (1.30)
             ==  myGCD 15 10           (1.31)
             ==  myGCD (15-10) 10      (1.32)
             ==  myGCD 5 10            (1.33)
             ==  myGCD 10 5            (1.34)
             ==  myGCD (10-5) 5        (1.35)
             ==  myGCD 5 5             (1.36)
             ==  5                     (1.37)
```

Example, with Haskell

Let us check simple example using Haskell:

```
*Ffield> myGCD 7 13
1
*Ffield> myGCD 7 14
7
*Ffield> myGCD (-15) (20)
5
*Ffield> myGCD (-299) (-13)
13
```

[2] Since Haskell language adopts lazy evaluation, i.e., call by need, not call by name.

The final result is from

```
*Ffield> 13*23
299
```

1.1.6 Extended Euclidean algorithm

Here we treat the extended Euclidean algorithm, this is a constructive solution for Bézout's lemma.

As intermediate steps, this algorithm makes sequences of integers $\{r_i\}_i$, $\{s_i\}_i$, $\{t_i\}_i$ and quotients $\{q_i\}_i$ as follows. The base cases are

$$(r_0, s_0, t_0) := (a, 1, 0) \tag{1.38}$$
$$(r_1, s_1, t_1) := (b, 0, 1) \tag{1.39}$$

and inductively, for $i \geq 2$,

$$q_i := \text{quot}(r_{i-2}, r_{i-1}) \tag{1.40}$$
$$r_i := r_{i-2} - q_i * r_{i-1} \tag{1.41}$$
$$s_i := s_{i-2} - q_i * s_{i-1} \tag{1.42}$$
$$t_i := t_{i-2} - q_i * t_{i-1}. \tag{1.43}$$

The termination condition[3] is

$$r_k = 0 \tag{1.44}$$

for some $k \in \mathbb{N}$ and

$$\gcd(a, b) = r_{k-1} \tag{1.45}$$
$$x = s_{k-1} \tag{1.46}$$
$$y = t_{k-1}. \tag{1.47}$$

Proof

By definition,

$$\gcd(r_{i-1}, r_i) = \gcd(r_{i-1}, r_{i-2} - q_i * r_{i-1}) \tag{1.48}$$
$$= \gcd(r_{i-1}, r_{i-2}) \tag{1.49}$$

[3] This algorithm will terminate eventually, since the sequence $\{r_i\}_i$ is non-negative by definition of q_i, but strictly decreasing. Therefore, $\{r_i\}_i$ will meet 0 in finite step k.

and this implies

$$\gcd(a,b) =: \gcd(r_0, r_1) = \cdots = \gcd(r_{k-1}, 0), \tag{1.50}$$

i.e.,

$$r_{k-1} = \gcd(a,b). \tag{1.51}$$

Next, for $i = 0, 1$ observe

$$a * s_i + b * t_i = r_i. \tag{1.52}$$

Let $i \geq 2$, then

$$\begin{aligned} r_i &= r_{i-2} - q_i * r_{i-1} && (1.53)\\ &= a * s_{i-2} + b * t_{i-2} - q_i * (a * s_{i-1} + b * t_{i-1}) && (1.54)\\ &= a * (s_{i-2} - q_i * s_{i-1}) + b * (t_{i-2} - q_i * t_{i-1}) && (1.55)\\ &=: a * s_i + b * t_i. && (1.56) \end{aligned}$$

Therefore, inductively we get

$$\gcd(a,b) = r_{k-1} = a * s_{k-1} + b * t_{k-1}. =: a * x + b * y. \tag{1.57}$$

This prove Bézout's lemma.
■

Haskell implementation

Here I use lazy lists for intermediate lists of qs, rs, ss, ts, and pick up (second) last elements for the results.

```
Here we would like to implement the extended Euclidean algorithm.
See the algorithm, examples, and pseudo code at:

  https://en.wikipedia.org/wiki/Extended_Euclidean_algorithm

> exGCD' :: (Integral n) => n -> n -> ([n], [n], [n], [n])
> exGCD' a b = (qs, rs, ss, ts)
>   where
>     qs = zipWith quot rs (tail rs)
>     rs = takeUntil (==0) r'
>     r' = steps a b
```

```
>     ss = steps 1 0
>     ts = steps 0 1
>     steps a b = rr
>       where
>         rr@(_:rs) = a:b: zipWith (-) rr (zipWith (*) qs rs)
>
> takeUntil :: (a -> Bool) -> [a] -> [a]
> takeUntil p = foldr func []
>   where
>     func x xs
>       | p x = []
>       | otherwise = x : xs
```

Here we have used so called lazy lists, and higher order function[4]. The gcd of a and b should be the last element of second list `rs`, and our targets (s, t) are second last elements of last two lists `ss` and `ts`. The following example is from wikipedia:

```
*Ffield> exGCD' 240 46
([5,4,1,1,2],[240,46,10,6,4,2],[1,0,1,-4,5,-9,23],[0,1,-5,21,-26,47,-120])
```

Look at the second lasts of `[1,0,1,-4,5,-9,23]`,`[0,1,-5,21,-26,47,-120]`, i.e., -9 and 47:

```
*Ffield> gcd 240 46
2
*Ffield> 240*(-9) + 46*(47)
2
```

It works, and we have other simpler examples:

```
*Ffield> exGCD' 15 25
([0,1,1,2],[15,25,15,10,5],[1,0,1,-1,2,-5],[0,1,0,1,-1,3])
*Ffield> 15 * 2 + 25*(-1)
5
*Ffield> exGCD' 15 26
([0,1,1,2,1,3],[15,26,15,11,4,3,1],[1,0,1,-1,2,-5,7,-26],[0,1,0,1,-1,3,-4,15])
*Ffield> 15*7 + (-4)*26
1
```

[4] Naively speaking, the function whose inputs and/or outputs are functions is called a higher order function.

Now what we should do is extract gcd of a and b, and (x, y) from the tuple of lists:

```
> -- a*x + b*y = gcd a b
> exGCD :: Integral t => t -> t -> (t, t, t)
> exGCD a b = (g, x, y)
>   where
>     (_,r,s,t) = exGCD' a b
>     g = last r
>     x = last . init $ s
>     y = last . init $ t
```

where the underscore _ is a special symbol in Haskell that hits every pattern, since we do not need the quotient list. So, in order to get gcd and (x, y) we don't need quotients list.

```
*Ffield> exGCD 46 240
(2,47,-9)
*Ffield> 46*47 + 240*(-9)
2
*Ffield> gcd 46 240
2
```

1.1.7 Coprime as a binary relation

Let us define a binary relation as follows:

```
coprime :: Integral a => a -> a -> Bool
coprime a b = (gcd a b) == 1
```

1.1.8 Corollary (Inverses in $\mathbb{Z}_n$)

For a non-zero element

$$a \in \mathbb{Z}_n, \tag{1.58}$$

there is a unique number

$$b \in \mathbb{Z}_n \text{ s.t. } ((a * b) \mod n) = 1 \tag{1.59}$$

iff a and n are coprime.

Proof

From Bézout's lemma, a and n are coprime iff

$$\exists s, t \in \mathbb{Z}, a * s + n * t = 1. \tag{1.60}$$

Therefore

$$\begin{aligned} a \text{ and } n \text{ are coprime} &\Leftrightarrow \exists s, t \in \mathbb{Z}, a * s + n * t = 1 && (1.61)\\ &\Leftrightarrow \exists s, t' \in \mathbb{Z}, a * s = 1 + n * t'. && (1.62) \end{aligned}$$

This s, by taking its modulo n is our $b = a^{-1}$:

$$a * s = 1 \mod n. \tag{1.63}$$

We will make a Haskell implementation in §1.1.9.
■

1.1.9 Corollary (Finite field $\mathbb{Z}_p$)

If p is prime, then

$$\mathbb{Z}_p := \{0, \cdots, (p-1)\} \tag{1.64}$$

with addition, subtraction and multiplication under modulo n is a field.

Proof

It suffices to show that

$$\forall a \in \mathbb{Z}_p, a \neq 0 \Rightarrow \exists a^{-1} \in \mathbb{K} \text{ s.t. } a * a^{-1} = 1 = a^{-1} * a, \tag{1.65}$$

but since p is prime, and

$$\forall a \in \mathbb{Z}_p, a \neq 0 \Rightarrow \texttt{gcd a p == 1} \tag{1.66}$$

so all non-zero element has its inverse in $\mathbb{Z}_p$.
■

Example and implementation

Let us pick 11 as a prime and consider $\mathbb{Z}_{11}$:

```
Example Z_{11}

*Ffield> isField 11
True
*Ffield> map (exGCD 11) [0..10]
[(11,1,0),(1,0,1),(1,1,-5),(1,-1,4),(1,-1,3)
,(1,1,-2),(1,-1,2),(1,2,-3),(1,3,-4),(1,-4,5),(1,1,-1)
]
```

This list of three-tuple let us know the candidate of inverse. Take the last one, `(1,1,-1)`. This is the image of `exGcd 11 10`, and

$$1 = 10 * 1 + 11 * (-1) \tag{1.67}$$

holds. This suggests -1 is a candidate of the inverse of 10 in $\mathbb{Z}_{11}$:

$$\begin{aligned} 10^{-1} &= -1 \mod 11 && (1.68)\\ &= 10 \mod 11 && (1.69) \end{aligned}$$

In fact,

$$10 * 10 = 11 * 9 + 1. \tag{1.70}$$

So, picking up the third elements in tuple and zipping with nonzero elements, we have a list of inverses:

```
*Ffield> map ((`mod` 11) . (\(_,_,x)->x) . exGCD 11) [1..10]
[1,6,4,3,9,2,8,7,5,10]

We get non-zero elements with its inverse:

*Ffield> zip [1..10] it
[(1,1),(2,6),(3,4),(4,3),(5,9),(6,2),(7,8),(8,7),(9,5),(10,10)]
```

Let us generalize these flow into a function[5]:

[5] From `https://hackage.haskell.org/package/base-4.9.0.0/docs/Data-Maybe.html`:

> The Maybe type encapsulates an optional value. A value of type Maybe a either contains a value of type a (represented as Just a), or it is empty (represented as Nothing). Using Maybe is a good way to deal with errors or exceptional cases without resorting to drastic measures such as error.

```
> -- a^{-1} (in Z_p) == a `inversep` p
> inversep :: Integral a => a -> a -> Maybe a
> a `inversep` p = let (g,x,_) = exGCD a p in
>   if (g == 1) then Just (x `mod` p)
>               else Nothing
```

This `inversep` function returns the inverse with respect to second argument, if they are coprime, i.e. gcd is 1. So the second argument should not be prime.

```
> inversesp :: Integral a => a -> [Maybe a]
> inversesp p = map (`inversep` p) [1..(p-1)]
```

```
*Ffield> inversesp 11
[Just 1,Just 6,Just 4,Just 3,Just 9,Just 2,Just 8,Just 7,Just 5,Just 10]
*Ffield> inversesp 9
[Just 1,Just 5,Nothing,Just 7,Just 2,Nothing,Just 4,Just 8]
```

1.2 Rational number reconstruction

1.2.1 A map from $\mathbb{Q}$ to $\mathbb{Z}_p$

Let p be a prime. Now we have a map

$$- \mod p : \mathbb{Z} \to \mathbb{Z}_p; a \mapsto (a \mod p), \tag{1.71}$$

and a natural inclusion (or a forgetful map)[6]

$$¿ : \mathbb{Z}_p \hookrightarrow \mathbb{Z}. \tag{1.73}$$

Then we can define a map

$$- \mod p : \mathbb{Q} \to \mathbb{Z}_p \tag{1.74}$$

by[7]

$$q = \frac{a}{b} \mapsto (q \mod p) := \left(\left(a \times ¿\left(b^{-1} \mod p\right)\right) \mod p\right). \tag{1.75}$$

[6] By introducing this forgetful map, we can use

$$\times : (\mathbb{Z}, \mathbb{Z}) \to \mathbb{Z} \tag{1.72}$$

of normal product on $\mathbb{Z}$.

[7] This is an example of operator overloadings.

Example and implementation

An easy implementation is the followings:[8]

```
A map from Q to Z_p.

> -- p should be prime.
> modp :: Integral a => Ratio a -> a -> a
> q `modp` p = (a * (bi `mod` p)) `mod` p
>   where
>     (a,b) = (numerator q, denominator q)
>     bi = fromJust (b `inversep` p)
```

Let us consider a rational number $\frac{3}{7}$ on a finite field $\mathbb{Z}_{11}$:

```
Example: on Z_{11}
Consider (3 % 7).

  *Ffield> let q = 3%7
  *Ffield> 3 `mod` 11
  3
  *Ffield> 7 `inversep` 11
  Just 8
  *Ffield> (3*8) `mod` 11
  2
```

For example, pick 7:

```
  *Ffield> 7*8 == 11*5+1
  True
```

Therefore, on $\mathbb{Z}_{11}$, $(7^{-1} \mod 11)$ is equal to $(8 \mod 11)$ and

$$\frac{3}{7} \in \mathbb{Q} \mapsto (3 \times ¿(7^{-1} \mod 11) \mod 11) \tag{1.78}$$
$$= (3 \times 8) \mod 11 \tag{1.79}$$
$$= 24 \mod 11 \tag{1.80}$$
$$= 2 \mod 11. \tag{1.81}$$

[8] The backquotes makes any binary function infix operator. For example,

```
add 1 2 == 1 `add` 2
```
(1.76)

Similarly, use parenthesis we can use an infix binary operator to a function:

```
(+) 1 2 == 1 + 2
```
(1.77)

Haskell returns the same result

```
*Ffield> q `modp` 11
2
```

and consistent.

1.2.2 Reconstruction from $\mathbb{Z}_p$ to $\mathbb{Q}$

Consider a rational number q and its image $a \in \mathbb{Z}_p$.

$$a := q \mod p \tag{1.82}$$

The extended Euclidean algorithm can be used for guessing a rational number q from the images $a := q \mod p$ of several primes p's.

At each step, the extended Euclidean algorithm satisfies eq.(1.52).

$$a * s_i + p * t_i = r_i \tag{1.83}$$

Therefore

$$r_i = a * s_i \mod p. \tag{1.84}$$

Hence $\frac{r_i}{s_i}$ is a possible guess for q. We take

$$r_i^2, s_i^2 < p \tag{1.85}$$

as the termination condition for this reconstruction.

Haskell implementation

Let us first try to reconstruct from the image ($\frac{1}{3} \mod p$) of some prime p. Here we have chosen three primes

```
Reconstruction Z_p -> Q
  *Ffield> let q = (1%3)
  *Ffield> take 3 $ dropWhile (<100) primes
  [101,103,107]
```

The images are basically given by the first elements of second lists (s_0's):

```
*Ffield> q `modp` 101
34
*Ffield> let try x = exGCD' (q `modp` x) x
*Ffield> try 101
([0,2,1,33],[34,101,34,33,1],[1,0,1,-2,3,-101],[0,1,0,1,-1,34])
*Ffield> try 103
([0,1,2,34],[69,103,69,34,1],[1,0,1,-1,3,-103],[0,1,0,1,-2,69])
*Ffield> try 107
([0,2,1,35],[36,107,36,35,1],[1,0,1,-2,3,-107],[0,1,0,1,-1,36])
```

Look at the first hit of termination condition eq.(1.85), $r_4 = 1$ and $s_4 = 3$. They give us the same guess $\frac{1}{3}$, and that the reconstructed number.

From the above observations we can make a simple "guess" function:

```
> guess :: Integral t =>
>          (t, t)       -- (q `modp` p, p)
>       -> (Ratio t, t)
> guess (a, p) = let (_,rs,ss,_) = exGCD' a p in
>   (select rs ss p, p)
>     where
>       select :: Integral t => [t] -> [t] -> t -> Ratio t
>       select [] _ _ = 0%1
>       select (r:rs) (s:ss) p
>         | s /= 0 && r^2 <= p && s^2 <= p = r%s
>         | otherwise = select rs ss p
```

We have put a list of big primes as follows.

```
> -- Hard code of big primes
> -- For chinese reminder theorem we declare it as [Integer].
> bigPrimes :: [Integer]
> bigPrimes = dropWhile (< 897473) $ takeWhile (< 978948) primes
```

We choose 3 times match as the termination condition.

```
> matches3 :: Eq a => [a] -> a
> matches3 (a:bb@(b:c:cs))
>   | a == b && b == c = a
>   | otherwise        = matches3 bb
```

Finally, we can check our gadgets.

What we know is a list of (q `modp` p) and prime p for several (big) primes.

```
*Ffield> let q = 10%19
*Ffield> let knownData = zip (map (modp q) bigPrimes) bigPrimes
*Ffield> take 3 knownData
[(614061,897473),(377894,897497),(566842,897499)]
*Ffield> matches3 $  map (fst . guess) knownData
10 % 19
```

The following is the function we need, its input is the list of tuple which first element is the image in $\mathbb{Z}_p$ and second element is that prime p.

```
> reconstruct :: Integral a =>
>                [(a, a)]  -- :: [(Z_p, primes)]
>             -> Ratio a
> reconstruct aps = matches3 $ map (fst . guess) aps
```

```
Here is a naive test:
  > let qs = [1 % 3,10 % 19,41 % 17,30 % 311,311 % 32
             ,869 % 232,778 % 123,331 % 739]
  > let modmap q = zip (map (modp q) bigPrimes) bigPrimes
  > let longList = map modmap qs
  > map reconstruct longList
  [1 % 3,10 % 19,41 % 17,30 % 311,311 % 32
  ,869 % 232,778 % 123,331 % 739]
  > it == qs
  True
```

For later use, let us define

```
> imagesAndPrimes ::  Rational-> [(Integer, Integer)]
> imagesAndPrimes q = zip (map (modp q) bigPrimes) bigPrimes
```

to generate a list of images (of our target rational number) in Z_p and the base primes.

As another example, we have slightly involved function:

```
> matches3' :: Eq a => [(a, t)] -> (a, t)
> matches3' (a0@(a,_):bb@((b,_):(c,_):cs))
>   | a == b && b == c = a0
>   | otherwise        = matches3' bb
```

Let us see the first good guess, Haskell tells us that in order to reconstruct, say $\frac{331}{739}$, we should take three primes start from 614693:

```
*Ffield> let knowData q = zip (map (modp q) primes) primes
*Ffield> matches3' $ map guess $ knowData (331%739)
(331 % 739,614693)
(18.31 secs, 12,393,394,032 bytes)

*Ffield> matches3' $ map guess $ knowData (11%13)
(11 % 13,311)
(0.02 secs, 2,319,136 bytes)
*Ffield> matches3' $ map guess $ knowData (1%13)
(1 % 13,191)
(0.01 secs, 1,443,704 bytes)
*Ffield> matches3' $ map guess $ knowData (1%3)
(1 % 3,13)
(0.01 secs, 268,592 bytes)
*Ffield> matches3' $ map guess $ knowData (11%31)
(11 % 31,1129)
(0.03 secs, 8,516,568 bytes)
*Ffield> matches3' $ map guess $ knowData (12%312)
(1 % 26,709)
```

A problem

Since our choice of `bigPrimes` are order 10^6, our reconstruction can fail for rational numbers of

$$\frac{O(10^3)}{O(10^3)}, \tag{1.86}$$

say

```
*Ffield> let q = 895%922
*Ffield> let knownData = imagesAndPrimes q
*Ffield> take 4 knownData
[(882873,897473)
,(365035,897497)
,(705735,897499)
,(511060,897517)
]
*Ffield> map guess it
[((-854) % 123,897473)
,((-656) % 327,897497)
,((-192) % 805,897499)
```

```
 ,((-491) % 497,897517)
 ]
```

We can solve this by introducing the following theorem.

1.2.3 Chinese remainder theorem

From wikipedia[9]

> There are certain things whose number is unknown. If we count them by threes, we have two left over; by fives, we have three left over; and by sevens, two are left over. How many things are there?

Here is a solution with Haskell:

```
*Ffield> let lst = [n|n<-[0..], mod n 3==2, mod n 5==3, mod n 7==2]
*Ffield> head lst
23
```

We define an infinite list of natural numbers that satisfy

$$n \mod 3 = 2, n \mod 5 = 3, n \mod 7 = 2. \tag{1.87}$$

Then take the first element, and this is the answer.

Claim

The statement for binary case is the following. Let $n_1, n_2 \in \mathbb{Z}$ be coprime, then for arbitrary $a_1, a_2 \in \mathbb{Z}$, the following a system of equations

$$x = a_1 \mod n_1 \tag{1.88}$$
$$x = a_2 \mod n_2 \tag{1.89}$$

have a unique solution modular $n_1 * n_2$[10].

[9] https://en.wikipedia.org/wiki/Chinese_remainder_theorem

[10] Note that, this is equivalent that there is a unique solution a in

$$0 \leq a < n_1 \times n_2. \tag{1.90}$$

Proof

(existence) With §1.1.6, there are $m_1, m_2 \in \mathbb{Z}$ s.t.

$$n_1 * m_1 + n_2 * m_2 = 1. \tag{1.91}$$

Now we have

$$\begin{aligned} n_1 * m_1 &= 1 \mod n_2 && (1.92)\\ n_2 * m_2 &= 1 \mod n_1 && (1.93) \end{aligned}$$

that is[11]

$$\begin{aligned} m_1 &= n_1^{-1} \mod n_2 && (1.94)\\ m_2 &= n_2^{-1} \mod n_1. && (1.95) \end{aligned}$$

Then

$$a := a_1 * n_2 * m_2 + a_2 * n_1 * m_1 \mod (n_1 * n_2) \tag{1.96}$$

is a solution.

(uniqueness) If a' is also a solution, then

$$\begin{aligned} a - a' &= 0 \mod n_1 && (1.97)\\ a - a' &= 0 \mod n_2. && (1.98) \end{aligned}$$

Since n_1 and n_2 are coprime, i.e., no common divisors, this difference is divisible by $n_1 * n_2$, and

$$a - a' = 0 \mod (n_1 * n_2). \tag{1.99}$$

Therefore, the solution is unique modular $n_1 * n_2$.

■

Haskell implementation

Let us see how our naive `guess` function fail one more time:

```
Chinese Remainder Theorem, and its usage

  *Ffield> let q = 895%922
  *Ffield> let knownData = imagesAndPrimes q
```

[11] Here we have used slightly different notions from 1. m_1 in 1 is our m_2 times our n_2.

```
*Ffield> let [(a1,p1),(a2,p2)] = take 2 knownData
*Ffield> take 2 knownData
[(882873,897473),(365035,897497)]
*Ffield> map guess it
[((-854) % 123,897473),((-656) % 327,897497)]
```

It suffices to make a binary version of Chinese Remainder theorem in Haskell:

```
> crtRec' :: Integral t => (t, t) -> (t, t) -> (t, t)
> crtRec' (a1,p1) (a2,p2) = (a,p)
>   where
>     a = (a1*p2*m2 + a2*p1*m1) `mod` p
>     m1 = fromJust (p1 `inversep` p2)
>     m2 = fromJust (p2 `inversep` p1)
>     p = p1*p2
```

`crtRec'` function takes two tuples of image in $\mathbb{Z}_p$ and primes, and returns these combination. Now let us fold.

```
> pile :: (a -> a -> a) -> [a] -> [a]
> pile f [] = []
> pile f dd@(d:ds) = d : zipWith' f (pile f dd) ds
```

Schematically, this `pile f` function takes

$$[d_0, d_1, d_2, d_3, \cdots] \tag{1.100}$$

and returns

$$[d_0, f(d_0, d_1), f(f(d_0, d_1), d_2), f(f(f(d_0, d_1), d_2), d_3), \cdots] \tag{1.101}$$

We have used another higher order function which is slightly modified from standard definition:

```
> -- Strict zipWith, from:
> --   http://d.hatena.ne.jp/kazu-yamamoto/touch/20100624/1277348961
> zipWith' :: (a -> b -> c) -> [a] -> [b] -> [c]
> zipWith' f (a:as) (b:bs) = (x `seq` x) : zipWith' f as bs
>   where x = f a b
> zipWith' _ _      _      = []
```

Let us check our implementation.

```
*Ffield> let q = 895%922
*Ffield> let knownData = imagesAndPrimes q
*Ffield> take 4 knownData
[(882873,897473)
,(365035,897497)
,(705735,897499)
,(511060,897517)
]
*Ffield> pile crtRec' it
[(882873,897473)
,(86488560937,805479325081)
,(397525881357811624,722916888780872419)
,(232931448259966259937614,648830197267942270883623)
]
*Ffield> map guess it
[((-854) % 123,897473)
,(895 % 922,805479325081)
,(895 % 922,722916888780872419)
,(895 % 922,648830197267942270883623)
```

So on a product ring $\mathbb{Z}_{805479325081}$, we get the right answer.

1.2.4 `recCRT`: from image in $\mathbb{Z}_p$ to rational number

From above discussion, here we can define a function which takes a list of images in $\mathbb{Z}_p$ and returns the rational number. What we do is, basically, to take a list of image (of our target rational number) and primes, then applying Chinese Remainder theorem recursively, return several guess of rational number.

```
> recCRT :: Integral a => [(a,a)] -> Ratio a
> recCRT = reconstruct . pile crtRec'

> recCRT' = matches3' . map guess . pile crtRec'
```

```
*Ffield> let q = 895%922
*Ffield> let knownData = imagesAndPrimes q
*Ffield> recCRT knownData
895 % 922
*Ffield> recCRT' knownData
(895 % 922,805479325081)
```

Here is some random checks and results.

```
todo: use QuickCheck

> trial = do
>   n <- randomRIO (0,10000) :: IO Integer
>   d <- randomRIO (1,10000) :: IO Integer
>   let q = (n%d)
>   putStrLn $ "input: " ++ show q
>   return $ recCRT' . imagesAndPrimes $ q

  *Ffield> trial
  input: 1080 % 6931
  (1080 % 6931,805479325081)
  *Ffield> trial
  input: 2323 % 1248
  (2323 % 1248,805479325081)
  *Ffield> trial
  input: 6583 % 1528
  (6583 % 1528,805479325081)
  *Ffield> trial
  input: 721 % 423
  (721 % 423,897473)
  *Ffield> trial
  input: 9967 % 7410
  (9967 % 7410,805479325081)
```

1.3 Polynomials and rational functions

The following discussion on an arbitrary field $\mathbb{K}$.

1.3.1 Notations

Let $n \in \mathbb{N}$ be positive. We use multi-index notation:

$$\alpha = (\alpha_1, \cdots, \alpha_n) \in \mathbb{N}^n. \tag{1.102}$$

A monomial is defined as

$$z^\alpha := \prod_i z_i^{\alpha_i}. \tag{1.103}$$

The total degree of this monomial is given by

$$|\alpha| := \sum_i \alpha_i. \tag{1.104}$$

1.3.2 Polynomials and rational functions

Let $\mathbb{K}$ be a field. Consider a map

$$f : \mathbb{K}^n \to \mathbb{K}; z \mapsto f(z) := \sum_\alpha c_\alpha z^\alpha, \tag{1.105}$$

where

$$c_\alpha \in \mathbb{K}. \tag{1.106}$$

We call the value $f(z)$ at the dummy $z \in \mathbb{K}^n$ a polynomial:

$$f(z) := \sum_\alpha c_\alpha z^\alpha. \tag{1.107}$$

We denote

$$\mathbb{K}[z] := \left\{ \sum_\alpha c_\alpha z^\alpha \right\} \tag{1.108}$$

as the ring of all polynomial functions in the variable z with $\mathbb{K}$-coefficients.

Similarly, a rational function can be expressed as a ratio of two polynomials $p(z), q(z) \in \mathbb{K}[z]$:

$$\frac{p(z)}{q(z)} = \frac{\sum_\alpha n_\alpha z^\alpha}{\sum_\beta d_\beta z^\beta}. \tag{1.109}$$

We denote

$$\mathbb{K}(z) := \left\{ \frac{\sum_\alpha n_\alpha z^\alpha}{\sum_\beta d_\beta z^\beta} \right\} \tag{1.110}$$

as the field of rational functions in the variable z with $\mathbb{F}$-coefficients. Similar to fractional numbers, there are several equivalent representation of a rational function, even if we simplify with gcd. However there still is an overall constant ambiguity. To have a unique representation, usually we put the lowest degree of term of the denominator to be 1.

1.3.3 As data, coefficients list

We can identify a polynomial

$$\sum_{\alpha} c_\alpha z^\alpha \tag{1.111}$$

as a set of coefficients

$$\{c_\alpha\}_\alpha. \tag{1.112}$$

Similarly, for a rational function, we can identify

$$\frac{\sum_\alpha n_\alpha z^\alpha}{\sum_\beta d_\beta z^\beta} \tag{1.113}$$

as an ordered pair of coefficients

$$(\{n_\alpha\}_\alpha, \{d_\beta\}_\beta). \tag{1.114}$$

However, there still is an overall factor ambiguity even after gcd simplifications.

1.4 Haskell implementation of univariate polynomials

Here we basically follows some part of §9 of ref.3, and its addendum[12].

`Univariate.lhs`

```
> module Univariate where
> import Data.Ratio
> import Polynomials
```

1.4.1 A polynomial as a list of coefficients

Let us start `instance` declaration, which enable us to use basic arithmetics, e.g., addition and multiplication.

[12] See `http://homepages.cwi.nl/~jve/HR/PolAddendum.pdf`

```
-- Polynomials.hs
-- http://homepages.cwi.nl/~jve/rcrh/Polynomials.hs

module Polynomials where

default (Integer, Rational, Double)

-- polynomials, as coefficients lists
instance (Num a, Ord a) => Num [a] where
  fromInteger c = [fromInteger c]
  -- operator overloading
  negate []     = []
  negate (f:fs) = (negate f) : (negate fs)

  signum [] = []
  signum gs
    | signum (last gs) < (fromInteger 0) = negate z
    | otherwise = z

  abs [] = []
  abs gs
    | signum gs == z = gs
    | otherwise      = negate gs

  fs     + []     = fs
  []     + gs     = gs
  (f:fs) + (g:gs) = f+g : fs+gs

  fs     * []     = []
  []     * gs     = []
  (f:fs) * gg@(g:gs) = f*g : (f .* gs + fs * gg)

delta :: (Num a, Ord a) => [a] -> [a]
delta = ([1,-1] *)

shift :: [a] -> [a]
shift = tail

p2fct :: Num a => [a] -> a -> a
p2fct [] x = 0
```

```
p2fct (a:as) x = a + (x * p2fct as x)

comp :: (Eq a, Num a, Ord a) => [a] -> [a] -> [a]
comp _     []      = error ".."
comp []     _      = []
comp (f:fs) g0@(0:gs) = f : gs * (comp fs g0)
comp (f:fs) gg@(g:gs) = ([f] + [g] * (comp fs gg))
                      + (0 : gs * (comp fs gg))

deriv :: Num a => [a] -> [a]
deriv []     = []
deriv (f:fs) = deriv1 fs 1
  where
    deriv1 []     _ = []
    deriv1 (g:gs) n = n*g : deriv1 gs (n+1)
```

Note that the above operators are overloaded, say `(*)`, `f*g` is a multiplication of two numbers but `fs*gg` is a multiplication of two list of coefficients. We can not extend this overloading to scalar multiplication, since Haskell type system takes the operands of `(*)` the same type

```
(*) :: Num a => a -> a -> a
```
(1.115)

```
> -- scalar multiplication
> infixl 7 .*
> (.*) :: Num a => a -> [a] -> [a]
> c .* []     = []
> c .* (f:fs) = c*f : c .* fs
```

Let us see few examples. If we take a scalar multiplication, say

$$3 * \left(1 + 2z + 3z^2 + 4z^3\right) \tag{1.116}$$

the result should be

$$3 * \left(1 + 2z + 3z^2 + 4z^3\right) = 3 + 6z + 9z^2 + 12z^3 \tag{1.117}$$

In Haskell

```
*Univariate> 3 .* [1,2,3,4]
[3,6,9,12]
```

and this is exactly same as map with section:

```
*Univariate> map (3*) [1,2,3,4]
[3,6,9,12]
```

When we multiply two polynomials, say

$$(1+2z)*(3+4z+5z^2+6z^3) \tag{1.118}$$

the result should be

$$\begin{aligned}(1+2z)*(3+4z+5z^2+6z^3) &= 1*(3+4z+5z^2+6z^3)+2z*(3+4z+5z^2+6z^3)\\ &= 3+(4+2*3)z+(5+2*4)z^2+(6+2*5)z^3+2*6z^4\\ &= 3+10z+13z^2+16z^3+12z^4 \end{aligned} \tag{1.119}$$

In Haskell,

```
*Univariate> [1,2] * [3,4,5,6]
[3,10,13,16,12]
```

Now the (dummy) variable is given as

```
> -- z of f(z), variable
> z :: Num a => [a]
> z = [0,1]
```

A polynomial of degree R is given by a finite sum of the following form:

$$f(z) := \sum_{i=0}^{R} c_i z^i. \tag{1.120}$$

Therefore, it is natural to represent $f(z)$ by a list of coefficient $\{c_i\}_i$. Here is the translator from the coefficient list to a polynomial function:

```
> p2fct :: Num a => [a] -> a -> a
> p2fct [] x = 0
> p2fct (a:as) x = a + (x * p2fct as x)
```

This gives us[13]

```
*Univariate> take 10 $ map (p2fct [1,2,3]) [0..]
[1,6,17,34,57,86,121,162,209,262]
*Univariate> take 10 $ map (\n -> 1+2*n+3*n^2) [0..]
[1,6,17,34,57,86,121,162,209,262]
```

[13] Here we have used lambda, or so called anonymous function. From `http://learnyouahaskell.com/higher-order-functions`

To make a lambda, we write a \(because it kind of looks like the greek

1.4.2 Difference analysis

We do not know in general this canonical form of the polynomial, nor the degree. That means, what we can access is the graph of f, i.e., the list of inputs and outputs. Without loss of generality, we can take

```
[0..]
```
(1.123)

as the input data. Usually we take a finite sublist of this, but we assume it is sufficiently long. The outputs should be

```
map f [0..]  = [f 0, f 1 ..]
```
(1.124)

For example

```
*Univariate> take 10 $ map (\n -> n^2+2*n+1) [0..]
[1,4,9,16,25,36,49,64,81,100]
```

Let us consider the difference sequence

$$\Delta(f)(n) := f(n+1) - f(n). \tag{1.125}$$

Its Haskell version is

```
> -- difference analysis
> difs :: (Num a) => [a] -> [a]
> difs [] = []
> difs [_] = []
> difs (i:jj@(j:js)) = j-i : difs jj
```

This gives

```
*Univariate> difs [1,4,9,16,25,36,49,64,81,100]
[3,5,7,9,11,13,15,17,19]
*Univariate> difs [3,5,7,9,11,13,15,17,19]
[2,2,2,2,2,2,2,2]
```

---

letter lambda if you squint hard enough) and then we write the parameters, separated by spaces.

For example,

$$f(x) := x^2 + 1 \tag{1.121}$$
$$f := \lambda x.x^2 + 1 \tag{1.122}$$

are the same definition.

We claim that if $f(z)$ is a polynomial of degree R, then $\Delta(f)(z)$ is a polynomial of degree $R-1$. Since the degree is given, we can write $f(z)$ in canonical form

$$f(n) = \sum_{i=0}^{R} c_i n^i \tag{1.126}$$

and

$$\begin{aligned}
\Delta(f)(n) &:= f(n+1) - f(n) && (1.127)\\
&= \sum_{i=0}^{R} c_i \left\{(n+1)^i - n^i\right\} && (1.128)\\
&= \sum_{i=1}^{R} c_i \left\{(n+1)^i - n^i\right\} && (1.129)\\
&= \sum_{i=1}^{R} c_i \left\{i * n^{i-1} + O(n^{i-2})\right\} && (1.130)\\
&= c_R * R * n^{R-1} + O(n^{R-2}) && (1.131)
\end{aligned}$$

where $O(n^{i-2})$ is some polynomial(s) of degree $i-2$.

This guarantees the following function will terminate in finite steps[14]; `difLists` keeps generating difference lists until the difference get constant.

```
> difLists :: (Eq a, Num a) => [[a]] -> [[a]]
> difLists [] = []
> difLists xx@(xs:xss) =
>   if isConst xs then xx
>                 else difLists $ difs xs : xx
>   where
>     isConst (i:jj@(j:js)) = all (==i) jj
>     isConst _ = error "difLists: lack of data, or not a polynomial"
```

Let us try:

```
*Univariate> difLists [[-12,-11,6,45,112,213,354,541,780,1077]]
[[6,6,6,6,6,6,6]
,[16,22,28,34,40,46,52,58]
,[1,17,39,67,101,141,187,239,297]
,[-12,-11,6,45,112,213,354,541,780,1077]
]
```

[14] If a given lists is generated by a polynomial.

The degree of the polynomial can be computed by difference analysis:

```
> degree' :: (Eq a, Num a) => [a] -> Int
> degree' xs = length (difLists [xs]) -1
```

For example,

```
*Univariate> degree [1,4,9,16,25,36,49,64,81,100]
2
*Univariate> take 10 $ map (\n -> n^2+2*n+1) [0..]
[1,4,9,16,25,36,49,64,81,100]
*Univariate> degree $ take 10 $ map (\n -> n^5+4*n^3+1) [0..]
5
```

Above `degree'` function can only treat finite list, however, the following function can compute the degree of infinite list.

```
> degreeLazy :: (Eq a, Num a) => [a] -> Int
> degreeLazy xs = helper xs 0
>   where
>     helper as@(a:b:c:_) n
>       | a==b && b==c = n
>       | otherwise    = helper (difs as) (n+1)
```

Note that this lazy function only sees the first two elements of the list (of difference). So first take the lazy `degreeLazy` and guess the degree, take sufficient finite sublist of output and apply `degree'`. Here is the hybrid version:

```
> degree :: (Num a, Eq a) => [a] -> Int
> degree xs = let l = degreeLazy xs in
>   degree' $ take (l+2) xs
```

Chapter 2

Functional reconstruction over $\mathbb{Q}$

The goal of a functional reconstruction algorithm is to identify the monomials appearing in their definition and the corresponding coefficients.

From here, we use $\mathbb{Q}$ as our base field, but every algorithm can be computed on any field, e.g., finite field $\mathbb{Z}_p$.

2.1 Univariate polynomials

2.1.1 Newtons' polynomial representation

Consider a univariate polynomial $f(z)$. Given a sequence of distinct values $y_n|_{n\in\mathbb{N}}$, we evaluate the polynomial form $f(z)$ sequentially:

$$
\begin{aligned}
f_0(z) &= a_0 && (2.1)\\
f_1(z) &= a_0 + (z-y_0)a_1 && (2.2)\\
&\vdots\\
f_r(z) &= a_0 + (z-y_0)\left(a_1 + (z-y_1)(\cdots + (z-y_{r-1})a_r\right) && (2.3)\\
&= f_{r-1}(z) + (z-y_0)(z-y_1)\cdots(z-y_{r-1})a_r, && (2.4)
\end{aligned}
$$

where

$$\begin{aligned} a_0 &= f(y_0) && (2.5)\\ a_1 &= \frac{f(y_1)-a_0}{y_1-y_0} && (2.6)\\ &\vdots \\ a_r &= \left(\left((f(y_r)-a_0)\,\frac{1}{y_r-y_0}-a_1\right)\frac{1}{y_r-y_1}-\cdots-a_{r-1}\right)\frac{1}{y_r-y_{r-1}} && (2.7) \end{aligned}$$

It is easy to see that, $f_r(z)$ and the original $f(z)$ match on the given data points, i.e.,

$$f_r(n) = f(n), 0 \leq n \leq r. \tag{2.8}$$

When we have already known the total degree of $f(z)$, say R, then we can terminate this sequential trial:

$$\begin{aligned} f(z) &= f_R(z) && (2.9)\\ &= \sum_{r=0}^{R} a_r \prod_{i=0}^{r-1}(z-y_i). && (2.10) \end{aligned}$$

In practice, a consecutive zero on the sequence a_r can be taken as the termination condition for this algorithm.[1]

2.1.2 Towards canonical representations

Once we get the Newton's representation

$$\sum_{r=0}^{R} a_r \prod_{i=0}^{r-1}(z-y_i) = a_0 + (z-y_0)\,(a_1 + (z-y_1)(\cdots + (z-y_{R-1})a_R) \tag{2.11}$$

as the reconstructed polynomial, it is convenient to convert it into the canonical form:

$$\sum_{r=0}^{R} c_r z^r. \tag{2.12}$$

This conversion only requires addition and multiplication of univariate polynomials. These operations are reasonably cheap, especially on $\mathbb{Z}_p$.

[1] We have not proved, but higher power will be dominant when we take sufficiently big input, so we terminate this sequence when we get a consecutive zero in a_r.

2.1.3 Simplification of our problem

Without loss of generality, we can put

```
[0..]
```
(2.13)

as our input list. We usually take its finite part but we assume it has enough length. Corresponding to above input,

```
map f [0..]  = [f 0, f 1, ..]
```
(2.14)

of `f :: Ratio Int -> Ratio Int` is our output list.

Then we have slightly simpler forms of coefficients:

$$
\begin{aligned}
f_r(z) &:= a_0 + z * (a_1 + (z-1)\,(a_2 + (z-2)\,(a_3 + \cdots + (z-r+1)a_r))) && (2.15)\\
a_0 &= f(0) && (2.16)\\
a_1 &= f(y_1) - a_0 && (2.17)\\
&= f(1) - f(0) =: \Delta(f)(0) && (2.18)\\
a_2 &= \frac{f(2)-a_0}{2} - a_1 && (2.19)\\
&= \frac{f(2)-f(0)}{2} - (f(1)-f(0)) && (2.20)\\
&= \frac{f(2)-2f(1)-f(0)}{2} && (2.21)\\
&= \frac{(f(2)-f(1))-(f(1)-f(0))}{2} =: \frac{\Delta^2(f)(0)}{2} && (2.22)\\
&\vdots\\
a_r &= \frac{\Delta^r(f)(0)}{r!}, && (2.23)
\end{aligned}
$$

where Δ is the difference operator in eq.(1.125):

$$\Delta(f)(n) := f(n+1) - f(n). \tag{2.24}$$

In order to simplify our expression, we introduce a falling power:

$$
\begin{aligned}
(x)_0 &:= 1 && (2.25)\\
(x)_n &:= x(x-1)\cdots(x-n+1) && (2.26)\\
&= \prod_{i=0}^{n-1}(x-i). && (2.27)
\end{aligned}
$$

Under these settings, we have

$$\begin{aligned} f(z) &= f_R(z) && (2.28)\\ &= \sum_{r=0}^{R} \frac{\Delta^r(f)(0)}{r!}(x)_r, && (2.29) \end{aligned}$$

where we have assume

$$\Delta^{R+1}(f) = [0, 0, \cdots]. \tag{2.30}$$

Example

Consider a polynomial

$$f(z) := 2 * z^3 + 3 * z, \tag{2.31}$$

and its out put list

$$[f(0), f(1), f(3), \cdots] = [0, 5, 22, 63, 140, 265, \cdots] \tag{2.32}$$

This polynomial is 3rd degree, so we compute up to $\Delta^3(f)(0)$:

$$\begin{aligned} f(0) &= 0 && (2.33)\\ \Delta(f)(0) &= f(1) - f(0) = 5 && (2.34)\\ \Delta^2(f)(0) &= \Delta(f)(1) - \Delta(f)(0) \\ &= f(2) - f(1) - 5 = 22 - 5 - 5 = 12 && (2.35)\\ \Delta^3(f)(0) &= \Delta^2(f)(1) - \Delta^2(f)(0) \\ &= f(3) - f(2) - \{f(2) - f(1)\} - 12 = 12 && (2.36) \end{aligned}$$

so we get

$$[0, 5, 12, 12] \tag{2.37}$$

as the first difference list. Therefore, we get the falling power representation of f:

$$\begin{aligned} f(z) &= 5(x)_1 + \frac{12}{2}(x)_2 + \frac{12}{3!}(x)_3 && (2.38)\\ &= 5(x)_1 + 6(x)_2 + 2(x)_3. && (2.39) \end{aligned}$$

2.2 Univariate polynomial reconstruction with Haskell

2.2.1 Newton interpolation formula with Haskell

First, the falling power is naturally given by recursively:

```
> infixr 8 ^- -- falling power
> (^-) :: (Integral a) => a -> a -> a
> x ^- 0 = 1
> x ^- n = (x ^- (n-1)) * (x - n + 1)
```

Assume the differences are given in a list

$$\texttt{xs} = \left[f(0), \Delta(f)(0), \Delta^2(f)(0), \cdots \right]. \tag{2.40}$$

Then the implementation of the Newton interpolation formula is as follows:

```
> newtonC :: (Fractional t, Enum t) => [t] -> [t]
> newtonC xs = [x / factorial k | (x,k) <- zip xs [0..]]
>   where
>     factorial k = product [1..fromInteger k]
```

Consider a polynomial

```
f x = 2*x^3+3*x
```
(2.41)

Let us try to reconstruct this polynomial from output list. In order to get the list `[x_0, x_1 ..]`, take `difLists` and pick the first elements:

```
> let f x = 2*x^3+3*x
> take 10 $ map f [0..]
[0,5,22,63,140,265,450,707,1048,1485]
> difLists [it]
[[12,12,12,12,12,12,12]
,[12,24,36,48,60,72,84,96]
,[5,17,41,77,125,185,257,341,437]
,[0,5,22,63,140,265,450,707,1048,1485]
]
> reverse $ map head it
[0,5,12,12]
```

This list is the same as eq.(2.37) and we get the same expression as eq.(2.39) $5(x)_1 + 6(x)_2 + 2(x)_3$:

```
> newtonC it
[0 % 1,5 % 1,6 % 1,2 % 1]
```

The list of first differences, i.e.,

$$\left[f(0), \Delta(f)(0), \Delta^2(f)(0), \cdots\right] \tag{2.42}$$

can be computed as follows:

```
> firstDifs :: (Eq a, Num a) => [a] -> [a]
> firstDifs xs = reverse $ map head $ difLists [xs]
```

Mapping a list of integers to a Newton representation:

```
> list2npol :: (Integral a) => [Ratio a] -> [Ratio a]
> list2npol = newtonC . firstDifs
```

```
*NewtonInterpolation> take 10 $ map f [0..]
[0,5,22,63,140,265,450,707,1048,1485]
*NewtonInterpolation> list2npol it
[0 % 1,5 % 1,6 % 1,2 % 1]
```

Therefore, we get the Newton coefficients from the output list.

2.2.2 Stirling numbers of the first kind

We need to map Newton falling powers to standard powers to get the canonical representation. This is a matter of applying combinatorics, by means of a convention formula that uses the so-called Stirling cyclic numbers

$$\left[\begin{array}{c} n \\ k \end{array} \right] \tag{2.43}$$

Its defining relation is, $\forall n > 0$,

$$(x)_n = \sum_{k=1}^{n} (-)^{n-k} \left[\begin{array}{c} n \\ k \end{array} \right] x^k, \tag{2.44}$$

and

$$\left[\begin{array}{c} 0 \\ 0 \end{array} \right] := 1. \tag{2.45}$$

From the highest order, x^n, we get

$$\left[\begin{array}{c} n \\ n \end{array} \right] = 1, \forall n > 0. \tag{2.46}$$

We also put

$$\left[\begin{array}{c} 0 \\ 1 \end{array} \right] = \left[\begin{array}{c} 0 \\ 2 \end{array} \right] = \left[\begin{array}{c} 0 \\ 3 \end{array} \right] = \cdots = 0, \tag{2.47}$$

and

$$\left[\begin{array}{c} 1 \\ 0 \end{array} \right] = \left[\begin{array}{c} 2 \\ 0 \end{array} \right] = \left[\begin{array}{c} 3 \\ 0 \end{array} \right] = \cdots = 0. \tag{2.48}$$

The key equation is

$$(x)_n = (x)_{n-1} * (x - n + 1) \tag{2.49}$$

and we get

$$\begin{aligned}
(x)_n &= \sum_{k=1}^{n} (-)^{n-k} \left[\begin{array}{c} n \\ k \end{array} \right] x^k && (2.50)\\
&= x^n + \sum_{k=1}^{n-1} (-)^{n-k} \left[\begin{array}{c} n \\ k \end{array} \right] x^k && (2.51)\\
(x)_{n-1} * (x - n + 1) &= \sum_{k=1}^{n-1} (-)^{n-1-k} \left\{ \left[\begin{array}{c} n-1 \\ k \end{array} \right] x^{k+1} - (n-1) \left[\begin{array}{c} n-1 \\ k \end{array} \right] x^k \right\} && (2.52)\\
&= \sum_{l=2}^{n} (-)^{n-l} \left[\begin{array}{c} n-1 \\ l-1 \end{array} \right] x^l + (n-1) \sum_{k=1}^{n-1} (-)^{n-k} \left[\begin{array}{c} n \\ k \end{array} \right] x^k && (2.53)\\
&= x^n + (n-1)(-)^{n-1} x \\
&\quad + \sum_{k=2}^{n-1} (-)^{n-k} \left\{ \left[\begin{array}{c} n-1 \\ k-1 \end{array} \right] + (n-1) \left[\begin{array}{c} n-1 \\ k \end{array} \right] \right\} x^k && (2.54)\\
&= x^n + \sum_{k=1}^{n-1} (-)^{n-k} \left\{ \left[\begin{array}{c} n-1 \\ k-1 \end{array} \right] + (n-1) \left[\begin{array}{c} n-1 \\ k \end{array} \right] \right\} x^k && (2.55)
\end{aligned}$$

Therefore, $\forall n, k > 0$,

$$\left[\begin{array}{c} n \\ k \end{array} \right] = \left[\begin{array}{c} n-1 \\ k-1 \end{array} \right] + (n-1) \left[\begin{array}{c} n-1 \\ k \end{array} \right] \tag{2.56}$$

Now we have the following canonical, power representation of reconstructed polynomial

$$\begin{aligned} f(z) &= f_R(z) && (2.57)\\ &= \sum_{r=0}^{R} \frac{\Delta^r(f)(0)}{r!}(x)_r && (2.58)\\ &= \sum_{r=0}^{R} \frac{\Delta^r(f)(0)}{r!} \sum_{k=1}^{r} (-)^{r-k} \left[\begin{array}{c} r \\ k \end{array} \right] x^k, && (2.59) \end{aligned}$$

So, what shall we do is to sum up order by order.

Here is an implementation, first the Stirling numbers:

```
> stirlingC :: Integer -> Integer -> Integer
> stirlingC 0 0 = 1
> stirlingC 0 _ = 0
> stirlingC n k = (n-1)*(stirlingC (n-1) k) + stirlingC (n-1) (k-1)
```

This definition can be used to convert from falling powers to standard powers.

```
> fall2pol :: (Integral a) => a -> [a]
> fall2pol 0 = [1]
> fall2pol n = 0   -- No constant term.
>            : [(-1)^(n-k) * stirlingC n k| k<-[1..n]]
```

We use `fall2pol` to convert Newton representations to standard polynomials in coefficients list representation. Here we have uses `sum` to collect same order terms in list representation.

```
> npol2pol :: (Integral a) => [Ratio a] -> [Ratio a]
> npol2pol xs = sum [ [x] * map fromInteger (fall2pol k)
>                   | (x,k) <- zip xs [0..]
>                   ]
```

2.2.3 `list2pol`: from output list to canonical coefficients

Finally, here is the function for computing a polynomial from an output sequence:

```
> list2pol :: (Integral a) => [Ratio a] -> [Ratio a]
> list2pol = npol2pol . list2npol
```

Here are some checks on these functions:

```
Reconstruction as curve fitting
  *NewtonInterpolation> list2pol $ map (\n -> 7*n^2+3*n-4) [0..100]
  [(-4) % 1,3 % 1,7 % 1]

  *NewtonInterpolation> list2pol [0,1,5,14,30]
  [0 % 1,1 % 6,1 % 2,1 % 3]
  *NewtonInterpolation> map (\n -> n%6 + n^2%2 + n^3%3) [0..4]
  [0 % 1,1 % 1,5 % 1,14 % 1,30 % 1]

  *NewtonInterpolation> map (p2fct $ list2pol [0,1,5,14,30]) [0..8]
  [0 % 1,1 % 1,5 % 1,14 % 1,30 % 1,55 % 1,91 % 1,140 % 1,204 % 1]
```

First example shows that from the sufficiently long output list, we can reconstruct the list of coefficients. Second example shows that from a given outputs, we have a list coefficients. Then use these coefficients, we define the output list of the function, and they match. The last example shows that from a limited (but sufficient) output information, we reconstruct a function and get extra outputs outside from the given data.

2.3 Univariate rational functions

We use the same notion, i.e., what we can know is the output-list of a univariate rational function, say `f::Int -> Ratio Int`:

```
map f [0..]  == [f 0, f 1 ..]
```
(2.60)

2.3.1 Thiele's interpolation formula

We evaluate the polynomial form $f(z)$ as a continued fraction:

$$
\begin{aligned}
f_0(z) &= a_0 && (2.61)\\
f_1(z) &= a_0 + \frac{z}{a_1} && (2.62)\\
&\vdots\\
f_r(z) &= a_0 + \cfrac{z}{a_1 + \cfrac{z-1}{a_2 + \cfrac{z-2}{\ddots + \cfrac{}{a_{r-2} + \cfrac{z-r+1}{a_{r-1} + \cfrac{}{a_r}}}}}}, && (2.63)
\end{aligned}
$$

where

$$
\begin{aligned}
a_0 &= f(0) && (2.64)\\
a_1 &= \frac{1}{f(1) - a_0} && (2.65)\\
a_2 &= \cfrac{1}{\cfrac{2}{f(2) - a_0} - a_1} && (2.66)\\
&\vdots\\
a_r &= \cfrac{1}{\cfrac{2}{\cfrac{3}{\ddots - a_{r-2}} - a_{r-1}} - a_2} - a_1 && (2.67)\\
&= \left(\left((f(r) - a_0)^{-1} r - a_1 \right)^{-1} (r-1) - \cdots - a_{r-1} \right)^{-1} 1 && (2.68)
\end{aligned}
$$

2.3.2 Towards canonical representations

In order to get a unique representation of canonical form

$$\frac{\sum_\alpha n_\alpha z^\alpha}{\sum_\beta d_\beta z^\beta} \tag{2.69}$$

we put

$$d_{\min r'} = 1 \tag{2.70}$$

as a normalization, instead of d_0. However, if we meet 0 as a singular value, then we can shift s.t. the new $d_0 \neq 0$. So without loss of generality, we can assume $f(0)$ is not singular, i.e., the denominator of f has a nonzero constant term:

$$d_0 = 1 \tag{2.71}$$

$$f(z) = \frac{\sum_i n_i z^i}{1 + \sum_{j>0} d_z^j}. \tag{2.72}$$

2.4 Univariate rational function reconstruction with Haskell

Here we the same notion of

`https://rosettacode.org/wiki/Thiele%27s_interpolation_formula`

and especially

`https://rosettacode.org/wiki/Thiele%27s_interpolation_formula#C`

2.4.1 Reciprocal difference

We claim, without proof[2], that the Thiele coefficients are given by

$$a_0 := f(0) \tag{2.73}$$

$$a_n := \rho_{n,0} - \rho_{n-2,0}, \tag{2.74}$$

[2] See the ref.4, Theorem (2.2.2.5) in 2nd edition.

where ρ is so called the reciprocal difference:

$$\begin{aligned} \rho_{n,i} &:= 0, n < 0 && (2.75)\\ \rho_{0,i} &:= f(i), i = 0, 1, 2, \cdots && (2.76)\\ \rho_{n,i} &:= \frac{n}{\rho_{n-1,i+1} - \rho_{n-1,i}} + \rho_{n-2,i+1} && (2.77) \end{aligned}$$

These preparation helps us to write the following codes:

```
Thiele's interpolation formula

Reciprocal difference rho, using the same notation of
https://rosettacode.org/wiki/Thiele%27s_interpolation_formula#C

> rho :: [Ratio Int] -- A list of output of f :: Int -> Ratio Int
>     -> Int -> Int -> Ratio Int
> rho fs 0 i = fs !! i
> rho fs n _
>   | n < 0 = 0
> rho fs n i = (n*den)%num + rho fs (n-2) (i+1)
>   where
>     num  = numerator next
>     den  = denominator next
>     next = (rho fs (n-1) (i+1)) - (rho fs (n-1) i)

Note that (%) has the following type,
  (%) :: Integral a => a -> a -> Ratio a

> a fs 0 = fs !! 0
> a fs n = rho fs n 0 - rho fs (n-2) 0
```

2.4.2 `tDegree` for termination

Now let us consider a simple example which is given by the following Thiele coefficients

$$a_0 = 1, a_1 = 2, a_2 = 3, a_3 = 4. \tag{2.78}$$

The function is now

$$f(x) := 1 + \cfrac{x}{2 + \cfrac{x-1}{3 + \cfrac{x-2}{4}}} \tag{2.79}$$

$$= \frac{x^2 + 16x + 16}{16 + 6x} \tag{2.80}$$

Using Maxima[3], we can verify this:

```
(%i25) f(x) := 1+(x/(2+(x-1)/(3+(x-2)/4)));
(%o25) f(x):=x/(2+(x-1)/(3+(x-2)/4))+1
(%i26) ratsimp(f(x));
(%o26) (x^2+16*x+16)/(16+6*x)
```

Let us come back Haskell, and try to get the Thiele coefficients of

```
*Univariate> let func x = (x^2 + 16*x + 16)%(6*x + 16)
*Univariate> let fs = map func [0..]
*Univariate> map (a fs) [0..]
[1 % 1,2 % 1,3 % 1,4 % 1,*** Exception: Ratio has zero denominator
```

This is clearly unsafe, so let us think more carefully. Observe the reciprocal differences

```
*Univariate> let fs = map func [0..]
*Univariate> take 5 $ map (rho fs 0) [0..]
[1 % 1,3 % 2,13 % 7,73 % 34,12 % 5]
*Univariate> take 5 $ map (rho fs 1) [0..]
[2 % 1,14 % 5,238 % 69,170 % 43,230 % 53]
*Univariate> take 5 $ map (rho fs 2) [0..]
[4 % 1,79 % 16,269 % 44,667 % 88,413 % 44]
*Univariate> take 5 $ map (rho fs 3) [0..]
[6 % 1,6 % 1,6 % 1,6 % 1,6 % 1]
```

So, the constancy of the reciprocal differences can be used to get the depth of Thiele series:

```
> tDegree :: [Ratio Int] -> Int
> tDegree fs = helper fs 0
```

[3] http://maxima.sourceforge.net

```
>   where
>     helper fs n
>       | isConstants fs' = n
>       | otherwise       = helper fs (n+1)
>       where
>         fs' = map (rho fs n) [0..]
>     isConstants (i:j:_) = i==j -- 2 times match
> --  isConstants (i:j:k:_) = i==j && j==k
```

Using this `tDegree` function, we can safely take the (finite) Thiele sequence.

2.4.3 `thieleC`: from output list to Thiele coefficients

From the equation (3.26) of ref.1,

```
*Univariate> let h t = (3+6*t+18*t^2)%(1+2*t+20*t^2)
*Univariate> let hs = map h [0..]
*Univariate> tDegree hs
4
```

So we get the Thiele coefficients

```
*Univariate> map (a hs) [0..(tDegree hs)]
[3 % 1,(-23) % 42,(-28) % 13,767 % 14,7 % 130]
```

Plug these in the continued fraction, and simplify with Maxima

```
(%i35) h(t):=3+t/((-23/42)+(t-1)/((-28/13)+(t-2)/((767/14)+(t-3)/(7/130))));
(%o35) h(t):=t/((-23)/42+(t-1)/((-28)/13+(t-2)/(767/14+(t-3)/(7/130))))+3
(%i36) ratsimp(h(t));
(%o36) (18*t^2+6*t+3)/(1+2*t+20*t^2)
```

Finally we make a function `thieleC` that returns the Thiele coefficients:

```
> thieleC :: [Ratio Int] -> [Ratio Int]
> thieleC lst = map (a lst) [0..(tDegree lst)]

*Univariate> thieleC hs
[3 % 1,(-23) % 42,(-28) % 13,767 % 14,7 % 130]
```

We need a convertor from this Thiele sequence to continuous form of rational function.

```
> nextStep [a0,a1] (v:_)  = a0 + v/a1
> nextStep (a:as)  (v:vs) = a + (v / nextStep as vs)
>
> -- From thiele sequence to (rational) function.
> thiele2ratf :: Integral a => [Ratio a] -> (Ratio a -> Ratio a)
> thiele2ratf as x
>   | x == 0 = head as
>   | otherwise = nextStep as [x,x-1 ..]
```

The following example shows that, the given output lists `hs`, we can interpolate the value between our discrete data.

```
*Univariate> let h t = (3+6*t+18*t^2)%(1+2*t+20*t^2)
*Univariate> let hs = map h [0..]
*Univariate> take 5 hs
[3 % 1,27 % 23,87 % 85,183 % 187,45 % 47]
*Univariate> let as = thieleC hs
*Univariate> as
[3 % 1,(-23) % 42,(-28) % 13,767 % 14,7 % 130]
*Univariate> let th x = thiele2ratf as x
*Univariate> map th [0..5]
[3 % 1,27 % 23,87 % 85,183 % 187,45 % 47,69 % 73]
*Univariate> th 0.5
3 % 2
```

2.4.4 Haskell representation for rational functions

We represent a rational function by a tuple of coefficient lists, like,

```
(ns,ds) ::  ([Ratio Int],[Ratio Int])
```
(2.81)

Here is a translator from coefficients lists to rational function.

```
> lists2ratf :: (Integral a) =>
>   ([Ratio a],[Ratio a]) -> (Ratio a -> Ratio a)
> lists2ratf (ns,ds) x = (p2fct ns x)/(p2fct ds x)
```

```
*Univariate> let frac x = lists2ratf ([1,1%2,1%3],[2,2%3]) x
*Univariate> take 10 $ map frac [0..]
[1 % 2,11 % 16,1 % 1,11 % 8,25 % 14,71 % 32,8 % 3,25 % 8,79 % 22,65 % 16]
*Univariate> let ffrac x = (1+(1%2)*x+(1%3)*x^2)/(2+(2%3)*x)
*Univariate> take 10 $ map ffrac [0..]
[1 % 2,11 % 16,1 % 1,11 % 8,25 % 14,71 % 32,8 % 3,25 % 8,79 % 22,65 % 16]
```

Simply taking numerator and denominator polynomials.

The following `canonicalizer` reduces the tuple-rep of rational function in canonical form, i.e., the coefficient of the lowest degree term of the denominator to be 1[4].

```
> canonicalize :: (Integral a) =>
>   ([Ratio a],[Ratio a]) -> ([Ratio a],[Ratio a])
> canonicalize rat@(ns,ds)
>   | dMin == 1 = rat
>   | otherwise = (map (/dMin) ns, map (/dMin) ds)
>   where
>     dMin = firstNonzero ds
>     firstNonzero [a] = a -- head
>     firstNonzero (a:as)
>       | a /= 0 = a
>       | otherwise = firstNonzero as
```

```
*Univariate> canonicalize ([1,1%2,1%3],[2,2%3])
([1 % 2,1 % 4,1 % 6],[1 % 1,1 % 3])
*Univariate> canonicalize ([1,1%2,1%3],[0,0,2,2%3])
([1 % 2,1 % 4,1 % 6],[0 % 1,0 % 1,1 % 1,1 % 3])
*Univariate> canonicalize ([1,1%2,1%3],[0,0,0,2%3])
([3 % 2,3 % 4,1 % 2],[0 % 1,0 % 1,0 % 1,1 % 1])
```

What we need is a translator from Thiele coefficients to this tuple-rep. Since the list of Thiele coefficients is finite, we can naturally think recursively.

Before we go to a general case, consider

$$f(x) := 1 + \cfrac{x}{2 + \cfrac{x-1}{3 + \cfrac{x-2}{4}}} \tag{2.82}$$

[4] Here our data point start from 0, i.e., the output data is given by `map f [0..]`, 0 is not singular, i.e., the denominator should have constant term and that means non empty. Therefore, the function firstNonzero is actually `head`.

When we simplify this expression, we should start from the bottom:

$$f(x) = 1 + \cfrac{x}{2 + \cfrac{x-1}{\cfrac{4*3+x-2}{4}}} \tag{2.83}$$

$$= 1 + \cfrac{x}{2 + \cfrac{x-1}{\cfrac{x+10}{4}}} \tag{2.84}$$

$$= 1 + \cfrac{x}{\cfrac{2*(x+10)+4*(x-1)}{x+10}} \tag{2.85}$$

$$= 1 + \cfrac{x}{\cfrac{6x+16}{x+10}} \tag{2.86}$$

$$= \frac{1*(6x+16)+x*(x+10)}{6x+16} \tag{2.87}$$

$$= \frac{x^2+16x+16}{6x+16} \tag{2.88}$$

Finally, if we need, we take its canonical form:

$$f(x) = \frac{1 + x + \frac{1}{16}x^2}{1 + \frac{3}{8}x} \tag{2.89}$$

In general, we have the following Thiele representation:

$$a_0 + \cfrac{z}{a_1 + \cfrac{z-1}{a_2 + \cfrac{z-2}{\cfrac{\vdots}{a_n + \cfrac{z-n}{a_{n+1}}}}}} \tag{2.90}$$

The base case should be

$$a_n + \frac{z-n}{a_{n+1}} = \frac{a_{n+1}*a_n - n + z}{a_{n+1}} \tag{2.91}$$

and induction step $0 \leq r \leq n$ should be

$$a_r(z) = a_r + \frac{z-r}{a_{r+1}(z)} \tag{2.92}$$

$$= \frac{a_r a_{r+1}(z) + z - r}{a_{r+1}(z)} \tag{2.93}$$

$$= \frac{a_r * \mathtt{num}\,(a_{r+1}(z)) + \mathtt{den}\,(a_{r+1}(z)) * (z-r)}{\mathtt{num}\,(a_{r+1}(z))} \tag{2.94}$$

where

$$a_{r+1}(z) = \frac{\mathtt{num}\,(a_{r+1}(z))}{\mathtt{den}\,(a_{r+1}(z))} \tag{2.95}$$

is a canonical representation of $a_{n+1}(z)$[5].

Thus, the implementation is the followings.

```
> thiele2coef :: (Integral a) =>
>   [Ratio a] -> ([Ratio a],[Ratio a])
> thiele2coef as = canonicalize $ t2r as 0
>   where
>     t2r [an,an'] n = ([an*an'-n,1],[an'])
>     t2r (a:as)   n = ((a .* num) + ([-n,1] * den), num)
>       where
>         (num, den) = t2r as (n+1)
```

From the first example,

```
*Univariate> let func x = (x^2+16*x+16)%(6*x+16)
*Univariate> let funcList = map func [0..]
*Univariate> tDegree funcList
3
*Univariate> take 5 funcList
[1 % 1,3 % 2,13 % 7,73 % 34,12 % 5]
*Univariate> let aFunc = thieleC funcList
*Univariate> aFunc
[1 % 1,2 % 1,3 % 1,4 % 1]
*Univariate> thiele2coef aFunc
([1 % 1,1 % 1,1 % 16],[1 % 1,3 % 8])
```

From the other example, equation (3.26) of ref.1,

[5] Not necessary being a canonical representation, it suffices to express $a_{n+1}(z)$ in a polynomial over polynomial form, that is, two lists in Haskell.

```
*Univariate> let h t = (3+6*t+18*t^2)%(1+2*t+20*t^2)
*Univariate> let hs = map h [0..]
*Univariate> take 5 hs
[3 % 1,27 % 23,87 % 85,183 % 187,45 % 47]
*Univariate> let th x = thiele2ratf as x
*Univariate> map th [0..5]
[3 % 1,27 % 23,87 % 85,183 % 187,45 % 47,69 % 73]
*Univariate> as
[3 % 1,(-23) % 42,(-28) % 13,767 % 14,7 % 130]
*Univariate> thiele2coef as
([3 % 1,6 % 1,18 % 1],[1 % 1,2 % 1,20 % 1])
```

2.4.5 `list2rat`: from output list to canonical coefficients

Finally, we get

```
> list2rat :: (Integral a) => [Ratio a] -> ([Ratio a], [Ratio a])
> list2rat = thiele2Coef . thieleC
```

as the reconstruction function from the output sequence.

```
*Univariate> let h t = (3+6*t+18*t^2)%(1+2*t+20*t^2)
*Univariate> list2rat $ map h [0..]
([3 % 1,6 % 1,18 % 1],[1 % 1,2 % 1,20 % 1])
```

2.5 Multivariate polynomials

From now on, we will use only the following functions from univariate cases.

`Multivariate.lhs`

```
> module Multivariate
>   where

> import Data.Ratio
> import Univariate
>   ( degree, list2pol
>   , thiele2ratf, lists2ratf, thiele2coef, lists2rat
>   )
```

2.5.1 Foldings as recursive applications

Consider an arbitrary multivariate polynomial

$$f(z_1, \cdots, z_n) \in \mathbb{K}[z_1, \cdots, z_n]. \tag{2.96}$$

First, fix all the variable but 1st and apply the univariate Newton's reconstruction:

$$f(z_1, z_2, \cdots, z_n) = \sum_{r=0}^{R} a_r(z_2, \cdots, z_n) \prod_{i=0}^{r-1}(z_1 - y_i) \tag{2.97}$$

Recursively, pick up one "coefficient" and apply the univariate Newton's reconstruction on z_2:

$$a_r(z_2, \cdots, z_n) = \sum_{s=0}^{S} b_s(z_3, \cdots, z_n) \prod_{j=0}^{s-1}(z_2 - x_j) \tag{2.98}$$

The terminate cotndition should be the univariate case.

2.5.2 Experiments, 2 variables case

Let us take a polynomial from the denominator in eq.(3.23) of ref.1.

$$f(z_1, z_2) = 3 + 2z_1 + 4z_2 + 7z_1^2 + 5z_1z_2 + 6z_2^2 \tag{2.99}$$

In Haskell, first, fix $z_2 = 0, 1, 2$ and identify $f(z_1, 0), f(z_1, 1), f(z_1, 2)$ as our univariate polynomials.

```
*Multivariate> let f z1 z2 = 3+2*z1+4*z2+7*z1^2+5*z1*z2+6*z2^2
*Multivariate> let fs z = map (`f` z) [0..]
*Multivariate> let llst = map fs [0,1,2]
*Multivariate> map degree llst
[2,2,2]
```

Fine, so the canonical form can be

$$f(z_1, z) = c_0(z) + c_1(z)z_1 + c_2(z)z_1^2. \tag{2.100}$$

Now our new target is three univariate polynomials $c_0(z), c_1(z), c_2(z)$.

```
*Multivariate> list2pol $ take 10 $ fs 0
[3 % 1,2 % 1,7 % 1]
*Multivariate> list2pol $ take 10 $ fs 1
[13 % 1,7 % 1,7 % 1]
*Multivariate> list2pol $ take 10 $ fs 2
[35 % 1,12 % 1,7 % 1]
```

That is

$$f(z,0) = 3 + 2z + 7z^2 \tag{2.101}$$
$$f(z,1) = 13 + 7z + 7z^2 \tag{2.102}$$
$$f(z,2) = 35 + 12z + 7z^2. \tag{2.103}$$

From these observation, we can determine $c_2(z)$, since it already a constant sequence.

$$c_2(z) = 7 \tag{2.104}$$

Consider $c_1(z)$, the sequence is now enough to determine $c_1(z)$:

```
*Multivariate> degree [2,7,12]
1
*Multivariate> list2pol [2,7,12]
[2 % 1,5 % 1]
```

i.e.,

$$c_1(z) = 2 + 5z. \tag{2.105}$$

However, for $c_1(z)$

```
*Multivariate> degree [3, 13, 35]
*** Exception: difLists: lack of data, or not a polynomial
CallStack (from HasCallStack):
  error, called at ./Univariate.lhs:61:19 in main:Univariate
```

so we need more numbers. Let us try one more:

```
*Multivariate> list2pol $ take 10 $ map (`f` 3) [0..]
[69 % 1,17 % 1,7 % 1]
*Multivariate> degree [3, 13, 35, 69]
2
*Multivariate> list2pol [3,13,35,69]
[3 % 1,4 % 1,6 % 1]
```

Thus we have

$$c_0(z) = 3 + 4z + 6z^2 \tag{2.106}$$

and these fully determine our polynomial:

$$f(z_1, z_2) = (3 + 4z_2 + 6z_2^2) + (2 + 5z_2)z_1 + 7z_1^2. \tag{2.107}$$

As another experiment, take the denominator.

```
*Multivariate> let g x y = 1+7*x + 8*y + 10*x^2 + x*y+9*y^2
*Multivariate> let gs x = map (g x) [0..]
*Multivariate> map degree $ map gs [0..3]
[2,2,2,2]
```

So the canonical form should be

$$g(x, y) = c_0(x) + c_1(x)y + c_2(x)y^2 \tag{2.108}$$

Let us look at these coefficient polynomial:

```
*Multivariate> list2pol $ take 10 $ gs 0
[1 % 1,8 % 1,9 % 1]
*Multivariate> list2pol $ take 10 $ gs 1
[18 % 1,9 % 1,9 % 1]
*Multivariate> list2pol $ take 10 $ gs 2
[55 % 1,10 % 1,9 % 1]
*Multivariate> list2pol $ take 10 $ gs 3
[112 % 1,11 % 1,9 % 1]
```

So we get

$$c_2(x) = 9 \tag{2.109}$$

and

```
*Multivariate> map (list2pol . (take 10) . gs) [0..4]
[[1 % 1,8 % 1,9 % 1]
,[18 % 1,9 % 1,9 % 1]
,[55 % 1,10 % 1,9 % 1]
,[112 % 1,11 % 1,9 % 1]
,[189 % 1,12 % 1,9 % 1]
]
*Multivariate> map head it
[1 % 1,18 % 1,55 % 1,112 % 1,189 % 1]
*Multivariate> list2pol it
[1 % 1,7 % 1,10 % 1]
*Multivariate> list2pol $ map (head . list2pol . (take 10) . gs) [0..4]
[1 % 1,7 % 1,10 % 1]
```

Using index operator `(!!)`,

```
*Multivariate> list2pol $ map ((!! 0) . list2pol . (take 10) . gs) [0..4]
[1 % 1,7 % 1,10 % 1]
*Multivariate> list2pol $ map ((!! 1) . list2pol . (take 10) . gs) [0..4]
[8 % 1,1 % 1]
*Multivariate> list2pol $ map ((!! 2) . list2pol . (take 10) . gs) [0..4]
[9 % 1]
```

Finally we get

$$c_0(x) = 1 + 7x + 10x^2, c_1(x) = 8 + x, (c_2(x) = 9,) \tag{2.110}$$

and

$$g(x, y) = (1 + 7x + 10x^2) + (8 + x)y + 9y^2 \tag{2.111}$$

2.5.3 Haskell implementation, 2 variables case

Let us assume that we are given a "table" of the values of a 2-variate function. We represent this table as a list of lists.

```
*Multivariate> let f z1 z2 = 3+2*z1+4*z2+7*z1^2+5*z1*z2+6*z2^2
*Multivariate> [[f x y | y <- [0..9]] | x <- [0..9]]
[[3,13,35,69,115,173,243,325,419,525]
,[12,27,54,93,144,207,282,369,468,579]
,[35,55,87,131,187,255,335,427,531,647]
,[72,97,134,183,244,317,402,499,608,729]
,[123,153,195,249,315,393,483,585,699,825]
,[188,223,270,329,400,483,578,685,804,935]
,[267,307,359,423,499,587,687,799,923,1059]
,[360,405,462,531,612,705,810,927,1056,1197]
,[467,517,579,653,739,837,947,1069,1203,1349]
,[588,643,710,789,880,983,1098,1225,1364,1515]
]
```

```
> tablize :: (Enum t1, Num t1) => (t1 -> t1 -> t) -> Int -> [[t]]
> tablize f n = [[f x y | y <- range] | x <- range]
>   where
>     range = take n [0..]
```

So, this "table" is like

$$\begin{pmatrix} f_{0,0} & f_{0,1} & \cdots \\ f_{1,0} & f_{1,1} & \cdots \\ f_{2,0} & f_{2,1} & \cdots \\ \vdots & & \ddots \end{pmatrix} \tag{2.112}$$

Then we can apply the univariate technique.

```
*Multivariate> let fTable = tablize f 10
*Multivariate> map list2pol fTable
[[3 % 1,4 % 1,6 % 1]
,[12 % 1,9 % 1,6 % 1]
,[35 % 1,14 % 1,6 % 1]
,[72 % 1,19 % 1,6 % 1]
,[123 % 1,24 % 1,6 % 1]
,[188 % 1,29 % 1,6 % 1]
,[267 % 1,34 % 1,6 % 1]
,[360 % 1,39 % 1,6 % 1]
,[467 % 1,44 % 1,6 % 1]
,[588 % 1,49 % 1,6 % 1]
]
```

Now we need to see the behavior of each coefficient, so take the "transpose" of it:

```
> wellOrd :: [[a]] -> [[a]]
> wellOrd xss
>   | null (head xss) = []
>   | otherwise       = map head xss : wellOrd (map tail xss)
```

```
*Multivariate> let f z1 z2 = 3+2*z1+4*z2+7*z1^2+5*z1*z2+6*z2^2
*Multivariate> let fTable = tablize f 10
*Multivariate> map list2pol fTable
[[3 % 1,4 % 1,6 % 1]
,[12 % 1,9 % 1,6 % 1]
,[35 % 1,14 % 1,6 % 1]
,[72 % 1,19 % 1,6 % 1]
,[123 % 1,24 % 1,6 % 1]
,[188 % 1,29 % 1,6 % 1]
,[267 % 1,34 % 1,6 % 1]
```

```
  ,[360 % 1,39 % 1,6 % 1]
  ,[467 % 1,44 % 1,6 % 1]
  ,[588 % 1,49 % 1,6 % 1]
  ]
  *Multivariate> wellOrd it
  [[3 % 1,12 % 1,35 % 1,72 % 1,123 % 1,188 % 1,267 % 1,360 % 1,467 % 1,588 % 1]
  ,[4 % 1,9 % 1,14 % 1,19 % 1,24 % 1,29 % 1,34 % 1,39 % 1,44 % 1,49 % 1]
  ,[6 % 1,6 % 1,6 % 1,6 % 1,6 % 1,6 % 1,6 % 1,6 % 1,6 % 1,6 % 1]
  ]
  *Multivariate> map list2pol it
  [[3 % 1,2 % 1,7 % 1]
  ,[4 % 1,5 % 1]
  ,[6 % 1]]
```

Therefore, the whole procedure becomes

```
> table2pol :: [[Ratio Integer]] -> [[Ratio Integer]]
> table2pol = map list2pol . wellOrd . map list2pol

  *Multivariate> let g x y = 1+7*x + 8*y + 10*x^2 + x*y+9*y^2
  *Multivariate> table2pol $ tablize g 5
  [[1 % 1,7 % 1,10 % 1],[8 % 1,1 % 1],[9 % 1]]
```

2.6 Multivariate rational functions

2.6.1 The canonical normalization

Our target is a pair of coefficients $(\{n_\alpha\}_\alpha, \{d_\beta\}_\beta)$ in

$$\frac{\sum_\alpha n_\alpha z^\alpha}{\sum_\beta d_\beta z^\beta} \tag{2.113}$$

A canonical choice is

$$d_0 = d_{(0,\cdots,0)} = 1. \tag{2.114}$$

Accidentally we might face $d_0 = 0$, but we can shift our function and make

$$d'_0 = d_s \neq 0. \tag{2.115}$$

2.6.2 An auxiliary t

Introducing an auxiliary variable t, let us define

$$h(z,t) := f(tz_1, \cdots, tz_n), \tag{2.116}$$

and reconstruct $h(t,z)$ as a univariate rational function of t:

$$h(z,t) = \frac{\sum_{r=0}^{R} p_r(z) t^r}{1 + \sum_{r'=1}^{R'} q_{r'}(z) t^{r'}} \tag{2.117}$$

where

$$p_r(z) = \sum_{|\alpha|=r} n_\alpha z^\alpha \tag{2.118}$$

$$q_{r'}(z) = \sum_{|\beta|=r'} n_\beta z^\beta \tag{2.119}$$

are homogeneous polynomials.

Thus, what we shall do is the (homogeneous) polynomial reconstructions of $p_r(z)|_{0\leq r\leq R}$, $q_{r'}|_{1\leq r'\leq R'}$.

A simplification

Since our new targets are homogeneous polynomials, we can consider, say,

$$p_r(1, z_2, \cdots, z_n) \tag{2.120}$$

instead of $p_r(z_1, z_2, \cdots, z_n)$, reconstruct it using multivariate Newton's method, and homogenize with z_1.

2.6.3 Experiments, 2 variables case

Consider the equation (3.23) in ref.1.

```
*Multivariate> let f x y = (3+2*x+4*y+7*x^2+5*x*y+6*y^2)
                           % (1+7*x+8*y+10*x^2+x*y+9*y^2)
*Multivariate> :t f
f :: Integral a => a -> a -> Ratio a
*Multivariate> let h x y t = f (t*x) (t*y)
*Multivariate> let hs x y = map (h x y) [0..]
*Multivariate> take 5 $ hs 0 0
[3 % 1,3 % 1,3 % 1,3 % 1,3 % 1]
```

```
*Multivariate> take 5 $ hs 0 1
[3 % 1,13 % 18,35 % 53,69 % 106,115 % 177]
*Multivariate> take 5 $ hs 1 0
[3 % 1,2 % 3,7 % 11,9 % 14,41 % 63]
*Multivariate> take 5 $ hs 1 1
[3 % 1,3 % 4,29 % 37,183 % 226,105 % 127]
```

Here we have introduced the auxiliary t as third argument.

We take $(x, y) = (1, 0), (1, 1), (1, 2), (1, 3)$ and reconstruct them[6].

```
*Multivariate> lists2rat $ hs 1 0
([3 % 1,2 % 1,7 % 1],[1 % 1,7 % 1,10 % 1])
*Multivariate> lists2rat $ hs 1 1
([3 % 1,6 % 1,18 % 1],[1 % 1,15 % 1,20 % 1])
*Multivariate> lists2rat $ hs 1 2
([3 % 1,10 % 1,41 % 1],[1 % 1,23 % 1,48 % 1])
*Multivariate> lists2rat $ hs 1 3
([3 % 1,14 % 1,76 % 1],[1 % 1,31 % 1,94 % 1])
```

So we have

$$h(1,0,t) = \frac{3+2t+7t^2}{1+7t+10t^2} \tag{2.121}$$

$$h(1,1,t) = \frac{3+6t+18t^2}{1+15t+20t^2} \tag{2.122}$$

$$h(1,2,t) = \frac{3+10t+41t^2}{1+23t+48t^2} \tag{2.123}$$

$$h(1,3,t) = \frac{3+14t+76t^2}{1+31t+94t^2} \tag{2.124}$$

Our next targets are the coefficients as polynomials in y [7].

Let us consider numerator first. This `list` is Haskell representation for eq.(2.121), eq.(2.122), eq.(2.123) and eq.(2.124).

```
*Multivariate> let list = map (lists2rat . (hs 1)) [0..4]
*Multivariate> let numf = map fst list
*Multivariate> list
[([3 % 1,2 % 1,7 % 1],[1 % 1,7 % 1,10 % 1])
,([3 % 1,6 % 1,18 % 1],[1 % 1,15 % 1,20 % 1])
```

[6] Eq.(3.26) in ref.1 is different from our reconstruction.

[7] In our example, we take $x = 1$ fixed and reproduce x-dependence using homogenization

```
,([3 % 1,10 % 1,41 % 1],[1 % 1,23 % 1,48 % 1])
,([3 % 1,14 % 1,76 % 1],[1 % 1,31 % 1,94 % 1])
,([3 % 1,18 % 1,123 % 1],[1 % 1,39 % 1,158 % 1])
]
*Multivariate> numf
[[3 % 1,2 % 1,7 % 1]
,[3 % 1,6 % 1,18 % 1]
,[3 % 1,10 % 1,41 % 1]
,[3 % 1,14 % 1,76 % 1]
,[3 % 1,18 % 1,123 % 1]
]
```

From this information, we reconstruct each polynomials

```
*Multivariate> list2pol $ map head numf
[3 % 1]
*Multivariate> list2pol $ map (head . tail) numf
[2 % 1,4 % 1]
*Multivariate> list2pol $ map last numf
[7 % 1,5 % 1,6 % 1]
```

that is we have $3, 2 + 4y, 7 + 5y + 6y^2$ as results. Similarly,

```
*Multivariate> let denf = map snd list
*Multivariate> denf
[[1 % 1,7 % 1,10 % 1]
,[1 % 1,15 % 1,20 % 1]
,[1 % 1,23 % 1,48 % 1]
,[1 % 1,31 % 1,94 % 1]
,[1 % 1,39 % 1,158 % 1]
]
*Multivariate> list2pol $ map head denf
[1 % 1]
*Multivariate> list2pol $ map (head . tail) denf
[7 % 1,8 % 1]
*Multivariate> list2pol $ map last denf
[10 % 1,1 % 1,9 % 1]
```

So we get

$$h(1, y, t) = \frac{3 + (2 + 4y)t + (7 + 5y + 6y^2)t^2}{1 + (7 + 8y)t + (10 + y + 9y^2)t^2} \tag{2.125}$$

Finally, we use the homogeneous property for each powers:

$$h(x,y,t) = \frac{3 + (2x+4y)t + (7x^2+5xy+6y^2)t^2}{1 + (7x+8y)t + (10x^2+xy+9y^2)t^2} \tag{2.126}$$

Putting $t = 1$, we get

$$\begin{aligned} f(x,y) &= h(x,y,1) && (2.127)\\ &= \frac{3 + (2x+4y) + (7x^2+5xy+6y^2)}{1 + (7x+8y) + (10x^2+xy+9y^2)} && (2.128) \end{aligned}$$

2.6.4 Haskell implementation, 2 variables case

Assume we have a "table" of data:

```
*Multivariate> let h x y = (3+2*x+4*y+7*x^2+5*x*y+6*y^2) % (1+7*x+8*y+10*x^2+x*y+9*y^2)
*Multivariate> let auxh x y t = h (t*x) (t*y)
*Multivariate> let h x y = (3+2*x+4*y+7*x^2+5*x*y+6*y^2)% (1+7*x+8*y+10*x^2+x*y+9*y^2)
*Multivariate> let auxh x y t = h (t*x) (t*y)
```

```
Using the homogenious property, we just take x=1:
```

```
*Multivariate> let auxhs = [map (auxh 1 y) [0..5] | y <- [0..5]]
*Multivariate> auxhs
[[3 % 1,2 % 3,7 % 11,9 % 14,41 % 63,94 % 143]
,[3 % 1,3 % 4,29 % 37,183 % 226,105 % 127,161 % 192]
,[3 % 1,3 % 4,187 % 239,201 % 251,233 % 287,77 % 94]
,[3 % 1,31 % 42,335 % 439,729 % 940,425 % 543,1973 % 2506]
,[3 % 1,8 % 11,59 % 79,291 % 385,681 % 895,528 % 691]
,[3 % 1,23 % 32,155 % 211,1707 % 2302,1001 % 1343,4663 % 6236]
]
```

Now, each list can be seen as a univariate rational function:

```
*Multivariate> map list2rat auxhs
[([3 % 1,2 % 1,7 % 1],[1 % 1,7 % 1,10 % 1])
,([3 % 1,6 % 1,18 % 1],[1 % 1,15 % 1,20 % 1])
,([3 % 1,10 % 1,41 % 1],[1 % 1,23 % 1,48 % 1])
,([3 % 1,14 % 1,76 % 1],[1 % 1,31 % 1,94 % 1])
,([3 % 1,18 % 1,123 % 1],[1 % 1,39 % 1,158 % 1])
,([3 % 1,22 % 1,182 % 1],[1 % 1,47 % 1,240 % 1])
]
```

```
*Multivariate> map fst it
[[3 % 1,2 % 1,7 % 1]
,[3 % 1,6 % 1,18 % 1]
,[3 % 1,10 % 1,41 % 1]
,[3 % 1,14 % 1,76 % 1]
,[3 % 1,18 % 1,123 % 1]
,[3 % 1,22 % 1,182 % 1]
]
```

We need to see the behavior of each coefficients:

```
*Multivariate> wellOrd it
[[3 % 1,3 % 1,3 % 1,3 % 1,3 % 1,3 % 1]
,[2 % 1,6 % 1,10 % 1,14 % 1,18 % 1,22 % 1]
,[7 % 1,18 % 1,41 % 1,76 % 1,123 % 1,182 % 1]
]
*Multivariate> map list2pol it
[[3 % 1],[2 % 1,4 % 1],[7 % 1,5 % 1,6 % 1]]
```

So, the numerator is given by

```
*Multivariate> map list2pol . wellOrd . map (fst . list2rat) $ auxhs
[[3 % 1],[2 % 1,4 % 1],[7 % 1,5 % 1,6 % 1]]
```

and the denominator is

```
*Multivariate> map list2pol . wellOrd . map (snd . list2rat) $ auxhs
[[1 % 1],[7 % 1,8 % 1],[10 % 1,1 % 1,9 % 1]]
```

Thus, we finally get the following function

```
> table2ratf table = (t2r fst table, t2r snd table)
>   where
>     t2r third = map list2pol . wellOrd . map (third . list2rat)
```

```
*Multivariate> table2ratf auxhs
([[3 % 1],[2 % 1,4 % 1],[7 % 1,5 % 1,6 % 1]],[[1 % 1],[7 % 1,8 % 1],[10 % 1,1 %
```

Chapter 3

Functional reconstruction over finite fields

3.1 Univariate polynomials

We choose our new target the first differences, since once we get it, to reconstruct polynomial is an easy task. Once we get the first differences of a polynomial, we get the coefficient list by applying `npol2pol . newtonC` on it.

3.1.1 Pre-cook

We need a convertor, or function-modular which takes function and prime, and returns a function on $\mathbb{Z}_p$.

```
> -- Function-modular.
> fmodp :: Integral c => (a -> Ratio c) -> c -> a -> c
> f `fmodp` p = (`modp` p) . f

  *FROverZp> let f x = (1%3) + (3%5)*x + (7%6)*x^2
  *FROverZp> let fs = map f [0..]
  *FROverZp> take 5 $ map (f `fmodp` 101) [0..]
  [34,93,87,16,82]
  *FROverZp> take 5 $ map (`modp` 101) fs
  [34,93,87,16,82]
```

What we can access is the output list of our target polynomial. On $\mathbb{Z}_p$,

our input is a finite list

```
[0,1,2 ..  (p-1)]
```
(3.1)

so as the output list.

```
> accessibleData :: (Ratio Int -> Ratio Int) -> Int -> [Int]
> accessibleData f p = take p $ map (f `fmodp` p) [0..]
>
> accessibleData' :: [Ratio Int] -> Int -> [Int]
> accessibleData' fs p = take p $ map (`modp` p) fs
```

3.1.2 Difference analysis on $\mathbb{Z}_p$

Play the same game over prime field $\mathbb{Z}_p$, i.e., every arithmetic in under mod p.

```
Difference analysis over Z_p
Every arithmetic should be on Z_p, i.e., (`mod` p).

> difsp :: Integral b => b -> [b] -> [b]
> difsp p xs = map (`mod` p) (zipWith (-) (tail xs) xs)

  *FROverZp> let f x = (1%3) + (3%5)*x + (7%6)*x^2
  *FROverZp> take 5 $ accessibleData f 101
  [34,93,87,16,82]
  *FROverZp> difsp 101 it
  [59,95,30,66]
  *FROverZp> difsp 101 it
  [36,36,36]
  *FROverZp> difsp 101 it
  [0,0]
```

Here what we do is, first to take the differences (`zipWith (-) (tail xs) xs`) and take modular p for all element (`map (`mod` p)`). Now we can recursively apply this `difsp` over our data.

```
> difListsp :: Integral b => b -> [[b]] -> [[b]]
> difListsp _ [] = []
> difListsp p xx@(xs:xxs) =
>   if isConst xs then xx
>                 else difListsp p $ difsp p xs : xx
```

```
>   where
>     isConst (i:jj@(j:js)) = all (==i) jj
>     isConst _ = error "difListsp: "
```

```
*FROverZp> let f x = (1%3) + (3%5)*x + (7%6)*x^2
*FROverZp> map head $ difListsp 101 [accessibleData f 101]
[36,59,34]
```

3.1.3 Eager and lazy degree

From the above difference analysis on $\mathbb{Z}_p$, we get degree of the polynomial. Here we have a combination of two degree functions, one is eager and the other lazy:

```
Degree, eager and lazy versions

> degreep' p xs = length (difListsp p [xs]) -1
> degreep'Lazy p xs = helper xs 0
>   where
>     helper as@(a:b:c:_) n
>       | a==b && b==c = n -- two times matching
>       | otherwise    = helper (difsp p as) (n+1)
>
> degreep :: Integral b => b -> [b] -> Int
> degreep p xs = let l = degreep'Lazy p xs in
>   degreep' p $ take (l+2) xs
```

```
*FROverZp> let f x = (1%3) + (3%5)*x + (7%6)*x^2
*FROverZp> let myDeg p = degreep p $ accessibleData f p
*FROverZp> myDeg 101
2
*FROverZp> myDeg 103
2
*FROverZp> myDeg 107
2
*FROverZp> degreep 101 $ accessibleData
  (\n -> (1%2)+(2%3)*n+(3%4)*n^2+(6%7)*n^7) 101
7
```

Now we can take first differences.

```
> firstDifsp :: Integral a => a -> [a] -> [a]
> firstDifsp p xs = reverse $ map head $ difListsp p [xs']
>   where
>     xs' = take n xs
>     n   = 2+ degreep p xs
```

```
*FROverZp> let f x = (1%3) + (3%5)*x + (7%6)*x^2
*FROverZp> firstDifsp 101 $ accessibleData f 101
[34,59,36]
*FROverZp> firstDifsp 101 $ accessibleData
  (\n -> (1%2)+(2%3)*n+(3%4)*n^2+(6%7)*n^7) 101
[51,66,59,33,29,58,32,78]
```

3.1.4 Term by term reconstruction

The output list of `firstDifsp` are basically the coefficients of Newton representation on $\mathbb{Z}_p$. So we zip it with our base prime p and map these pair over several primes.

```
> wellOrd :: [[a]] -> [[a]]
> wellOrd xss
>   | null (head xss) = []
>   | otherwise       = map head xss : wellOrd (map tail xss)
```

```
*FROverZp> let f x = (1%3) + (3%5)*x + (7%6)*x^2
*FROverZp> let fps p = accessibleData f p
*FROverZp> let ourData p = firstDifsp p (fps p)
*FROverZp> let fivePrimes = take 5 bigPrimes
*FROverZp> map (\p -> zip (ourData p) (repeat p)) fivePrimes
[[(299158,897473),(867559,897473),(299160,897473)]
,[(299166,897497),(329084,897497),(299168,897497)]
,[(598333,897499),(388918,897499),(598335,897499)]
,[(598345,897517),(29919,897517),(598347,897517)]
,[(299176,897527),(329095,897527),(299178,897527)]
]
*FROverZp> wellOrd it
[[(299158,897473),(299166,897497),(598333,897499)
 ,(598345,897517),(299176,897527)]
,[(867559,897473),(329084,897497),(388918,897499)
 ,(29919,897517),(329095,897527)]
```

```
 ,[(299160,897473),(299168,897497),(598335,897499)
  ,(598347,897517),(299178,897527)]
 ]
 *FROverZp> :t it
 it :: [[(Int, Int)]]
```

Finally we get the images of first differences over prime fields.

One minor issue is to change the data type, since our tools (say the functions of Chinese Remainder Theorem) use limit-less integer `Integer`.

```
> toInteger2 :: (Integral a1, Integral a) => (a, a1) -> (Integer, Integer)
> toInteger2 (a,b) = (toInteger a, toInteger b)
```

Let us take an example:

```
 *FROverZp> let f x = (895 % 922) + (1080 % 6931)*x + (2323 % 1248)*x^2
 *FROverZp> let fps p = accessibleData f p
 *FROverZp> let longList = map (map toInteger2) $ wellOrd $
   map (\p -> zip (firstDifsp p (fps p)) (repeat p)) bigPrimes
 *FROverZp> map recCRT' longList
 [(895 % 922,805479325081)
 ,(17448553 % 8649888,722916888780872419)
 ,(2323 % 624,805479325081)
 ]
```

This result is consistent to that of on Q:

```
 *FROverZp> :l Univariate
 [1 of 2] Compiling Polynomials      ( Polynomials.hs, interpreted )
 [2 of 2] Compiling Univariate       ( Univariate.lhs, interpreted )
 Ok, modules loaded: Univariate, Polynomials.
 *Univariate> let f x = (895 % 922) + (1080 % 6931)*x + (2323 % 1248)*x^2
 *Univariate> firstDifs (map f [0..20])
 [895 % 922,17448553 % 8649888,2323 % 624]
```

3.1.5 `list2polZp`: from the output list to coefficient lists

Finally we get the function which takes an output list of our unknown univariate polynomial and returns the coefficient.

```
> list2firstDifZp' fs =
>   map recCRT' $ map (map toInteger2) $ wellOrd $ map helper bigPrimes
```

```
>   where helper p = zip (firstDifsp p (accessibleData' fs p)) (repeat p)
```

```
  *FROverZp> let f x = (895 % 922) + (1080 % 6931)*x + (2323 % 1248)*x^2
  *FROverZp> let fs = map f [0..]
  *FROverZp> list2firstDifZp' fs
  [(895 % 922,805479325081)
  ,(17448553 % 8649888,722916888780872419)
  ,(2323 % 624,805479325081)
  ]
  *FROverZp> map fst it
  [895 % 922,17448553 % 8649888,2323 % 624]
  *FROverZp> newtonC it
  [895 % 922,17448553 % 8649888,2323 % 1248]
  *FROverZp> npol2pol it
  [895 % 922,1080 % 6931,2323 % 1248]
```

```
> list2polZp :: [Ratio Int] -> [Ratio Integer]
> list2polZp = npol2pol . newtonC . (map fst) . list2firstDifZp'
```

3.2 TBA Univariate rational functions

Chapter 4

Codes

4.1 Ffield.lhs

Listing 4.1: Ffield.lhs

```
1  Ffield.lhs
2
3  https://arxiv.org/pdf/1608.01902.pdf
4
5  > module Ffield where
6
7  > import Data.Ratio
8  > import Data.Maybe
9  > import Data.Numbers.Primes
10 > import Test.QuickCheck
11
12 > coprime :: Integral a => a -> a -> Bool
13 > coprime a b = gcd a b == 1
14
15 Consider a finite ring
16   Z_n := [0..(n-1)]
17 of some Int number.
18 If any non-zero element has its multiplication inverse,
       then the ring is a field:
19
20 > -- Our target should be in Int.
21 > isField :: Int -> Bool
22 > isField = isPrime
23
24 Here we would like to implement the extended Euclidean
       algorithm.
```

```
25 See the algorithm, examples, and pseudo code at:
26
27   https://en.wikipedia.org/wiki/
        Extended_Euclidean_algorithm
28   http://qiita.com/bra_cat_ket/items/205c19611e21f3d422b7
29
30 > exGCD' :: (Integral n) => n -> n -> ([n], [n], [n], [n
       ])
31 > exGCD' a b = (qs, rs, ss, ts)
32 >   where
33 >     qs = zipWith quot rs (tail rs)
34 >     rs = takeUntil (==0) r'
35 >     r' = steps a b
36 >     ss = steps 1 0
37 >     ts = steps 0 1
38 >     steps a b = rr
39 >       where
40 >         rr@(_:rs) = a:b: zipWith (-) rr (zipWith (*) qs
       rs)
41 >
42 > takeUntil :: (a -> Bool) -> [a] -> [a]
43 > takeUntil p = foldr func []
44 >   where
45 >     func x xs
46 >       | p x       = []
47 >       | otherwise = x : xs
48 >
49 > -- a*x + b*y = gcd a b
50 > exGCD
51 >   :: Integral t =>
52 >      t -> t -> (t, t, t)
53 > exGCD a b = (g, x, y)
54 >   where
55 >     (_,r,s,t) = exGCD' a b
56 >     g = last r
57 >     x = last . init $ s
58 >     y = last . init $ t
59 >
60 > -- a^{-1} (in Z_p) == a `inversep` p
61 > inversep
62 >   :: Integral a =>
63 >      a -> a -> Maybe a -- We also use in CRT.
64 > a `inversep` p = let (g,x,_) = exGCD a p in
65 >   if (g == 1)
66 >     then Just (x `mod` p) -- g==1 <=> coprime a p
```

```
67  >     else Nothing
68  >
69  > -- If a is "safe" value, we can use this.
70  > inversep' :: Int -> Int -> Int
71  > 0 `inversep'` _ = error "zero␣division"
72  > a `inversep'` p = (x `mod` p)
73  >   where (_,x,_) = exGCD a p
74  >
75  > inversesp :: Int -> [Maybe Int]
76  > inversesp p = map (`inversep` p) [1..(p-1)]
77  >
78  > -- A map from Q to Z_p, where p is a prime.
79  > modp :: Ratio Int -> Int -> Maybe Int
80  > q `modp` p
81  >   | coprime b p = Just $ (a * (bi `mod` p)) `mod` p
82  >   | otherwise   = Nothing
83  >   where
84  >     (a,b)   = (numerator q, denominator q)
85  >     Just bi = b `inversep` p
86  >
87  > -- When the denominator of q is not proprtional to p,
        use this.
88  > modp' :: Ratio Int -> Int -> Int
89  > q `modp'` p = (a * (bi `mod` p)) `mod` p
90  >   where
91  >     (a,b)   = (numerator q, denominator q)
92  >     bi = b `inversep'` p
93  >
94  > -- This is guess function without Chinese Reminder
        Theorem.
95  > guess :: Integral t =>
96  >          (Maybe t, t)       -- (q `modp` p, p)
97  >       -> Maybe (Ratio t, t)
98  > guess (Nothing, _) = Nothing
99  > guess (Just a, p) = let (_,rs,ss,_) = exGCD' a p in
100 >   Just (select rs ss p, p)
101 >     where
102 >       select :: Integral t => [t] -> [t] -> t -> Ratio
        t
103 >       select [] _ _ = 0%1
104 >       select (r:rs) (s:ss) p
105 >         | s /= 0 && r*r <= p && s*s <= p = r%s
106 >         | otherwise = select rs ss p
107 >
108 > -- Hard code of big primes
```

```
> -- We have chosen a finite number (100) version.
> bigPrimes :: [Int]
> -- bigPrimes = take 100 $ dropWhile (<10^4) primes
> bigPrimes = take 100 $ dropWhile (<10^6) primes

  *Ffield> let knownData q = zip (map (modp q) bigPrimes)
      bigPrimes
  *Ffield> let ds = knownData (12%13)
  *Ffield> map guess ds
  [Just (12 % 13,10007)
  ,Just (12 % 13,10009)
  ,Just (12 % 13,10037)
  ,Just (12 % 13,10039) ..

  *Ffield> let ds = knownData (112%113)
  *Ffield> map guess ds
  [Just ((-39) % 50,10007)
  ,Just ((-41) % 48,10009)
  ,Just ((-69) % 20,10037)
  ,Just ((-71) % 18,10039) ..

--

Chinese Remainder Theorem, and its usage

> imagesAndPrimes :: Ratio Int -> [(Maybe Int, Int)]
> imagesAndPrimes q = zip (map (modp q) bigPrimes)
    bigPrimes

Our data is a list of the type
  [(Maybe Int, Int)]
In order to use CRT, we should cast its type.

> toInteger2 :: [(Maybe Int, Int)] -> [(Maybe Integer,
    Integer)]
> toInteger2 = map helper
>   where
>     helper (x,y) = (fmap toInteger x, toInteger y)
>
> crtRec'':: Integral a => (Maybe a, a) -> (Maybe a, a) ->
    (Maybe a, a)
> crtRec' (Nothing,p) (_,q)       = (Nothing, p*q)
> crtRec' (_,p)       (Nothing,q) = (Nothing, p*q)
> crtRec' (Just a1,p1) (Just a2,p2) = (Just a,p)
>   where
```

```
150 >     a = (a1*p2*m2 + a2*p1*m1) `mod` p
151 >     Just m1 = p1 `inversep` p2
152 >     Just m2 = p2 `inversep` p1
153 >     p = p1*p2
154 >
155 > matches3 :: Eq a => [Maybe (a,b)] -> Maybe (a,b)
156 > matches3 (b1@(Just (q1,p1)):bb@((Just (q2,_)):(Just (q3
      ,_)):_))
157 >   | q1==q2 && q2==q3 = b1
158 >   | otherwise        = matches3 bb
159 > matches3 _ = Nothing
160
161   *Ffield> let ds = imagesAndPrimes (1123%1135)
162   *Ffield> map guess ds
163   [Just (25 % 52,10007)
164   ,Just ((-81) % 34,10009)
165   ,Just ((-88) % 63,10037) ..
166
167   *Ffield> matches3 it
168   Nothing
169
170   *Ffield> scanl1 crtRec' ds
171
172   *Ffield> scanl1 crtRec' . toInteger2 $ ds
173   [(Just 3272,10007)
174   ,(Just 14913702,100160063)
175   ,(Just 298491901442,1005306552331) ..
176
177   *Ffield> map guess it
178   [Just (25 % 52,10007)
179   ,Just (1123 % 1135,100160063)
180   ,Just (1123 % 1135,1005306552331)
181   ,Just (1123 % 1135,10092272478850909) ..
182
183   *Ffield> matches3 it
184   Just (1123 % 1135,100160063)
185
186 > reconstruct :: [(Maybe Int, Int)] -> Maybe (Ratio
      Integer)
187 > reconstruct = matches 10 . makeList -- 10 times match
188 >   where
189 >     matches n (a:as)
190 >       | all (a==) $ take (n-1) as = a
191 >       | otherwise                 = matches n as
192 >     makeList = map (fmap fst . guess) . scanl1 crtRec'
```

```
    . toInteger2 . filter (isJust . fst)
193
194 -- QuickCheck
195
196   *Ffield> let q = 513197683989569 % 1047805145658 ::
          Ratio Int
197   *Ffield> let ds = imagesAndPrimes q
198   *Ffield> let answer = fmap fromRational . reconstruct $
          ds
199   *Ffield> answer :: Maybe (Ratio Int)
200   Just (513197683989569 % 1047805145658)
201
202 > prop_rec :: Ratio Int -> Bool
203 > prop_rec q = Just q == answer
204 >   where
205 >     answer :: Maybe (Ratio Int)
206 >     answer = fmap fromRational . reconstruct $ ds
207 >     ds = imagesAndPrimes q
208
209   *Ffield> quickCheckWith stdArgs { maxSuccess = 100000 }
          prop_rec
210   +++ OK, passed 100000 tests.
```

4.2 Polynomials.hs

Listing 4.2: Polynomials.hs

```
1  -- Polynomials.hs
2  -- http://homepages.cwi.nl/~jve/rcrh/Polynomials.hs
3
4  module Polynomials where
5
6  default (Integer, Rational, Double)
7
8  -- scalar multiplication
9  infixl 7 .*
10 (.*) :: Num a => a -> [a] -> [a]
11 c .* []     = []
12 c .* (f:fs) = c*f : c .* fs
13
14 z :: Num a => [a]
15 z = [0,1]
16
17 -- polynomials, as coefficients lists
18 instance (Num a, Ord a) => Num [a] where
```

```
19    fromInteger c = [fromInteger c]
20    -- operator overloading
21    negate []     = []
22    negate (f:fs) = (negate f) : (negate fs)
23
24    signum [] = []
25    signum gs
26      | signum (last gs) < (fromInteger 0) = negate z
27      | otherwise = z
28
29    abs [] = []
30    abs gs
31      | signum gs == z = gs
32      | otherwise      = negate gs
33
34    fs     + []     = fs
35    []     + gs     = gs
36    (f:fs) + (g:gs) = f+g : fs+gs
37
38    fs     * []     = []
39    []     * gs     = []
40    (f:fs) * gg@(g:gs) = f*g : (f .* gs + fs * gg)
41
42  delta :: (Num a, Ord a) => [a] -> [a]
43  delta = ([1,-1] *)
44
45  shift :: [a] -> [a]
46  shift = tail
47
48  p2fct :: Num a => [a] -> a -> a
49  p2fct [] x = 0
50  p2fct (a:as) x = a + (x * p2fct as x)
51
52  comp :: (Eq a, Num a, Ord a) => [a] -> [a] -> [a]
53  comp _      []      = error ".."
54  comp []      _      = []
55  comp (f:fs) g0@(0:gs) = f : gs * (comp fs g0)
56  comp (f:fs) gg@(g:gs) = ([f] + [g] * (comp fs gg))
57                        + (0 : gs * (comp fs gg))
58
59  deriv :: Num a => [a] -> [a]
60  deriv []     = []
61  deriv (f:fs) = deriv1 fs 1
62    where
63      deriv1 []     _ = []
```

```
64      deriv1 (g:gs) n = n*g : deriv1 gs (n+1)
```

4.3 Univariate.lhs

Listing 4.3: Univariate.lhs

```
1  Univariate.lhs
2
3  > module Univariate where
4
5  References
6  J. Stoer, R. Bulirsch
7    Introduction to Numerical Analysis (2nd edition)
8  L. M. Milne-Thomson
9    THE CALCULUS OF FINITE DIFFERENCES
10
11 > import Control.Applicative
12 > import Control.Monad
13 > import Data.Ratio
14 > import Data.Maybe
15 > import Data.List
16 > -- import Control.Monad.Catch
17 >
18 > import Polynomials
19
20 From the output list
21   map f [0..]
22 of a polynomial
23   f :: Int -> Ratio Int
24 we reconstrunct the canonical form of f.
25
26 > difs :: (Num a) => [a] -> [a]
27 > difs [] = []
28 > difs [_] = []
29 > difs (i:jj@(j:js)) = j-i : difs jj
30 >
31 > difLists :: (Eq a, Num a) => [[a]] -> [[a]]
32 > difLists []          = []
33 > difLists xx@(xs:_) =
34 >   if isConst xs
35 >     then xx
36 >     else difLists $ difs xs : xx
37 >   where
38 >     isConst (i:jj@(j:_)) = all (==i) jj
39 >     isConst _ = error "difLists: lack of data, or not a
```

```
    ␣polynomial"
40  >
41  > -- This degree function is "strict", so only take
        finite list.
42  > degree' :: (Eq a, Num a) => [a] -> Int
43  > degree' xs = length (difLists [xs]) - 1
44  >
45  > -- This degree function can compute the degree of
        infinite list.
46  > degreeLazy :: (Eq a, Num a) => [a] -> Int
47  > degreeLazy xs = helper xs 0
48  >   where
49  >     helper as@(a:b:c:_) n
50  >       | a==b && b==c = n
51  >       | otherwise    = helper (difs as) (n+1)
52  >
53  > -- This is a hyblid version, safe and lazy.
54  > degree :: (Num a, Eq a) => [a] -> Int
55  > degree xs = let l = degreeLazy xs in
56  >   degree' $ take (l+2) xs
57
58  > -- m-times match version
59  > degreeTimes :: (Num a, Eq a) => Int -> [a] -> Int
60  > degreeTimes m xs = helper xs 0
61  >   where
62  >     helper aa@(a:as) n
63  >       | all (== a) (take (m-1) as) = n
64  >       | otherwise                  = helper (difs aa) (
        n+1)
65
66  Newton interpolation formula
67  First we introduce a new infix symbol for the operation
68  of taking a falling power.
69
70  > infixr 8 ^- -- falling power
71  > (^-) :: (Eq a, Num a) => a -> a -> a
72  > x ^- 0 = 1
73  > x ^- n = (x ^- (n-1)) * (x - n + 1)
74
75  Claim (Newton interpolation formula):
76    A polynomial f of degree n is expressed as
77      f(z) = \sum_{k=0}^n  (diff^n(f)(0)/k!) * (x ^- k)
78    where diff^n(f) is the n-th difference of f.
79
80  Example
```

```
Consider a polynomial f(x) = 2*x^3+3*x.

In general, we have no prior knowledge of this form,
but we know the sequences as a list of outputs (map f
    [0..]):

  Univariate> let f x = 2*x^3+3*x
  Univariate> take 10 $ map f [0..]
  [0,5,22,63,140,265,450,707,1048,1485]
  Univariate> degree $ take 10 $ map f [0..]
  3

Let us try to get differences:

  Univariate> difs $ take 10 $ map f [0..]
  [5,17,41,77,125,185,257,341,437]
  Univariate> difs it
  [12,24,36,48,60,72,84,96]
  Univariate> difs it
  [12,12,12,12,12,12,12]

Or more simply take difLists:

  Univariate> difLists [take 10 $ map f [0..]]
  [[12,12,12,12,12,12,12]
  ,[12,24,36,48,60,72,84,96]
  ,[5,17,41,77,125,185,257,341,437]
  ,[0,5,22,63,140,265,450,707,1048,1485]
  ]

What we need is the heads of above lists.

  Univariate> map head it
  [12,12,5,0]

Newton interpolation formula gives
  f' x = 0*(x ^- 0) `div` (0!) + 5*(x ^- 1) `div` (1!)
         + 12*(x ^- 2) `div` (2!) + 12*(x ^- 3) `div`
            (3!)
       = 5*(x ^- 1) + 6*(x ^- 2) + 2*(x ^- 3)
So

  Univariate> let f x = 2*x^3+3*x
  Univariate> let f' x = 5*(x ^- 1) + 6*(x ^- 2) + 2*(x
     ^- 3)
```

```
123   Univariate> take 10 $ map f [0..]
124   [0,5,22,63,140,265,450,707,1048,1485]
125   Univariate> take 10 $ map f' [0..]
126   [0,5,22,63,140,265,450,707,1048,1485]
127
128 Assume the differences are given in a list
129   [x_0, x_1 ..]
130 where x_k = diff^k(f)(0).
131 Then the implementation of the Newton interpolation
       formula is as follows:
132
133 > newtonC
134 >   :: (Fractional t, Enum t) =>
135 >      [t] -- first differences
136 >   -> [t] -- Newton coefficients
137 > newtonC xs = [x / factorial k | (x,k) <- zip xs [0..]]
138 >   where
139 >     factorial k = product [1..fromInteger k]
140
141   Univariate> let f x = 2*x^3+3*x
142   Univariate> take 10 $ map f [0..]
143   [0,5,22,63,140,265,450,707,1048,1485]
144   Univariate> difLists [it]
145   [[12,12,12,12,12,12,12]
146   ,[12,24,36,48,60,72,84,96]
147   ,[5,17,41,77,125,185,257,341,437]
148   ,[0,5,22,63,140,265,450,707,1048,1485]
149   ]
150   Univariate> reverse $ map head it
151   [0,5,12,12]
152   Univariate> newtonC it
153   [0 % 1,5 % 1,6 % 1,2 % 1]
154
155 The list of first differences can be computed as follows:
156
157 > firstDifs
158 >   :: (Eq a, Num a) =>
159 >      [a] -- map f [0..]
160 >   -> [a]
161 > firstDifs xs = reverse . map head . difLists $ [xs]
162
163 Mapping a list of integers to a Newton representation:
164
165 > -- This implementation can take infinite list.
166 > list2npol :: (Integral a) => [Ratio a] -> [Ratio a]
```

```
> list2npol xs = newtonC . firstDifs $ take n xs
>   where n = (degree xs) + 2
>
> -- m-times matches version
> list2npolTimes :: (Integral a) => Int -> [Ratio a] -> [Ratio a]
> list2npolTimes m xs = newtonC . firstDifs $ take n xs
>   where n = (degreeTimes m xs) + 2

  *Univariate> let f x = x*(x-1)*(x-2)*(x-3)*(x-4)*(x-5)
  *Univariate> let fs = map f [0..]
  *Univariate> list2npolTimes 10 fs
  [0 % 1,0 % 1,0 % 1,0 % 1,0 % 1,0 % 1,1 % 1]
  *Univariate> npol2pol it
  [0 % 1,(-120) % 1,274 % 1,(-225) % 1,85 % 1,(-15) % 1,1 % 1]
  *Univariate> list2npol fs
  [0 % 1]

We need to map Newton falling powers to standard powers.
This is a matter of applying combinatorics, by means of a convention formula
that uses the so-called Stirling cyclic numbers (of the first kind.)
Its defining relation is
  (x ^- n) = \sum_{k=1}^n (stirlingC n k) * (-1)^(n-k) * x^k.
The key equation is
  (x ^- n) = (x ^- (n-1)) * (x-n+1)
           = x*(x ^- (n-1)) - (n-1)*(x ^- (n-1))

Therefore, an implementation is as follows:

> stirlingC :: (Integral a) => a -> a -> a
> stirlingC 0 0 = 1
> stirlingC 0 _ = 0
> stirlingC n k = stirlingC (n-1) (k-1) + (n-1) * stirlingC (n-1) k

This definition can be used to convert from falling powers to standard powers.

> fall2pol :: (Integral a) => a -> [a]
> fall2pol 0 = [1]
> fall2pol n = 0   -- No constant term.
```

```
205 >            : [(-1)^(n-k) * stirlingC n k| k<-[1..n]]
206
207 We use this to convert Newton representations to standard
        polynomials
208 in coefficients list representation.
209 Here we have uses
210   sum
211 to collect same order terms in list representation.
212
213 > npol2pol :: (Ord t, Num t) => [t] -> [t]
214 > npol2pol xs = sum [ [x] * map fromInteger (fall2pol k)
215 >                    | (x,k) <- zip xs [0..]
216 >                    ]
217
218 Finally, here is the function for reconstruction the
       polynomial
219 from an output sequence:
220
221 > list2pol :: (Integral a) => [Ratio a] -> [Ratio a]
222 > list2pol = npol2pol . list2npol
223
224 Reconstruction as curve fitting
225
226   *Univariate> let f x = 2*x^3 + 3*x + 1%5
227   *Univariate> take 10 $ map f [0..]
228   [1%5, 26%5, 111%5, 316%5, 701%5, 1326%5, 2251%5,
         3536%5, 5241%5, 7426%5]
229   *Univariate> list2npol it
230   [1 % 5,5 % 1,6 % 1,2 % 1]
231   *Univariate> list2npol $ map f [0..]
232   [1 % 5,5 % 1,6 % 1,2 % 1]
233   *Univariate> list2pol $ map (\n -> 1%3 + (3%5)*n +
         (5%7)*n^2) [0..]
234   [1 % 3,3 % 5,5 % 7]
235   *Univariate>  list2pol [0,1,5,14,30,55]
236   [0 % 1,1 % 6,1 % 2,1 % 3]
237   *Univariate> map (p2fct $ list2pol [0,1,5,14,30,55])
         [0..6]
238   [0 % 1,1 % 1,5 % 1,14 % 1,30 % 1,55 % 1,91 % 1]
239
240 Here is n-times match version:
241
242 > list2polTimes :: Integral a => Int -> [Ratio a] -> [
       Ratio a]
243 > list2polTimes n = npol2pol . list2npolTimes n
```

```
244
245 --
246
247 Thiele's interpolation formula
248 https://rosettacode.org/wiki/Thiele%27
       s_interpolation_formula#Haskell
249 http://mathworld.wolfram.com/ThielesInterpolationFormula.
       html
250
251 reciprocal difference
252 Using the same notation of
253 https://rosettacode.org/wiki/Thiele%27
       s_interpolation_formula#C
254
255 > rho :: (Integral a) =>
256 >         [Ratio a] -- A list of output of f :: a -> Ratio
        a
257 >      -> a -> Int  -- "matrix"
258 >      -> Maybe (Ratio a) -- Nothing means 1/0 type
       infinity
259 > rho fs 0 i = Just $ fs !! i
260 > rho fs n i
261 >   | n < 0         = Just 0
262 >   | num == Just 0 = Nothing -- "infinity"
263 >   | otherwise     = (+) <$> recipro <*> rho fs (n-2) (i
       +1)
264 >    where
265 >      recipro = ((%) . (* n) <$> den) <*> num -- (den*n)%
       num
266 > --             (%) <$> (*n) <$> den <*> num -- functor
       law
267 >      num  = numerator <$> next
268 >      den  = denominator <$> next
269 > --    next = (-) <$> rho fs (n-1) (i+1) <*> rho fs (n
       -1) i
270 >      next = x `seq` y `seq` (-) <$> x <*> y
271 >        where x = rho fs (n-1) (i+1)
272 >              y = rho fs (n-1) i
273
274 Note that (%) has the following type,
275   (%) :: Integral a => a -> a -> Ratio a
276
277   *Univariate> (%) <$> (*2) <$> Just 5 <*> Just 3
278   Just (10 % 3)
279
```

```
280 The follwoing reciprocal differences match the table of
281 Milne-Thompson[1951] page 106:
282
283   *Univariate> map (\p -> map (rho (map (\t -> 1%(1+t^2))
          [0..]) p) [0..3]) [0..5]
284   [[Just (1 % 1)   ,Just (1 % 2)    ,Just (1 % 5)    ,
         Just (1 % 10)]
285   ,[Just ((-2) % 1),Just ((-10) % 3),Just ((-10) % 1),
         Just ((-170) % 7)]
286   ,[Just ((-1) % 1),Just ((-1) % 10),Just ((-1) % 25),
         Just ((-1) % 46)]
287   ,[Just (0 % 1)   ,Just (40 % 1)   ,Just (140 % 1)  ,
         Just (324 % 1)]
288   ,[Just (0 % 1)   ,Just (0 % 1)    ,Just (0 % 1)    ,
         Just (0 % 1)]
289   ,[Nothing,Nothing,Nothing,Nothing]
290   ]
291
292 > -- Thiele coefficients (continuous fraction)
293 > a :: (Integral a) => [Ratio a] -> a -> Maybe (Ratio a)
294 > a fs 0 = Just $ head fs
295 > a fs n = (-) <$> rho fs n 0 <*> rho fs (n-2) 0
296 >
297 > -- shifted Thiele coefficients
298 > a' :: Integral a => [Ratio a] -> Int -> a -> Maybe (
       Ratio a)
299 > a' fs p 0 = Just $ fs !! p
300 > a' fs p n = (-) <$> rho fs n p <*> rho fs (n-2) p
301
302   *Univariate> map (\p -> map (rho (map (\t -> t%(1+t^2))
          [0..]) p) [0..5]) [0..5]
303   [[Just (0%1),Just (1%2)    ,Just (2%5)    ,Just (3%10)
           ,Just (4%17)     ,Just (5%26)]
304   ,[Just (2%1),Just ((-10)%1),Just ((-10)%1),Just ((-170)
         %11),Just ((-442)%19),Just ((-962)%29)]
305   ,[Just (1%3),Nothing       ,Just ((-1)%15),Just ((-1)
         %48)  ,Just ((-1)%105) ,Just ((-1)%192)]
306   ,[Nothing   ,Nothing       ,Just (50%1)   ,Just (242%1)
           ,Just (662%1)    ,Just (1430%1)]
307   ,[Nothing   ,Nothing       ,Just (0%1)    ,Just (0%1)
            ,Just (0%1)      ,Just (0%1)]
308   ,[Nothing   ,Nothing       ,Nothing       ,Nothing
              ,Nothing         ,Nothing]
309   ]
310
```

```
311 Here, the consecutive Just ((-10) % 1) in second list
       make "fake" infinity (Nothing).
312
313   *Univariate> let f t = t%(1+t^2)
314   *Univariate> let fs = map f [0..]
315   *Univariate> let aMat = [map (a' fs i) [0..] | i <-
          [0..]]
316   *Univariate> take 20 $ map (length . takeWhile isJust)
          $ aMat
317   [3,2,5,5,5,5,5,5,5,5,5,5,5,5,5,5,5,5,5,5]
318
319 > -- Thiele coefficients with shifts.
320 > aMatrix :: Integral a => [Ratio a] -> [[Maybe (Ratio a)
       ]]
321 > aMatrix fs = [map (a' fs i) [0..] | i <- [0..]]
322 >
323 > tDegree :: Integral a => [Ratio a] -> Int
324 > tDegree = isConsts' 3 . map (length . takeWhile isJust)
        . aMatrix
325 >
326 > -- To find constant sub sequence.
327 > isConsts' :: Eq t => Int -> [t] -> t
328 > isConsts' n (l:ls)
329 >   | all (==l) $ take (n-1) ls = l
330 >   | otherwise                 = isConsts' n ls
331
332 we also need the shift, in this case, p=2 to get full
       Thiele coefficients.
333
334 > shiftaMatrix
335 >   :: Integral a =>
336 >      [Ratio a] -> [Maybe [Ratio a]]
337 > shiftaMatrix gs = map (sequence . (\q -> map (a' gs q)
       [0..(thieleD-1)])) [0..]
338 >   where
339 >     thieleD = fromIntegral $ tDegree gs
340 >
341 > shiftAndThieleC
342 >   :: Integral a =>
343 >      [Ratio a] -> (Maybe Int, Maybe [Ratio a])
344 > shiftAndThieleC fs = (findIndex isJust gs, join $ find
       isJust gs)
345 >   where
346 >     gs = shiftaMatrix fs
347
```

```
348   *Univariate> take 10 $ map sequence $ transpose $ take
         (tDegree fs) m
349   [Just [0 % 1,1 % 2,2 % 5,3 % 10,4 % 17]
350   ,Just [2 % 1,(-10) % 1,(-10) % 1,(-170) % 11,(-442) %
         19]
351   ,Nothing,Nothing,Nothing,Nothing,Nothing,Nothing,
         Nothing,Nothing]
352
353 Packed version, this scans the given data only once.
354
355 > degSftTC
356 >   :: Integral a =>
357 >      [Ratio a] -> (Int, Maybe Int, Maybe (Maybe [Ratio
      a]))
358 > --                  |    |          + Thiele
      coefficients
359 > --                  |    + shift
360 > --                  + degree
361 > degSftTC fs = (d,s,ts)
362 >   where
363 >     m = [map (a' fs i) [0..] | i <- [0..]]
364 >     d = isConsts' 3 . map (length . takeWhile isJust) $
      m -- 3 times match
365 >     m' = map (sequence . take d) m
366 >     s = findIndex isJust m'
367 >     ts = find isJust m'
368
369   *Univariate Control.Monad> let g t = t%(1+t^2)
370   *Univariate Control.Monad> let gs = map g [0..]
371   *Univariate Control.Monad> shiftAndThieleC $
         shiftaMatrix gs
372   (Just 2,Just [2 % 5,(-10) % 1,(-7) % 15,60 % 1,1 % 15])
373   *Univariate Control.Monad> let f t = 1%(1+t^2)
374   *Univariate Control.Monad> let fs = map f [0..]
375   *Univariate Control.Monad> shiftAndThieleC $
         shiftaMatrix fs
376   (Just 0,Just [1 % 1,(-2) % 1,(-2) % 1,2 % 1,1 % 1])
377
378 We need a convertor from this thiele sequence to
      continuous fractional form of rational function.
379
380 > nextStep [a0,a1] (v:_)  = a0 + v/a1
381 > nextStep (a:as)  (v:vs) = a + (v / nextStep as vs)
382 >
383 > -- From thiele sequence to (rational) function.
```

```
> thiele2ratf :: Integral a => [Ratio a] -> (Ratio a ->
    Ratio a)
> thiele2ratf as x
>   | x == 0    = head as -- only constant term
>   | otherwise = nextStep as [x,x-1 ..]

  *Univariate> let h t = (3+6*t+18*t^2)%(1+2*t+20*t^2)
  *Univariate> let hs = map h [0..]
  *Univariate> let as = thieleC hs
  *Univariate> as
  [3 % 1,(-23) % 42,(-28) % 13,767 % 14,7 % 130]
  *Univariate> let th x = thiele2ratf as x
  *Univariate> take 5 hs
  [3 % 1,27 % 23,87 % 85,183 % 187,45 % 47]
  *Univariate> map th [0..5]
  [3 % 1,27 % 23,87 % 85,183 % 187,45 % 47,69 % 73]

We represent a rational function by a tuple of
    coefficient lists:
  (ns,ds) :: ([Ratio Int],[Ratio Int])
where ns and ds are coef-list-rep of numerator polynomial
     and denominator polynomial.
Here is a translator from coefficients lists to rational
    function.

> -- similar to p2fct
> lists2ratf :: (Integral a) =>
>               ([Ratio a],[Ratio a]) -> (Ratio a ->
    Ratio a)
> lists2ratf (ns,ds) x = p2fct ns x / p2fct ds x

  *Univariate> let frac x = lists2ratf
     ([1,1%2,1%3],[2,2%3]) x
  *Univariate> take 10 $ map frac [0..]
  [1 % 2,11 % 16,1 % 1,11 % 8,25 % 14,71 % 32,8 % 3,25 %
     8,79 % 22,65 % 16]
  *Univariate> let ffrac x = (1+(1%2)*x+(1%3)*x^2)
     /(2+(2%3)*x)
  *Univariate> take 10 $ map ffrac [0..]
  [1 % 2,11 % 16,1 % 1,11 % 8,25 % 14,71 % 32,8 % 3,25 %
     8,79 % 22,65 % 16]

The following canonicalizer reduces the tuple-rep of
    rational function in canonical form
That is, the coefficien of the lowest degree term of the
```

```
    denominator to be 1.
However, since our input starts from 0 and this means
    firstNonzero is the same as head.

> canonicalize :: (Integral a) => ([Ratio a],[Ratio a])
    -> ([Ratio a],[Ratio a])
> canonicalize rat@(ns,ds)
>   | dMin == 1 = rat
>   | otherwise = (map (/ dMin) ns, map (/ dMin) ds)
>   where
>     dMin = firstNonzero ds
>     firstNonzero [a] = a -- head
>     firstNonzero (a:as)
>       | a /= 0    = a
>       | otherwise = firstNonzero as

What we need is a translator from Thiele coefficients to
    this tuple-rep.

> thiele2coef :: (Integral a) => [Ratio a] -> ([Ratio a
    ],[Ratio a])
> thiele2coef as = canonicalize $ t2r as 0
>   where
>     t2r [an,an'] n = ([an*an'-n,1],[an'])
>     t2r (a:as)   n = ((a .* num) + ([-n,1] * den), num)
>       where
>         (num, den) = t2r as (n+1)
>

  *Univariate> let h t = (3+6*t+18*t^2)%(1+2*t+20*t^2)
  *Univariate> let hs = map h [0..]
  *Univariate> take 5 hs
  [3 % 1,27 % 23,87 % 85,183 % 187,45 % 47]
  *Univariate> let th x = thiele2ratf as x
  *Univariate> map th [0..5]
  [3 % 1,27 % 23,87 % 85,183 % 187,45 % 47,69 % 73]
  *Univariate> as
  [3 % 1,(-23) % 42,(-28) % 13,767 % 14,7 % 130]
  *Univariate> thiele2coef as
  ([3 % 1,6 % 1,18 % 1],[1 % 1,2 % 1,20 % 1])

> thiele2coef' -- shifted version (0 -> sft)
>   :: Integral a =>
>      Ratio a -> [Ratio a] -> ([Ratio a], [Ratio a])
> thiele2coef' sft [a] = ([a],1)
```

```
459 > thiele2coef' sft as = canonicalize $ t2r as sft
460 >   where
461 >     t2r [an,an'] n = (([an*an'-n] + z),[an'])
462 >     t2r (a:as)   n = ((a .* num) + ((z - [n]) * den),
      num)
463 >       where
464 >         (num, den) = t2r as (n+1)
465
466   *Univariate> let f x = x^2%(1+x^2)
467   *Univariate> shiftAndThieleC $ map f [0..]
468   (Just 0,Just [0 % 1,2 % 1,2 % 1,(-2) % 1,(-1) % 1])
469   *Univariate> take 3 $ shiftaMatrix (map f [0..])
470   [Just [0 % 1,2 % 1,2 % 1,(-2) % 1,(-1) % 1]
471   ,Just [1 % 2,10 % 3,3 % 5,(-130) % 3,(-1) % 10]
472   ,Just [4 % 5,10 % 1,6 % 25,(-150) % 1,(-1) % 25]
473   ]
474   *Univariate> thiele2coef' 0 [0 % 1,2 % 1,2 % 1,(-2) %
        1,(-1) % 1]
475   ([0 % 1,0 % 1,1 % 1],[1 % 1,0 % 1,1 % 1])
476   *Univariate> thiele2coef' 1 [1 % 2,10 % 3,3 % 5,(-130)
        % 3,(-1) % 10]
477   ([0 % 1,0 % 1,1 % 1],[1 % 1,0 % 1,1 % 1])
478
479 > shiftAndThiele2coef (Just sft, Just ts) = Just $
      thiele2coef' (fromIntegral sft) ts
480 > shiftAndThiele2coef _                   = Nothing
481 >
482 > list2rat' :: (Integral a) => [Ratio a] -> Maybe ([Ratio
      a], [Ratio a])
483 > list2rat' = shiftAndThiele2coef . shiftAndThieleC
484 >
485 > list2rat'' lst = let (_,s,ts) = degSftTC lst in
486 >   shiftAndThiele2coef (s, join ts)
487
488   *Univariate> let f t = t%(1+t^2)
489   *Univariate> let fs = map (\t -> 1%(1+t^2)) [0..]
490   *Univariate> list2rat' fs
491   Just ([1 % 1,0 % 1,0 % 1],[1 % 1,0 % 1,1 % 1])
492   *Univariate> shiftAndThieleC fs
493   (Just 0,Just [1 % 1,(-2) % 1,(-2) % 1,2 % 1,1 % 1])
494   *Univariate> let fs = map (\t -> t%(1+t^2)) [0..]
495   *Univariate> let gs = map (\t -> t%(1+t^2)) [0..]
496   *Univariate> shiftAndThieleC gs
497   (Just 2,Just [2 % 5,(-10) % 1,(-7) % 15,60 % 1,1 % 15])
498   *Univariate> list2rat' gs
```

```
499   Just ([0 % 1,1 % 1,0 % 1],[1 % 1,0 % 1,1 % 1])
500   *Univariate> let hs = map (\t -> t^2%(1+t^2)) [0..]
501   *Univariate> shiftAndThieleC hs
502   (Just 0,Just [0 % 1,2 % 1,2 % 1,(-2) % 1,(-1) % 1])
503   *Univariate> list2rat' hs
504   Just ([0 % 1,0 % 1,1 % 1],[1 % 1,0 % 1,1 % 1])
505
506   *Univariate> let f t = t%(1+t^2)
507   *Univariate> let fs = map f [0..]
508   *Univariate> let aMat = [map (\j -> a' fs j i) [0..] |
          i <- [0..]]
509   *Univariate> take 6 $ map (take 5) aMat
510   [[Just (0 % 1),Just (1 % 2),Just (2 % 5),Just (3 % 10),
          Just (4 % 17)]
511   ,[Just (2 % 1),Just ((-10) % 1),Just ((-10) % 1),Just
          ((-170) % 11),Just ((-442) % 19)]
512   ,[Just (1 % 3),Nothing,Just ((-7) % 15),Just ((-77) %
          240),Just ((-437) % 1785)]
513   ,[Nothing,Nothing,Just (60 % 1),Just (2832 % 11),Just
          (13020 % 19)]
514   ,[Nothing,Nothing,Just (1 % 15),Just (1 % 48),Just (1 %
           105)]
515   ,[Nothing,Nothing,Nothing,Nothing,Nothing]
516   ]
```

4.4 Multivariate.lhs

Listing 4.4: Multivariate.lhs

```
1  Multivariate.lhs
2
3  > module Multivariate where
4
5  > import Data.Ratio
6  > import Data.List (transpose)
7  > import Univariate -- ( list2pol, list2npolTimes
8  >                   -- , tDegree, list2rat'
9  >                   -- )
10
11 Let us start 2-variate polynomials.
12 First, make a naive grid.
13
14   *Multivariate> let f z1 z2 = 3+2*z1+4*z2+7*z1^2+5*z1*z2
         +6*z2^2
15   *Multivariate> [[f x y | y <- [0..9]] | x <- [0..9]]
```

```
  [[3,13,5,69,115,173,243,325,419,525]
  ,[12,27,54,93,144,207,282,369,468,579]
  ,[35,55,87,131,187,255,335,427,531,647]
  ,[72,97,134,183,244,317,402,499,608,729]
  ,[123,153,195,249,315,393,483,585,699,825]
  ,[188,223,270,329,400,483,578,685,804,935]
  ,[267,37,359,423,499,587,687,799,923,1059]
  ,[360,405,462,531,612,705,810,927,1056,1197]
  ,[467,517,579,653,739,837,947,1069,1203,1349]
  ,[588,643,710,789,880,983,1098,1225,1364,1515]
  ]

Assuming the list of lists is a matrix of 2-variate
   function's values,
  f i j

> tablize
>  :: (Enum t1, Num t1) =>
>     (t1 -> t1 -> t) -> Int -> [[t]]
> tablize f n = [[f x y | y <- range] | x <- range]
>   where
>     range = take n [0..]

  *Multivariate> tablize (\x y -> (x,y)) 4
  [[(0,0),(0,1),(0,2),(0,3)]
  ,[(1,0),(1,1),(1,2),(1,3)]
  ,[(2,0),(2,1),(2,2),(2,3)]
  ,[(3,0),(3,1),(3,2),(3,3)]
  ]

  *Multivariate> let fTable = tablize f 10
  *Multivariate> map list2pol fTable
  [[3 % 1,4 % 1,6 % 1]
  ,[12 % 1,9 % 1,6 % 1]
  ,[35 % 1,14 % 1,6 % 1]
  ,[72 % 1,19 % 1,6 % 1]
  ,[123 % 1,24 % 1,6 % 1]
  ,[188 % 1,29 % 1,6 % 1]
  ,[267 % 1,34 % 1,6 % 1]
  ,[360 % 1,39 % 1,6 % 1]
  ,[467 % 1,44 % 1,6 % 1]
  ,[588 % 1,49 % 1,6 % 1]
  ]

Let us take the transpose of this "matrix" to see the
```

```
    behavior of coefficients.

  *Multivariate> let f z1 z2 = 3+2*z1+4*z2+7*z1^2+5*z1*z2
     +6*z2^2
  *Multivariate> let fTable = tablize f 10
  *Multivariate> map list2pol fTable
  [[3 % 1,4 % 1,6 % 1]
  ,[2 % 1,9 % 1,6 % 1]
  ,[35 % 1,14 % 1,6 % 1]
  ,[72 % 1,19 % 1,6 % 1]
  ,[123 % 1,24 % 1,6 % 1]
  ,[188 % 1,29 % 1,6 % 1]
  ,[267 % 1,34 % 1,6 % 1]
  ,[360 % 1,39 % 1,6 % 1]
  ,[467 % 1,44 % 1,6 % 1]
  ,[588 % 1,49 % 1,6 % 1]
  ]
  *ultivariate> transpose it
  [[3 % 1,12 % 1,35 % 1,72 % 1,123 % 1,188 % 1,267 %
     1,360 % 1,467 % 1,588 % 1]
  [4 % 1,9 % 1,14 % 1,19 % 1,24 % 1,29 % 1,34 % 1,39 %
     1,44 % 1,49 % 1]
  ,[6 % 1,6 % 1,6 % 1,6 % 1,6 % 1,6 % 1,6 % 1,6 % 1,6 %
     1,6 % 1]
  ]
  *Multivariate> map list2pol it
  [[3 % 1,2 % 1,7 % 1]
  ,[4 % 1,5 % 1]
  ,[6 % 1]]

> table2pol :: [[Ratio Integer]] -> [[Ratio Integer]]
> table2pol = map list2pol . transpose . map list2pol

  *Multivariate> let g x y = 1+7*x + 8*y + 10*x^2 + x*y
     +9*y^2
  *Multivariate> table2pol $ tablize g 5
  [[1 % 1,7 % 1,10 % 1],[8 % 1,1 % 1],[9 % 1]]

There are some bad-behavior polynomials;
  *Multivariate> table2pol $ tablize (\x y -> x*y) 20
  [[0 % 1],[1 % 1,1 % 1]]
  *Multivariate> tablize (\x y -> (x,y)) 5
  [[(0,0),(0,1),(0,2),(0,3),(0,4)]
  ,[(1,0),(1,1),(1,2),(1,3),(1,4)]
  ,[(2,0),(2,1),(2,2),(2,3),(2,4)]
```

```
  ,[(3,0),(3,1),(3,2),(3,3),(3,4)]
  ,[(4,0),(4,1),(4,2),(4,3),(4,4)]
  ]
  *Multivariate> tablize (\x y -> (x*y)) 5
  [[0,0,0,0,0]
  ,[0,1,2,3,4]
  ,[0,2,4,6,8]
  ,[0,3,6,9,12]
  ,[0,4,8,12,16]
  ]

Here we have assumed that the list of functions has the
    same length, but

  *Multivariate> map list2pol $ tablize (\x y -> x*y) 5
  [[0 % 1],[0 % 1,1 % 1],[0 % 1,2 % 1],[0 % 1,3 % 1],[0 %
      1,4 % 1]]

So, we should repeat 0's if we have zero-function.

> xyDegree f = (dX, dY)
>   where
>     dX = length . list2pol $ map (\t -> f t 1) [0..]
>     dY = length . list2pol $ map (\t -> f 1 t) [0..]

  *Multivariate> let test x y = x^2*(2*y + y^3)
  *Multivariate> uncurry (*) . xyDegree $ test
  6
  *Multivariate> maximum . map (length . list2pol) .
     tablize test $ 6
  4
  *Multivariate> map (take 4 . (++ (repeat (0%1))) .
     list2pol) . tablize test $ 6
  [[0 % 1,0 % 1,0 % 1,0 % 1]
  ,[0 % 1,2 % 1,0 % 1,1 % 1]
  ,[0 % 1,8 % 1,0 % 1,4 % 1]
  ,[0 % 1,18 % 1,0 % 1,9 % 1]
  ,[0 % 1,32 % 1,0 % 1,16 % 1]
  ,[0 % 1,50 % 1,0 % 1,25 % 1]
  ]
  *Multivariate> map list2pol . transpose $ it
  [[0 % 1],[0 % 1,0 % 1,2 % 1],[0 % 1],[0 % 1,0 % 1,1 %
     1]]

  *Multivariate> let test x y = (1%3)*x^2*((2%5)*y +
```

```
      ((3%4)*x*y^3))
139                                   -- = (2%15)*x^2*y + (1%4)
                                         *x^3*y^3
140   *Multivariate> xyDegree test
141   (3,3)
142   *Multivariate> map (take 4 . (++ (repeat (0%1))) .
         list2pol) . tablize test $ 9
143   [[0 % 1,0 % 1,0 % 1,0 % 1]
144   ,[0 % 1,2 % 15,0 % 1,1 % 4]
145   ,[0 % 1,8 % 15,0 % 1,2 % 1]
146   ,[0 % 1,6 % 5,0 % 1,27 % 4]
147   ,[0 % 1,32 % 15,0 % 1,16 % 1]
148   ,[0 % 1,10 % 3,0 % 1,125 % 4]
149   ,[0 % 1,24 % 5,0 % 1,54 % 1]
150   ,[0 % 1,98 % 15,0 % 1,343 % 4]
151   ,[0 % 1,128 % 15,0 % 1,128 % 1]
152   ]
153   *Multivariate> map list2pol . transpose $ it
154   [[0 % 1],[0 % 1,0 % 1,2 % 15],[0 % 1],[0 % 1,0 % 1,0 %
         1,1 % 4]]
155
156 > xyPol2Coef :: (Enum t, Integral a, Num t) =>
157 >                (t -> t -> Ratio a) -> [[Ratio a]]
158 > xyPol2Coef f = map list2pol . transpose . map (take num
       . (++ (repeat (0%1))) . list2pol) . tablize f $ rank
159 >   where
160 >     rank = uncurry (*) . xyDegree $ f
161 >     num  = maximum . map (length . list2pol) . tablize
      f $ rank
162
163   *Multivariate> let test x y = (1%3)*x^2*((2%5)*y +
         ((3%4)*x*y^3))
164   *Multivariate> xyPol2Coef test
165   [[0 % 1],[0 % 1,0 % 1,2 % 15],[0 % 1],[0 % 1,0 % 1,0 %
         1,1 % 4]]
166   *Multivariate> let test2 x y = x*y
167   *Multivariate> xyPol2Coef test2
168   [[0 % 1],[0 % 1,1 % 1]]
169   *Multivariate> let test3 x y = x^3*y^4
170   *Multivariate> xyPol2Coef test3
171   [[0 % 1],[0 % 1],[0 % 1],[0 % 1],[0 % 1,0 % 1,0 % 1,1 %
          1]]
172
173 > table2pol' :: Integral a => [[Ratio a]] -> [[Ratio a]]
174 > table2pol' tbl = map list2pol . transpose . map (take
```

```
       num . (++ (repeat (0%1))) . list2pol) $ tbl
>    where
>      num = maximum . map (length . list2pol) $ tbl

--

2-variable rational functions

  *Univariate> let h x y = (1+2*x+4*y+7*x^2+5*x*y+6*y^2)
     % (1+7*x+8*y+10*x^2+x*y+9*y^2)
  *Univariate> let auxh x y t = h (t*x) (t*y)
  *Univariate> let auxhs = [map (auxh 1 y) [0..] | y <-
     [0..]]
  *Univariate> take 10 $ map list2rat' auxhs
  [Just ([1 % 1,2 % 1,7 % 1],[1 % 1,7 % 1,10 % 1])
  ,Just ([1 % 1,6 % 1,18 % 1],[1 % 1,15 % 1,20 % 1])
  ,Just ([1 % 1,10 % 1,41 % 1],[1 % 1,23 % 1,48 % 1])
  ,Just ([1 % 1,14 % 1,76 % 1],[1 % 1,31 % 1,94 % 1])
  ,Just ([1 % 1,18 % 1,123 % 1],[1 % 1,39 % 1,158 % 1])
  ,Just ([1 % 1,22 % 1,182 % 1],[1 % 1,47 % 1,240 % 1])
  ,Just ([1 % 1,26 % 1,253 % 1],[1 % 1,55 % 1,340 % 1])
  ,Just ([1 % 1,30 % 1,336 % 1],[1 % 1,63 % 1,458 % 1])
  ,Just ([1 % 1,34 % 1,431 % 1],[1 % 1,71 % 1,594 % 1])
  ,Just ([1 % 1,38 % 1,538 % 1],[1 % 1,79 % 1,748 % 1])
  ]
  *Univariate> sequence it
  Just [([1 % 1,2 % 1,7 % 1],[1 % 1,7 % 1,10 % 1])
       ,([1 % 1,6 % 1,18 % 1],[1 % 1,15 % 1,20 % 1])
       ,([1 % 1,10 % 1,41 % 1],[1 % 1,23 % 1,48 % 1])
       ,([1 % 1,14 % 1,76 % 1],[1 % 1,31 % 1,94 % 1])
       ,([1 % 1,18 % 1,123 % 1],[1 % 1,39 % 1,158 % 1])
       ,([1 % 1,22 % 1,182 % 1],[1 % 1,47 % 1,240 % 1])
       ,([1 % 1,26 % 1,253 % 1],[1 % 1,55 % 1,340 % 1])
       ,([1 % 1,30 % 1,336 % 1],[1 % 1,63 % 1,458 % 1])
       ,([1 % 1,34 % 1,431 % 1],[1 % 1,71 % 1,594 % 1])
       ,([1 % 1,38 % 1,538 % 1],[1 % 1,79 % 1,748 % 1])
       ]
  *Univariate> fmap (map fst) it
  Just [[1 % 1,2 % 1,7 % 1]
       ,[1 % 1,6 % 1,18 % 1]
       ,[1 % 1,10 % 1,41 % 1]
       ,[1 % 1,14 % 1,76 % 1]
       ,[1 % 1,18 % 1,123 % 1]
       , ..
       ]
```

```
217    *Univariate> fmap transpose it
218    Just [[1 % 1,1 % 1,1 % 1,1 % 1,1 % 1,1 % 1,1 % 1,1 % 1,1 %
          1,1 % 1,1 % 1]
219          ,[2 % 1,6 % 1,10 % 1,14 % 1,18 % 1,22 % 1,26 %
             1,30 % 1,34 % 1,38 % 1]
220          ,[7 % 1,18 % 1,41 % 1,76 % 1,123 % 1,182 % 1,253 %
              1,336 % 1,431 % 1,538 % 1]
221          ]
222    *Univariate> fmap (map list2pol) it
223    Just [[1 % 1],[2 % 1,4 % 1],[7 % 1,5 % 1,6 % 1]]
224
225    *Multivariate> let h x y = (1+2*x+4*y+7*x^2+5*x*y
          +(6%13)*y^2) / (1+(7%3)*x+8*y+10*x^2+x*y+9*y^2)
226    *Multivariate> let auxh x y t = h (t*x) (t*y)
227    *Multivariate> let auxhs = [map (auxh 1 y) [0..100] | y
           <- [0..100]]
228    *Multivariate> fmap (map list2pol . transpose . map fst
          ) . sequence . map list2rat' $ auxhs
229    Just [[1 % 1],[2 % 1,4 % 1],[7 % 1,5 % 1,6 % 13]]
230    *Multivariate> fmap (map list2pol . transpose . map snd
          ) . sequence . map list2rat' $ auxhs
231    Just [[1 % 1],[7 % 3,8 % 1],[10 % 1,1 % 1,9 % 1]]
232
233  > -- SUPER SLOW IMPLEMENTATION, DO NOT USE THIS!
234  > table2ratf
235  >   :: Integral a =>
236  >      [[Ratio a]] -> (Maybe [[Ratio a]], Maybe [[Ratio a
        ]])
237  > table2ratf table = (t2r fst table, t2r snd table)
238  >   where
239  > --    t2r' third = fmap (map third) . sequence . map
        list2rat'
240  >
241  >     myMax Nothing   = 0
242  >     myMax (Just ns) = ns
243  >
244  >     t2r third = fmap (map list2pol . transpose . map (
        take num . (++ (repeat (0%1))) . third)) .
245  >                 mapM list2rat'
246  >       where
247  >         num = myMax . fmap (maximum . map (length . fst
        )) . mapM list2rat' $ table
248  >
249  > -- fmap (maximum . map (length . fst)) . sequence . map
        list2rat'
```

```
> -- map (take num . (++ (repeat (0%1))) . list2pol)
>
> tablizer :: (Num a, Enum a) => (a -> a -> b) -> a -> [[
    b]]
> tablizer f n = [map (f_t 1 y) [0..(n-1)] | y <- [0..(n
    -1)]]
>   where
>     f_t x y t = f (t*x) (t*y)

  *Multivariate> let h x y = (1+2*x+4*y+7*x^2+5*x*y
     +(6%13)*y^2) / (1+(7%3)*x+8*y+10*x^2+x*y+9*y^2)
  *Multivariate> let hTable = tablizer h 20
  *Multivariate> table2ratf hTable
  (Just [[1 % 1],[2 % 1,4 % 1],[7 % 1,5 % 1,6 % 13]],Just
      [[1 % 1],[7 % 3,8 % 1],[10 % 1,1 % 1,9 % 1]])

Note that, the sampling points for n=10 case are

  *Multivariate> tablizer (\x y -> (x,y)) 10
  [[(0,0),(1,0),(2,0) ,(3,0) ,(4,0) ,(5,0) ,(6,0) ,(7,0)
     ,(8,0) ,(9,0)]
  ,[(0,0),(1,1),(2,2) ,(3,3) ,(4,4) ,(5,5) ,(6,6) ,(7,7)
     ,(8,8) ,(9,9)]
  ,[(0,0),(1,2),(2,4) ,(3,6) ,(4,8) ,(5,10),(6,12),(7,14)
     ,(8,16),(9,18)]
  ,[(0,0),(1,3),(2,6) ,(3,9) ,(4,12),(5,15),(6,18),(7,21)
     ,(8,24),(9,27)]
  ,[(0,0),(1,4),(2,8) ,(3,12),(4,16),(5,20),(6,24),(7,28)
     ,(8,32),(9,36)]
  ,[(0,0),(1,5),(2,10),(3,15),(4,20),(5,25),(6,30),(7,35)
     ,(8,40),(9,45)]
  ,[(0,0),(1,6),(2,12),(3,18),(4,24),(5,30),(6,36),(7,42)
     ,(8,48),(9,54)]
  ,[(0,0),(1,7),(2,14),(3,21),(4,28),(5,35),(6,42),(7,49)
     ,(8,56),(9,63)]
  ,[(0,0),(1,8),(2,16),(3,24),(4,32),(5,40),(6,48),(7,56)
     ,(8,64),(9,72)]
  ,[(0,0),(1,9),(2,18),(3,27),(4,36),(5,45),(6,54),(7,63)
     ,(8,72),(9,81)]
  ]

  *Multivariate> let f x y = (1 + 2*x + 3*y + 4*x^2 +
     (1%5)*x*y + (1%6)*y^2)
                           / (7 + 8*x + (1%9)*y + x^2 + x
                              *y + 10*y^2)
```

```
279   *Multivariate> let g x y = (11 + 10*x + 9*y) / (8 + 7*x
          ^2 + (1%6)*x*y + 5*y^2)
280   *Multivariate> table2ratf $ tablizer f 20
281   (Just [[1 % 7],[2 % 7,3 % 7],[4 % 7,1 % 35,1 % 42]]
282   ,Just [[1 % 1],[8 % 7,1 % 63],[1 % 7,1 % 7,10 % 7]])
283   *Multivariate> table2ratf $ tablizer g 20
284   (Just [[11 % 8],[5 % 4,9 % 8],[0 % 1]]
285   ,Just [[1 % 1],[0 % 1],[7 % 8,1 % 48,5 % 8]])
286   *Multivariate> let h x y = (f x y) / (g x y)
287   *Multivariate> table2ratf $ tablizer h 20
288   (Just [[8 % 77],[16 % 77,24 % 77],[39 % 77,53 % 2310,19
          % 231]
289            ,[2 % 11,64 % 231,3 % 22,15 % 77],[4 % 11,31 %
               1155,106 % 385,37 % 2772,5 % 462]]
290   ,Just [[1 % 1],[158 % 77,578 % 693],[13 % 11,757 %
          693,111 % 77]
291            ,[10 % 77,19 % 77,109 % 77,90 % 77]]
292   )
293   *Multivariate> table2ratf $ tablizer f 10
294   (*** Exception: Prelude.!!: index too large
295   *Multivariate> table2ratf $ tablizer f 13
296   (*** Exception: Prelude.!!: index too large
297   *Multivariate> table2ratf $ tablizer f 15
298   (Just [[1 % 7],[2 % 7,3 % 7],[4 % 7,1 % 35,1 % 42]]
299   ,Just [[1 % 1],[8 % 7,1 % 63],[1 % 7,1 % 7,10 % 7]])
300   *Multivariate> table2ratf $ tablizer g 11
301   (*** Exception: Prelude.!!: index too large
302   *Multivariate> table2ratf $ tablizer g 13
303   (*** Exception: Prelude.!!: index too large
304   *Multivariate> table2ratf $ tablizer g 15
305   (Just [[11 % 8],[5 % 4,9 % 8],[0 % 1]],Just [[1 % 1],[0
          % 1],[7 % 8,1 % 48,5 % 8]])
306   *Multivariate> table2ratf $ tablizer (\x y -> x/(1+x^2)
          ) 10
307   (*** Exception: Prelude.!!: index too large
308   *Multivariate> table2ratf $ tablizer (\x y -> x/(1+x^2)
          ) 13
309   (*** Exception: Prelude.!!: index too large
310   *Multivariate> table2ratf $ tablizer (\x y -> x/(1+x^2)
          ) 20
311   (Just [[0 % 1],[1 % 1],[0 % 1]],Just [[1 % 1],[0 %
          1],[1 % 1]])
312   *Multivariate> table2ratf $ tablizer (\x y -> x/(1+x^2)
          ) 13
313   (*** Exception: Prelude.!!: index too large
```

```
314   *Multivariate> table2ratf $ tablizer (\x y -> x/(1+x^2)
         ) 15
315   (Just [[0 % 1],[1 % 1],[0 % 1]],Just [[1 % 1],[0 %
         1],[1 % 1]])
316
317 > -- Alternative transpose, filling with the default
       value.
318 > -- I basically followed the implementation of standard
       Prelude.
319 > transposeWith :: a -> [[a]] -> [[a]]
320 > transposeWith _ [] = []
321 > transposeWith z ([] : xss)
322 >   | all null xss = []
323 >   | otherwise    = (z : [h | (h:_) <- xss])
324 >                  : transposeWith z ([] : [t | (_:t) <-
       xss])
325 > transposeWith z ((x:xs) : xss) = (x : [h | (h:_) <- xss
       ])
326 >                                 : transposeWith z (xs :
       [t | (_:t) <- xss])
327
328   *Multivariate> let f x y = (x*y)%(1+y)^2
329   *Multivariate> let tbl = tablizer f 20
330   *Multivariate> fmap (map list2pol . (transposeWith
         (0%1)) . map fst) . sequence . map list2rat' $ tbl
331   Just [[0 % 1],[0 % 1],[0 % 1,1 % 1]]
332   *Multivariate> fmap (map list2pol . (transposeWith
         (0%1)) . map snd) . sequence . map list2rat' $ tbl
333   Just [[1 % 1],[0 % 1,2 % 1],[0 % 1,0 % 1,1 % 1]]
334
335 > table2ratf' table = (t2r fst table, t2r snd table)
336 >   where
337 >     t2r third = fmap (map list2pol . transposeWith
       (0%1) . map third) . mapM list2rat'
338 >
339 > table2ratf'' table = (t2r fst table, t2r snd table)
340 >   where
341 >     t2r third = fmap (map list2pol . transposeWith
       (0%1) . map third) . mapM list2rat''
342
343 > wilFunc x y = (x^2*y^2) % ((1 + y)^3)
344
345   *Multivariate> table2ratf $ tablizer wilFunc 20
346   (Just [[0 % 1],[0 % 1],[0 % 1],[0 % 1],[0 % 1,0 % 1,1 %
         1]]
```

```
347   ,Just [[1 % 1],[0 % 1,3 % 1],[0 % 1,0 % 1,3 % 1],[0 %
         1,0 % 1,0 % 1,1 % 1],[0 % 1]])
348   (3.91 secs, 2,850,226,792 bytes)
349   *Multivariate> table2ratf' $ tablizer wilFunc 20
350   (Just [[0 % 1],[0 % 1],[0 % 1],[0 % 1],[0 % 1,0 % 1,1 %
          1]]
351   ,Just [[1 % 1],[0 % 1,3 % 1],[0 % 1,0 % 1,3 % 1],[0 %
         1,0 % 1,0 % 1,1 % 1]])
352   (2.00 secs, 1,425,753,744 bytes)
353   *Multivariate> table2ratf'' $ tablizer wilFunc 20
354   (Just [[0 % 1],[0 % 1],[0 % 1],[0 % 1],[0 % 1,0 % 1,1 %
          1]]
355   ,Just [[1 % 1],[0 % 1,3 % 1],[0 % 1,0 % 1,3 % 1],[0 %
         1,0 % 1,0 % 1,1 % 1]])
356   (1.73 secs, 1,234,282,424 bytes)
```

4.5 FROverZp.lhs

Listing 4.5: FROverZp.lhs

```
1  FROverZp.lhs
2
3  > module FROverZp where
4
5  Functional Reconstruction over finite field Z_p
6
7  > import Data.Ratio
8  > import Data.Maybe
9  > import Data.Numbers.Primes
10 > import Data.List (null, transpose)
11 > import Control.Monad (sequence)
12 >
13 > import Ffield (modp, bigPrimes, reconstruct)
14 > -- , inversep, bigPrimes, recCRT, recCRT')
15 > import Univariate (newtonC, firstDifs, list2npol)
16
17 Univariate Polynomial case
18 Our target is a univariate polynomial
19   f :: (Integral a) =>
20        Ratio a -> Ratio a -- Real?
21
22 > -- Function-modular, now our modp function is wrapped
       by Maybe.
23 > fmodp :: (a -> Ratio Int) -> Int -> a -> Maybe Int
24 > f `fmodp` p = (`modp` p) . f
```

```
25
26    *FROverZp> let f x = (1%3) + (3%5)*x + (7%13)*x^2
27    *FROverZp> take 10 $ map (f `fmodp` 13)  [0..]
28    [Just 9,Nothing,Nothing,Nothing,Nothing,Nothing,Nothing
         ,Nothing,Nothing,Nothing]
29    *FROverZp> take 10 $ map (f `fmodp` 19)  [0..]
30    [Just 13,Just 8,Just 7,Just 10,Just 17,Just 9,Just 5,
         Just 5,Just 9,Just 17]
31
32  Since we have chosen
33    *Ffield> last bigPrimes
34    10939
35  we can put a finite number of outputs as our accessible "
       row" data.
36
37  Difference analysis over Z_p
38  Every arithmetic should be on Z_p, i.e., (`mod` p).
39
40  > accessibleData :: (Num a, Enum a) =>
41  >                   (a -> Ratio Int) -> Int -> [Maybe Int
       ]
42  > accessibleData f p = take p $ map (f `fmodp` p) [0..]
43  >
44  > accessibleData' :: [Ratio Int] -> Int -> [Maybe Int]
45  > accessibleData' fs p = take p $ map (`modp` p) fs
46
47    *FROverZp> let helper x y = (-) <$> x <*> y
48    *FROverZp> :type helper
49    helper :: (Applicative f, Num b) => f b -> f b -> f b
50    *FROverZp> :t map helper
51    map helper :: (Applicative f, Num b) => [f b] -> [f b
         -> f b]
52    *FROverZp> let myDif xs = zipWith helper (tail xs) xs
53    *FROverZp> myDif [Just 5,Just 0,Just 2,Just 4,Just 6,
         Just 1,Just 3]
54    [Just (-5),Just 2,Just 2,Just 2,Just (-5),Just 2]
55    *FROverZp> map (fmap (`mod` 13)) it
56    [Just 8,Just 2,Just 2,Just 2,Just 8,Just 2]
57    *FROverZp> map (fmap (`mod` 13)) . myDif $ [Just 5,Just
          0,Just 2,Just 4,Just 6,Just 1,Just 3, Nothing,
         Just 1, Just 2]
58    [Just 8,Just 2,Just 2,Just 2,Just 8,Just 2,Nothing,
         Nothing,Just 1]
59
60  > -- difsp :: (Applicative f, Integral b) => b -> [f b]
```

```
    -> [f b]
> difsp :: Applicative f => Int -> [f Int] -> [f Int]
> difsp p = map (fmap (`mod` p)) . difsp'
>   where
>     difsp' :: (Applicative f, Num b) => [f b] -> [f b]
>     difsp' xs = zipWith helper (tail xs) xs
>     helper :: (Applicative f, Num b) => f b -> f b -> f
    b
>     helper x y = (-) <$> x <*> y

  *FROverZp> difsp 13 [Just 5,Just 0,Just 2,Just 4,Just
     6,Just 1,Just 3, Nothing, Just 1, Just 2]
  [Just 8,Just 2,Just 2,Just 2,Just 8,Just 2,Nothing,
     Nothing,Just 1]

> difListsp :: (Applicative f, Eq (f Int)) => Int -> [[f
    Int]] -> [[f Int]]
> difListsp _ [] = []
> difListsp p xx@(xs:xss) =
>   if isConst xs
>     then xx
>     else difListsp p $ difsp p xs : xx
>   where
>     isConst (i:jj@(j:js)) = all (==i) jj
>     isConst _ = error "difListsp:␣"

  *FROverZp> let f x = (1%3) + (3%5)*x + (7%13)*x^2
  *FROverZp> let ds = accessibleData f 101
  *FROverZp> map head $ difListsp 101 [ds]
  [Just 71,Just 26,Just 34]

Degree, eager and lazy versions

> degreep' :: (Applicative f, Eq (f Int)) => Int -> [f
    Int] -> Int
> degreep' p xs = length (difListsp p [xs]) - 1
>
> degreepLazy :: (Applicative f, Num t, Eq (f Int)) =>
    Int -> [f Int] -> t
> degreepLazy p xs = helper xs 0
>   where
>     helper as@(a:b:c:_) n
>       | a==b && b==c = n -- two times matching
>       | otherwise    = helper (difsp p as) (n+1)
>
```

```
 99 > degreep :: (Applicative f, Eq (f Int)) => Int -> [f Int
      ] -> Int
100 > degreep p xs = let l = degreepLazy p xs in
101 >   degreep' p $ take (l+2) xs
102
103   *FROverZp> let f x = (1%3) + (3%5)*x + (7%13)*x^2
104   *FROverZp> degreep 101 $ accessibleData f 101
105   2
106   *FROverZp> degreep 103 $ accessibleData f 103
107   2
108   *FROverZp> degreep 107 $ accessibleData f 107
109   2
110   *FROverZp> degreep 11 $ accessibleData f 11
111   2
112   *FROverZp> degreep 13 $ accessibleData f 13
113   1
114   *FROverZp> degreep 17 $ accessibleData f 17
115   2
116
117 > firstDifsp :: (Applicative f, Eq (f Int)) => Int -> [f
      Int] -> [f Int]
118 > firstDifsp p xs = reverse $ map head $ difListsp p [xs
      ']
119 >   where
120 >     xs' = take n xs
121 >     n   = 2 + degreep p xs
122 >
123 > makeAPair :: [Int] -> [Ratio Int] -> [(Int, Maybe [Int
      ])]
124 > makeAPair ps fs = zip ps . map (sequence . (\p ->
      firstDifsp p (fsp p))) $ ps
125 >   where
126 >     fsp = accessibleData' fs
127
128   *FROverZp> let f x = (1%3) + (3%5)*x + (7%13)*x^2
129   *FROverZp> let fs = map f [0..]
130   *FROverZp> let smallerPrimes = filter isPrime [11, 13,
        19, 23, 101, 103, 107]
131   *FROverZp> makeAPair smallerPrimes  fs
132   [(11,Just [4,3,7])
133   ,(13,Nothing)
134   ,(19,Just [13,14,4])
135   ,(23,Just [8,16,17])
136   ,(101,Just [34,26,71])
137   ,(103,Just [69,36,9])
```

```
138    ,(107,Just [36,39,34])
139    ]
140
141    *FROverZp>  makeAPair smallerPrimes  fs
142    [(11,Just [4,3,7])
143    ,(13,Nothing)
144    ,(19,Just [13,14,4])
145    ,(23,Just [8,16,17])
146    ,(101,Just [34,26,71])
147    ,(103,Just [69,36,9])
148    ,(107,Just [36,39,34])
149    ]
150    *FROverZp> filter (isJust . snd) it
151    [(11,Just [4,3,7])
152    ,(19,Just [13,14,4])
153    ,(23,Just [8,16,17])
154    ,(101,Just [34,26,71])
155    ,(103,Just [69,36,9])
156    ,(107,Just [36,39,34])
157    ]
158    *FROverZp> map (\(p, xs) -> (zip (sequence xs) (repeat
          p))) it
159    [[(Just 4,11),(Just 3,11),(Just 7,11)]
160    ,[(Just 13,19),(Just 14,19),(Just 4,19)]
161    ,[(Just 8,23),(Just 16,23),(Just 17,23)]
162    ,[(Just 34,101),(Just 26,101),(Just 71,101)]
163    ,[(Just 69,103),(Just 36,103),(Just 9,103)]
164    ,[(Just 36,107),(Just 39,107),(Just 34,107)]
165    ]
166    *FROverZp> transpose it
167    [[(Just 4,11),(Just 13,19),(Just 8,23),(Just 34,101),(
          Just 69,103),(Just 36,107)]
168    ,[(Just 3,11),(Just 14,19),(Just 16,23),(Just 26,101),(
          Just 36,103),(Just 39,107)]
169    ,[(Just 7,11),(Just 4,19),(Just 17,23),(Just 71,101),(
          Just 9,103),(Just 34,107)]
170    ]
171    *FROverZp> :t it
172    it :: [[(Maybe Int, Int)]]
173    *FROverZp> :t reconstruct
174    reconstruct :: [(Maybe Int, Int)] -> Maybe (Ratio
          Integer, Integer)
175    *FROverZp> map reconstruct it
176    [Just (1 % 3,209),Just (74 % 65,485507),Just (14 %
          13,209)]
```

```
177
178   *FROverZp> let f x = (1%3) + (3%5)*x + (7%13)*x^2
179   *FROverZp> let fs = map f [0..]
180   *FROverZp> list2npol fs
181   [1 % 3,74 % 65,7 % 13]
182
183 > -- firstDifs' :: [Ratio Int] -> [Maybe (Ratio Integer,
        Integer)]
184 > -- firstDifs' = map reconstruct . transpose . map (\(p,
        xs) -> (zip (sequence xs) (repeat p))) .
185 > --               filter (isJust .snd) . makeAPair
        bigPrimes
186
187   *FROverZp> let g x = 1%153 + x*(133%122) + (x^2)
        *(1%199) + (x^3)*(922%855)
188   *FROverZp> let gs = map g [0..]
189   *FROverZp> fmap (newtonC . map fst) . sequence .
        firstDifs' $ take 100 gs
190   Just [1 % 153,45117911 % 20757690,183763 % 56715,922 %
        855]
191   *FROverZp> list2npol gs
192   [1 % 153,45117911 % 20757690,183763 % 56715,922 % 855]
193
194 > -- Eager-version, i.e., input should be finite.
195 > -- list2npolp :: [Ratio Int] -> Maybe [Ratio Integer]
196 > -- list2npolp = fmap (newtonC . map fst) . sequence .
        firstDifs'
197
198   *FROverZp> let g x = 1%153 + x*(133%122) + (x^2)
        *(1%199) + (x^3)*(922%855)
199   *FROverZp> let gs = map g [0..100]
200   *FROverZp> list2npol gs
201   [1 % 153,45117911 % 20757690,183763 % 56715,922 % 855]
202   *FROverZp> list2npolp gs
203   Just [1 % 153,45117911 % 20757690,183763 % 56715,922 %
        855]
204
205
206 --
207 Univariate Rational function case
208 Since thiele2coef uses only (*), (+) and (-) operations,
      we don't have to do these calculation over prime
      fields.
209 So, our target should be rho function (matrix?)
      calculation.
```

```
210
211 Reciprocal difference
212
213 > {-
214 > rhoZp :: Integral a => [Ratio a] -> a -> Int -> a -> a
215 > rhoZp fs 0 i p = (fs !! i) `modp` p
216 > rhoZp fs n i p
217 >   | n <= 0     = 0
218 >   | otherwise = (n*inv + rhoZp fs (n-2) (i+1) p) `mod`
      p
219 >   where
220 >     inv = fromJust inv'
221 >     inv'= (rhoZp fs (n-1) (i+1) p - rhoZp fs (n-1) i p)
      `inversep` p
222 >
223 > aZp :: Integral a => [Ratio a] -> a -> a -> a
224 > aZp fs 0 p = head fs `modp` p
225 > aZp fs n p = (rhoZp fs n 0 p - rhoZp fs (n-2) 0 p) `mod
    ` p
226 >
227 > tDegreeZp fs p = helper fs 0 p
228 >   where
229 >     helper fs n p
230 >       | isConst fs' = n
231 >       | otherwise   = helper fs (n+1) p
232 >       where
233 >         fs' = map (rhoZp fs n p) [0..]
234 >     isConst (i:j:_) = i==j
235
236   *FROverZp> let h t = (3+6*t+18*t^2)%(1+2*t+20*t^2)
237   *FROverZp> let hs = map h [0..]
238   *FROverZp> take 5 $ map (\n -> rhoZp hs 0 n 101) [0..]
239   [3,89,64,8,16]
240   *FROverZp> take 5 $ map (\n -> rhoZp hs 1 n 101) [0..]
241   [74,4,9,38,65]
242   *FROverZp> take 5 $ map (\n -> rhoZp hs 2 n 101) [0..]
243   [*** Exception: Maybe.fromJust: Nothing
244
245 > -}
```